
      While attending college in the early 1980s, I wrote my first program of personal significance (a never-published text adventure game called Stowaway to Oblivion). It was fun to develop, but not particularly sophisticated. Written in BASIC on an Apple II, it used a simple verb–noun parser for player commands. Although I was pleased with what I’d accomplished, I still felt that my code was not very elegant, and wondered how text adventures such as Zork (which had a much more sophisticated parser) were implemented. I stored the program away with the desire to revisit the code one day and create a better solution.

      After graduating, I was hired by NASA and assigned to work on a rule-based programming language called CLIPS. Starting in 1986, this language became available to groups outside of NASA and in a few years was in widespread use. While the three volume CLIPS Reference Manual was somewhat dry (as one might expect for technical material), the CLIPS User’s Guide, a tutorial written by Dr. Joseph Giarratano, was often praised by readers for its light hearted introduction to rule-based programming.

      In the late 1980s, Dr. Giarratano and I coauthored a textbook, Expert Systems: Principles and Programming, with a target audience of college level students in an artificial intelligence course. The first half of the book, written by Dr. Giarratano, covered expert system theory, and the second half, which I wrote, covered expert system development using CLIPS.

      As CLIPS gained new features, the CLIPS User’s Guide was updated. At its largest size, it was a hefty 600 pages. That was a bit excessive for a tutorial, so when the last major update to its content was made in 1993, it was pared down to a quarter of that size. Unfortunately, it’s hard to trim so many pages without losing the cohesion of the original content. Dr. Giarratano and I also continued to update our textbook. The fourth edition, released in 2005, was over 800 pages (and even with just the CLIPS portion, it was still too long for a tutorial).

      Amidst all this activity, I started to consider that rule-based programming might be the solution I was seeking for my text adventure. I toyed with implementation approaches in the early 1990s, but wasn’t sufficiently motivated, so I didn’t get very far in revisiting my code.

      I left NASA in 1996, but continued to independently support CLIPS as public domain software. Over the years, I’ve answered a lot of CLIPS questions on various internet forums and as a result thought a lot about the best approach to teach new users about CLIPS. A few years back, I started thinking about writing a new tutorial to incorporate my ideas. The challenge would be writing something short and fun to read (like the original User’s Guide) that would also cover the most important steps in developing a CLIPS application. It occurred to me that revisiting my text adventure would allow me to achieve both objectives.

      Thanks to the internet, I was pleasantly surprised to discover that even though the commercial market for text adventures had collapsed several decades ago, there was still an active community for these games. Not only that, but there were tools for developing them. After studying their implementation approaches, I was convinced that a text adventure could be elegantly written using CLIPS.

      And that’s how this book came to be, from a program I wrote in college over thirty-five years ago.

      Annoyingly, sometime, somewhere, I lost the source code for Stowaway to Oblivion, but at least this book allowed me to revisit the spirit of that adventure.

      This book will teach you how to build a CLIPS application from start to finish. It will cover not just the syntax and semantics of a CLIPS program, but also how to structure, debug, and test your code. It will start by building a very simple program, and as you progress through the chapters, the functionality and complexity of the code will be incrementally increased. Once you’ve completed this tutorial, you will have a small, yet fully functional, text adventure game.

      This is a hands-on tutorial. It is designed with the expectation that you will be creating code and interacting with CLIPS as you read through the chapters. Instructions for downloading and installing CLIPS are in the first chapter. Each chapter has listings of all interactions, however, so if you choose you can still read through the tutorial without having to install CLIPS. For your convenience, the source code described for each chapter can be downloaded from clipsrules.net/airbp.html. The various stages of the developed source code are stored in separate directories.

      Throughout this book, different fonts are used to help convey meaning. References to code, language keywords, and function names use a monospace font. Input that is to be typed, additions to existing code, and the first use of keywords and function names use a bold monospace font. Other new terms that are introduced use a bold serif font. Deletion of existing code is denoted with a strikethrough monospace font. References to menus, menu items, files, directories, and web pages use a sans serif font.

      This tutorial does not attempt to cover all of the features provided by CLIPS; however, it does cover the most important features to learn first. The reference manuals, particularly the Basic Programming Guide, will be useful as you read through this tutorial and want additional details about a specific feature. If you install one of the Integrated Development Environments for CLIPS, selecting the Online Documentation menu item in the Help menu will launch your web browser and take you to a web page with links to the CLIPS reference manuals and other documentation.

      What is Declarative Programming?

      The first programming language most people learn is an imperative programming language. This includes languages such as C, C++, C#, Java, and Python. With imperative programming, you not only specify the logic of what must be done, but also the specific sequence of steps that will control the flow of how it is done.

      With declarative programming, you specify only the logic of what must be done; the specific sequence of steps that control the flow of how it is done do not need to be specified.

      As an example of declarative logic, consider some guidelines for bagging groceries:

      
        	Pack light or fragile items separately or on top of heavier items.

        	Distribute heavy items among several bags.

        	Pack items of similar temperature together.

        	Pack food items separate from non-food items.

        	Pack raw foods separate from foods that must be cooked.

      

      These guidelines specify what needs to be done, but not how to do it. There’s no indication of when a new bag should be opened or how to select the next item for bagging. By examining bags of packed groceries, it can be determined if the guidelines have been correctly applied, but there are multiple ways the groceries can be correctly packed (including the rather inefficient method of placing just one item in each bag). Generally if the groceries are packed in a reasonable number of bags, it’s not a concern whether the groceries are packed in the fewest number of bags possible. In fact, packing groceries in the fewest number of bags may not be “optimal” since some people may prefer a larger number of lighter bags.

      With a general understanding of how to pack things, instead of packing groceries, a different set of guidelines could be used for packing boxes for shipment or packing a suitcase for travel. That is one of the benefits of declarative programming. The “how” something is done can be reused to solve a different problem by just specifying “what” needs to be done.

      What is Rule-Based Programming?

      Rule-based programming is a form of declarative programming where a rule is the primary unit of code. A rule embodies a set of conditions and the actions to be taken when those conditions are satisfied:

      
        
        When

           Conditions

        Then

           Actions

      

      

      It’s important to note the use of when/then rather than if/then. With the imperative if/then, the actions are processed if the conditions are true when program control reaches the if/then statement. With the declarative when/then, the actions can be processed whenever the conditions become true.

      Typically, rules make sense as units of code when you have numerous rules with similar structure. For example, diagnosis rules would have symptoms or test results as conditions, and tests to run or a diagnosis as the actions:

      
        
        Measles rule

           When

              Patient has high fever and

              Patient has cough and

              Patient has runny nose and

              Patient has red, watery eyes and

              Patient has rash

           Then

              Diagnose illness as measles

      

      

      Even if your program can be expressed using rules as units of code, it’s natural to question why you would want to use an esoteric rule-based language instead of an imperative programming language such as Java or C#. There are several good reasons.

      First, ownership of the program code may need to be distributed. There may be one or more architects who provide the basic application framework, but the creation and maintenance of application logic is distributed among many contributors. For example, a shipping company might provide an event notification system that allows their customers to create logic for receiving notifications when certain events occur (such as freight being loaded); or a company might have a pricing system that allows different departments to create and edit pricing logic for their products. In both cases, individual contributors cannot be expected to manage the flow of control for logic  they haven’t authored. Using declarative logic allows the burden of flow of control to be shifted from the users to the programming language. The separation of flow of control from program logic also makes it easier to add and remove rules with minimal impact on other program logic. For example, in a pricing system, using declarative programming makes it easier to add or remove pricing logic when products are launched or discontinued.

      Second, if the program requirements have been expressed using a rule-like format, then maintaining the program as rules makes it easier to see the mapping between the requirements and the code that implements the requirements. Additionally, converting declarative requirements to imperative code is burdensome if rules are frequently added or deleted.

      Finally, if your program is naturally expressed using rules, but you implement it using an imperative programming paradigm, you’ll likely end up reinventing much of the functionality already provided by a rule-based programming language.

      How Rules Work

      Rule-based systems are comprised of three components: data, rules, and an inference engine.

      The data component is a collection of global objects that represent the current state of the program. This data may be represented in the system using internally defined data structures or may be be represented as native objects of a language such as Java.

      The rules component is a collection of rules, each of which has a set of conditions and actions. When the conditions of a rule are satisfied, its actions can be applied. The rules represent the declarative logic of what should be done. The conditions of a rule are also referred to as the antecedent or left hand side. The actions of a rule are also referred to as the consequent or right hand side.

      The inference engine determines how the declarative logic is processed. First, it determines which rules have their conditions satisfied by the data. It then determines which rule should be selected to have its actions applied. This process is then repeated until there are no rules remaining to execute.

      Using the packing groceries example, the data would be the grocery items to pack, the rules would be the rules for packing the items (such as packing items with similar temperatures together), and the inference engine would handle the mechanics of selecting the item to pack next and when a new bag should be opened.

      Rule-based programming is classically used for developing expert systems, programs that emulate human experts in a specific domain. The pattern matching capabilities of rules, however, make them useful for a variety of applications, so they are not strictly limited to expert systems.

      Inference engines can be categorized by whether they use forward chaining, backward chaining, or a combination of both. Forward chaining systems start with premises and work forward to the conclusions supported by those premises. Backward chaining systems start with a conclusion to be proved and work backward to the premises that would support that conclusion.

      There are a wide variety of expert system applications. Some are more easily solved using forward chaining and some more easily using backward chaining, but it’s possible to solve these applications using either approach.

      Text Adventures

      One of the earliest genres of computer games were text adventures. In this form of interactive fiction, players interact with a virtual world using text commands that allow them to explore, solve puzzles, and complete goals.

      The sophistication of the command parsers ranged from the simple to the complex. Adventure International1 games accepted just verb–noun combinations, and only the first three characters of each word were used for matching, so the commands scream bear and screw bear were treated identically.2 The parser for Infocom games was much more sophisticated and allowed complete sentences. Zork, one of the most well-known Infocom games, can still be played online.3 The popularity of commercial text adventures peaked in the 1980s, and today most commercial interactive fiction uses a graphical interface.

      This listing shows typical player/game interaction in Zork:

      
        
        West of House

        You are standing in an open field west of a white house, with a boarded front door.

        There is a small mailbox here.

      

      

      
        
        > walk west

        Forest

        This is a forest, with trees in all directions. To the east, there appears to be sunlight.

      

      

      
        
        > go north

        Clearing

        You are in a clearing, with a forest surrounding you on all sides. A path leads south.

        On the ground is a pile of leaves.

      

      

      
        
        > move the pile of leaves

        Done.

        In disturbing the pile of leaves, a grating is revealed.

      

      

      
        
        > open the grate

        The grating is locked.

      

      

      
        
        >

      

      

      Despite the lack of commercial prospects for text adventures, there exist a number of tools for creating them:

      
        	Inform (inform7.com)

        	Quest (textadventures.co.uk/quest)

        	Twine (twinery.org)

        	Adrift (adrift.co)

        	Ink (inklestudios.com/ink)

      

      Inform 7 is the most interesting of these tools since it is a declarative programming language. The previous interaction in Zork can by recreated with this Inform 7 source code:

      
        
        The description of West of House is "You are standing in an open field west of a white house, with a boarded front door."

        The small mailbox is in West of House.

      

      

      
        
        West of West of House is Forest.

        The description of Forest is "This is a forest, with trees in all directions. To the east, there appears to be sunlight."

      

      

      
        
        North of Forest is Clearing.

        Clearing has description "You are in a clearing, with a forest surrounding you on all sides. A path leads south."

        A pile of leaves is in Clearing.

      

      

      
        
        A secret door is a kind of door.

        A secret door can be revealed or unrevealed.

        A secret door is unrevealed.

        A secret door is scenery.

      

      

      
        
        Understand "grate" as grating.

        The grating is a secret door. The grating is down from Clearing.

        The grating is locked and lockable.

        After taking or pushing or pulling the leaves for the first time: say "In disturbing the pile of leaves, a grating is revealed."; now the grating is revealed.

      

      

      Text adventures are a natural fit for a declarative programming language. The location descriptions, paths between locations, objects with which the player can interact, and command effects can be expressed declaratively. The general mechanics of an adventure (such as parsing commands, moving between locations, examining items, and picking up or dropping items) can be implemented once and reused.

      For this tutorial, a small text adventure called Escape from the Goblin Pit will be created. When fully implemented, you’ll be able to have this interaction with the game:

      
        
        Captured by goblins, you've been

        tossed in a pit at their lair.

      

      

      
        
        You're at the pit's north end.

        A giant mushroom is here. The

        ground is littered with the

        bodies of dead adventurers.

      

      

      
        
        > look at the mushroom

        It's squished. I wouldn't

        try landing on it again.

      

      

      
        
        > climb up

        The walls are too slick.

      

      

      
        
        > go south

        You're at the pit's south end.

        A large pile of rubble has

        collapsed from the wall above.

      

      

      
        
        > search the rubble

        You find a dead goblin.

        He probably fell from above.

      

      

      
        
        > search the dead goblin

        Some beans are in his pockets.

      

      

      
        
        > get beans

        You take the beans.

      

      

      
        
        > plant the beans

        The ground is too hard.

      

      

      
        
        > north

        You're at the pit's north end.

        A giant mushroom is here. The

        ground is littered with the

        bodies of dead adventurers.

      

      

      
        
        > plant the beans

        Apparently the beans are magical.

        They grow into a giant stalk.

      

      

      
        
        > climb stalk

        You have escaped!

      

      

      CLIPS

      The C Language Integrated Production System (CLIPS) is a rule-based programming language that was originally developed at NASA’s Johnson Space Center. It is now available as public domain software and can be downloaded from this web page:

      
        
        sourceforge.net/projects/clipsrules/files/CLIPS/6.40

      

      

      Windows users should download one of the following two installers (depending upon whether you’re using a 32- or 64-bit version of Windows):

      
        
        clips_windows_32_bit_installer_640.msi

        clips_windows_64_bit_installer_640.msi

      

      

      Double click the installer to launch it. Once installed, you can launch the CLIPS Integrated Development Environment (IDE) by opening the Start menu, clicking on the CLIPS 6.4 folder, and then clicking CLIPSIDE.

      To install on macOS, download the following file:

      
        
        clips_macos_executable_640.dmg

      

      

      Double click the dmg file to open it, and then drag the CLIPS IDE application into the Applications folder. Once installed, you can launch CLIPS by clicking on Launchpad in the dock and then double clicking on the CLIPS IDE application.

      For other operating systems (such as Linux), you must compile CLIPS as a terminal application. To do so, first download one of the following two files containing the CLIPS source code:

      
        
        clips_core_source_640.zip

        clips_core_source_640.tar.gz

      

      

      Typically you can extract the contents of a .zip file by double clicking it (for macOS) or by right clicking on it and selecting the extract menu option (for Windows or Linux). Similarly, you can extract the contents of a .tar.gz file by double clicking it (for macOS) or by clicking on it and selecting the extract menu option (for Linux).

      Once extracted, you can launch the terminal application for your operating system, navigate to the core directory of the CLIPS source code, and execute a make command:

      
        
        .../clips_core_source_640/core$ make

           .

           .

           .

        .../clips_core_source_640/core$

      

      

      Once compilation is complete, you can launch the clips executable by executing a ./clips command:

      
        
        .../clips_core_source_640/core$ ./clips

                 CLIPS (6.4 2/9/21)

        CLIPS> (exit)

        .../clips_core_source_640/core$

      

      

      These same steps can be used to create a CLIPS terminal application for macOS. In addition to the IDE, the Windows installer also installs a CLIPS terminal application. You can launch this application by opening the Start menu, clicking on the CLIPS 6.4 folder, and then clicking CLIPSDOS.

      Because the IDE provides a number of useful features, it is preferable to use it rather than the terminal application while learning CLIPS. For systems other than Windows and macOS, you can use a cross-platform CLIPS IDE written in Java Swing. To use this IDE, download one of the CLIPSJNI source code files:

      
        
        clips_jni_640.zip

        clips_jni_640.gz.tar

      

      

      In addition, download one of the files containing the CLIPS documentation:

      
        
        clips_documentation_640.zip

        clips_documentation_640.gz.tar

      

      

      Once you’ve extracted the contents of both files, read the CLIPS Interfaces Guide for instructions on compiling and running the Java Swing CLIPS IDE.

      Read Eval Print Loop

      In developing CLIPS programs, you’ll interact with the Read-Eval-Print Loop (REPL). When you start CLIPS, you’ll be greeted with a command prompt:

      
        
        CLIPS>

      

      

      If CLIPS is run as a terminal application, you’ll see the command prompt appear in the terminal window; otherwise, if you’re using one of the IDEs, the command prompt will appear in its own window.

      If you’ve used LISP (as well as some other languages such as Python), you’ll already be familiar with a REPL. If you’ve used a DOS command prompt or a Unix shell, you’ll also be familiar with the concept.

      The command prompt reads the input you type, evaluates it, prints the result, and then repeats the process (hence the Read-Eval-Print Loop). As an example, try entering a number at the command prompt and press the return key:

      
        
        CLIPS> 738

        738

        CLIPS>

      

      

      After you enter the value 738 and press the return key, CLIPS analyzes the input you’ve typed and determines whether it can evaluate what’s been entered. In many cases when you press the return key, the input cannot be evaluated (as it’s possible to have values or code spread across multiple lines). In this case, 738 is an integer, one of the primitive data types recognized by CLIPS, so the input can be evaluated. The result of evaluating a primitive data type is the value of that data type, so 738 is printed as the result of the evaluation.

      Here’s an example of entering a string (another primitive data type) at the command prompt:

      
        
        CLIPS>

        "a string with

        a carriage return"

        "a string with

        a carriage return"

        CLIPS>

      

      

      In this case, immediately pressing the return key at the command prompt does not cause an evaluation of the input. A string is specified by enclosing a sequence of characters between quotation marks. When the return key is pressed after entering "a string with, a closing quotation mark has not been entered so the input is not evaluated. It’s only when the closing quotation mark has been entered and the return key is pressed that the input is evaluated. Again, since a string is one of the CLIPS primitive data values, the result of the evaluation is simply the string.

      Functions calls can also be entered at the CLIPS command prompt. As in other languages, functions are procedural code that can be invoked with a number of input values. When executed, they perform a task such as opening a file, reading input, or computing a mathematical value. When complete, they may or may not return a value displayed as output.

      In this tutorial and other CLIPS documentation, you’ll frequently see the words function, command, and action used interchangeably; however, there are no practical differences between the three. If a function is typically entered at the command prompt (such as functions for loading and debugging code), it’s usually referred to as a command; otherwise it’s referred to as a function, or if used in the actions of a rule, an action.

      Since functions can be called directly from the command prompt, the traditional “Hello, world!” program can be implemented using the println function.

      
        
        CLIPS> (println "Hello, world!")

        Hello, world!

        CLIPS>

      

      

      The println (short for “print line”) function outputs each of its arguments to standard output (the screen of your computer) and then outputs the carriage return/line feed characters so that subsequent output begins on the next line.

      The syntax of function calls may appear a little odd if you’ve never used a language with prefix notation. CLIPS was originally designed to be syntax compatible with a programming language written in LISP (an acronym derived from LISt Processing). The use of parentheses to represent expressions as well as data allows LISP to treat code as data, which is pretty cool when you think about it, but it has also caused many to say that LISP is actually an abbreviation for Lost In Stupid Parentheses or Lots of Irritating Superfluous Parentheses.

      So if you’re used to function calls with syntax like

      
        
        println("Hello, world!");

      

      

      all you need to do to convert it to CLIPS syntax is remove the semicolon, move the left parenthesis in front of the function name, and delimit arguments using spaces rather than commas:

      
        
        (println "Hello, world!")

      

      

      CLIPS also uses a prefix notation for arithmetic and boolean operators, rather than infix operators. For example, the following expression uses infix operators:

      
        
        (3 + 4) < 6

      

      

      In CLIPS this would be expressed as:

      
        
        (< (+ 3 4) 6)

      

      

      You can exit CLIPS using the exit command:

      
        
        CLIPS> (exit)

      

      

      In subsequent chapters, you’ll see the command prompt used to load programs, start execution, and step through code. When you deploy a CLIPS application, you’ll typically embed CLIPS within another program, or you’ll start execution of CLIPS using a batch file (and in both cases the user won’t interact with the command prompt).

      All of the CLIPS IDEs support a command history, so you can use the up and down arrows to cycle through previous commands. You can also edit text or insert text anywhere within the current command. If you run CLIPS as a terminal application, no command history is supported, and once you’ve pressed the return key you can’t edit prior lines of a multiline command.

      What’s Next

      Now that you have CLIPS installed and know the basics of interacting with it, you’ll start building Escape from the Goblin Pit in the next chapter.

      The Defrule Construct

      The building blocks of a CLIPS program are referred to as constructs. Like a function call, constructs utilize parentheses as delimiters. However, constructs persist until explicitly removed, while function calls are transitory (although they can have effects that persist, such as opening a file).

      Six constructs will be covered in this book: defrule, deftemplate, deffacts, deffunction, defglobal, and defmodule. As you may have noticed, all of the construct keywords use the prefix ‘def’, which is short for ‘define’, and so a construct type such as defrule is just an abbreviation for ‘define rule’.

      Escape from the Goblin Pit will begin with a rule created using the defrule construct. Although you’ll typically define constructs by loading them from a file, they can also be entered at the command prompt:

      
        
        CLIPS>

        (defrule exposition

           =>

           (println "Captured by goblins, you've been")

           (println "tossed in a pit at their lair."))

        CLIPS>

      

      

      All constructs begin with a left parenthesis that is followed by the keyword for the construct type. For the defrule construct, the construct type is followed by the rule name. Construct names must be symbols (another CLIPS primitive data type). You can think of a symbol as a string that doesn’t require quotation marks, but can’t contain any white space (such as spaces or tabs). Basically anything that’s a valid variable or function name in languages with syntax similar to C or Java is also a CLIPS symbol, but there are additional characters that can be used (such as hyphens). Note that different construct types can make use of the same name, but entering a construct name that already exists for a specific construct type will redefine the existing construct. An optional comment string can be specified after the rule name, but for this example none has been specified.

      The then symbol, =>, separates the conditions of the rule from the actions of the rule. A rule without any conditions (such as this one) can be used to execute actions unconditionally. The actions (one or more function calls) of a rule are also optional, but a rule without any actions is not particularly useful. All construct definitions are closed by a final right parenthesis.

      Unlike functions, constructs never have return values. An improperly defined construct will display an error message (and in some cases, properly defined constructs may generate a warning).

      The Agenda

      When a rule has its conditions satisfied, an activation for the rule is created and added to the agenda (the list of rule activations ready for execution). It’s possible for a rule to have its conditions satisfied in multiple ways, so there can be multiple activations for the same rule. An activation is eligible, but not guaranteed, to have its actions executed. If there are multiple activations on the agenda, the inference engine selects one activation and executes the actions of the associated rule (the details of this selection will be discussed later). Executing a rule is also referred to as ‘firing’ it (like a neuron). Commonly, the actions of a rule can add and/or remove activations from the agenda. Once an activation has been executed, it is removed from the agenda.

      The exposition rule has no conditions, so once defined it is immediately activated and placed on the agenda. The agenda command is used to display the list of activations on the agenda:

      
        
        CLIPS> (agenda)

        0      exposition: *

        For a total of 1 activation.

        CLIPS>

      

      

      For each activation, the salience, name, and matching data for the associated rule is displayed. The salience, which will be explored later, is an integer representing the priority of the rule. By default, the salience of a rule is 0. For rules with no conditions, the character * is used to indicate that the rule is matched, but not by specific data.

      The run command begins execution of activations on the agenda:

      
        
        CLIPS> (run)

        Captured by goblins, you've been

        tossed in a pit at their lair.

        CLIPS>

      

      

      If the run command is called without any arguments, it will continue selecting activations and executing the associated rule until there are no activations remaining. In this case, there is only one rule and it can be matched in only one way, so once it’s executed and its activation is removed from the agenda, there will be no remaining activations so the run command will finish.

      Calling the agenda command again confirms that the agenda is now empty:

      
        
        CLIPS> (agenda)

        CLIPS>

      

      

      Creating Facts

      Facts are the data that match the conditions of a rule. If the adventurer is to find their way out of this unfortunate situation, a fact will be needed to represent their current location. Since there are numerous things that will be represented as facts—the adventurer, places, items that can be taken or manipulated, etc.—it will be useful to create these facts using a shared representation. The deftemplate construct is used to define the structure of facts sharing common attributes:

      
        
        CLIPS>

        (deftemplate thing

           (slot id)

           (slot category)

           (slot location))

        CLIPS>

      

      

      The name of the template and an optional comment string follow the deftemplate keyword. Next, any number of slot definitions can be specified. Each slot stores the value of an attribute for a fact. Slots are specified using the slot and multislot keywords. A slot defined using the slot keyword holds a single value. A slot defined using the multislot keyword holds zero or more values. The name of each slot is a symbol.

      For the thing template, three slots have been defined: id, category, and location. Each of these slots will hold a single value. The id slot will contain a symbol (such as adventurer or pit_north) that uniquely identifies a thing. The category slot will contain a symbol (such as actor or place) classifying things that function similarly. The location slot will contain a symbol (such as pit_north or pit_south) indicating the place or current location of a thing.

      With the thing template defined, a thing fact representing the adventurer can be created:

      
        
        CLIPS>

        (assert (thing (id adventurer)

                       (category actor)

                       (location pit_north)))

        <Fact-1>

        CLIPS>

      

      

      The assert command creates one or more facts. In this case, a single fact is created, but several more could have been added by including them before the closing right parenthesis of the command. The return value of the assert command is a fact-address (a C pointer to the memory for the data structure used by CLIPS to represent a fact), but this is printed as <Fact-#> rather than as a memory address. Each fact created is assigned an integer value called a fact-index, which starts at 1 and is incremented each time a new fact is created. Since this thing fact is the first fact created, it has a fact-index of 1 (hence <Fact-1>). If multiple facts had been created, the return value would have been the fact-address of the last fact created.

      Special Forms

      You may have noticed something different about the syntax of the assert command. Back in Chapter 1, there was an example of one function call nested within another:

      
        
        (< (+ 3 4) 6)

      

      

      In this case, the first argument to the < function is a function call to the + function. Generally when you see a left parenthesis in a function call, whatever immediately follows is the name of a function. So without knowing there’s something different about the command

      
        
        (assert (thing (id adventurer)

                       (category actor)

                       (location pit_north)))

      

      

      you’d expect that thing is a function rather than a template name and that id, category, and location are functions rather than slot names.

      A special form is any function that uses non-standard syntax or semantics for expressing and evaluating a function call. In the case of the assert command, a special syntax is used to specify the type of fact and the slot values that will be asserted when it is evaluated.

      Special forms are a small fraction of the total number of functions in CLIPS, but many of the most important functions that will be introduced in this book are special forms.

      The Fact-List

      The collection of all existing facts is referred to as the fact-list and can be displayed using the facts command:

      
        
        CLIPS> (facts)

        f-1     (thing (id adventurer)

                       (category actor)

                       (location pit_north))

        For a total of 1 fact.

        CLIPS>

      

      

      For each fact, its abbreviated fact-index (e.g. f-1 instead of <Fact-1>) is printed followed by the template name for the fact and a list of its slot values. The facts command accepts additional integer arguments that allow you to specify the starting fact-index, ending fact-index, and maximum number of facts to display. For example, the command

      
        
        (facts 50 90 10)

      

      

      would print the first 10 facts with fact-indices ranging from 50 to 90.

      The output of the facts command places each fact on a single line (unlike the output shown in the prior listing, which was formatted to improve readability). You can use the ppfact (short for pretty print fact) command to display each slot of a single fact on a separate line:

      
        
        CLIPS> (ppfact 1)

        (thing

           (id adventurer)

           (category actor)

           (location pit_north))

        CLIPS>

      

      

      When using one of the CLIPS IDEs, you can select the Fact Browser menu item from the Debug menu to open a dedicated window for displaying the contents of the fact-list. Figure 1-1 shows the fact browser for the Windows IDE.

      The left pane in the browser specifies the module for which facts should be displayed (modules will be covered in Chapter 7). The middle pane allows you to specify which fact should be displayed in the right pane.

      The search bar in the browser allows you to further restrict the facts that are displayed. You can type a fact-index (such as f-1); a single value (such as thing, category, or pit_north); a combination of slot name and value (such as category actor); or a combination of template name, slot name, and slot value (such as thing category actor). After entering the search text, press the return key to update the list of facts displayed.

      When a fact is first created, slots that have not been given an explicit value are assigned a default value. Unchecking the Display Defaulted Values check box allows you to prune out the slot values in the right pane that have been assigned a default value.

      
        
          [image: Fact Browser displaying slot values for a selected fact.]
        

        Figure 1-1
      
      Patterns

      Now that the adventurer is represented by a thing fact, a new rule will print a description of the north side of the pit when the adventurer is located there:

      
        
        CLIPS>

        (defrule pit_north

           (thing (id adventurer)

                  (location pit_north))

           =>

           (println "You're at the pit's north end.")

           (println "A giant mushroom is here. The")

           (println "ground is littered with the")

           (println "bodies of dead adventurers."))

        CLIPS>

      

      

      The conditions of a rule are composed of zero or more conditional elements (or CEs for short). The exposition rule had no conditional elements. This new rule has a single pattern conditional element:

      
        
        (thing (id adventurer)

               (location pit_north))

      

      

      A pattern conditional element (or pattern for short) is composed of constraints on the values contained in a fact’s slots. If these constraints are satisfied, then the fact matches (and thus satisfies the pattern). A pattern conditional element begins with a template name (which indicates the type of fact to match) and is followed by zero or more slot patterns. A slot pattern begins with a valid slot name for the template followed by zero or more fields containing a field constraint.

      Slot patterns may be specified in any order regardless of the order in which they were defined in their associated deftemplate. The order of field constraints in a slot pattern, however, is important. Each field constraint in a slot pattern must match to a sequence of values from a fact’s slot for a match to occur. The comparison of fields from a slot pattern to the values in a fact’s slot occurs from left to right (which is why the ordering of the fields in a slot pattern is important).

      Field constraints can either be single-field constraints or multifield constraints. A slot pattern for a slot defined using the slot keyword must contain exactly one single-field constraint. A slot pattern for a slot defined using the multislot keyword may contain any combination of single-field or multifield constraints (including none at all).

      The simplest type of single-field constraint is a literal constraint, where the slot value of the fact must match the literal value in the pattern. For the thing pattern this means that the id slot value for the fact must be adventurer and the location slot value must be pit_north.

      Slots that are not specified in a pattern can have any value. The thing pattern has no constraints for the category slot, so the category slot value contained in a thing fact will not be considered when determining whether the pattern is matched. The single-field wildcard constraint, specified using the ? (question mark) character, can also be used to explicitly indicate that any value contained in a fact’s slot will satisfy the field constraint. The thing pattern could be rewritten with a single-field wildcard to achieve the same behavior:

      
        
        (thing (id adventurer)

               (category ?)

               (location pit_north))

      

      

      In practice, however, it’s better not to do this since it can obscure the important parts of the pattern and is cumbersome if there are a large number of slots.

      The agenda can be checked to verify that this new rule has been activated:

      
        
        CLIPS> (agenda)

        0      pit_north: f-1

        For a total of 1 activation.

        CLIPS>

      

      

      The agenda command shows that the pit_north rule has been activated by the fact f-1, which was asserted before the rule was created. Typically, you’ll load your rules first and then create facts, but you can do the opposite as well. The inference engine is always working in the background, so you can interactively create some facts first and then add rules, or add the rules and then create the facts.

      If this rule is allowed to execute, a description of the north side of the pit will be printed, and then the agenda will be empty:

      
        
        CLIPS> (run)

        You're at the pit's north end.

        A giant mushroom is here. The

        ground is littered with the

        bodies of dead adventurers.

        CLIPS> (agenda)

        CLIPS>

      

      

      The Matches Command

      Commands are an integral part of a text adventure. The player issues commands, the adventurer attempts to follow them, and the results are displayed. Commands will be represented using a new deftemplate:

      
        
        CLIPS>

        (deftemplate command

           (multislot action))

        CLIPS>

      

      

      In the command deftemplate, the multislot keyword specifies that the action slot can hold any number of values (including none at all). This will allow commands containing multiple symbols, such as south, climb up, or look at mushroom.

      The first obstacle the adventurer will discover is that the walls of the pit are too slick to climb. When the adventurer is given a climb up command at the north side of the pit, the player will be notified of this predicament by a new rule:

      
        
        CLIPS>

        (defrule no_escape

           (thing (id adventurer)

                  (location pit_north))

           (command (action climb up))

           =>

           (println "The walls are too slick."))

        CLIPS>

      

      

      The first pattern is matched if the adventurer is at the north side of the pit. The second pattern is matched if the values in the action slot of the command fact are climb up. When matching a multislot, each constraint in the multislot pattern must match to a value in the fact’s slot and vice versa. So if the action slot contains climb or climb up walls, this pattern would not be matched.

      An agenda command reveals that this rule is not yet activated:

      
        
        CLIPS> (agenda)

        CLIPS>

      

      

      The first pattern of this rule should be matched because it’s identical to the first pattern of the pit_north rule. But since no command facts have been created yet, the command pattern can’t be matched.

      When there’s only one pattern and one fact, it’s easy to determine why a rule is or isn’t matched. But as the number of patterns and facts increases, this task becomes more difficult. The matches command simplifies this task:

      
        
        CLIPS> (matches no_escape)

        Matches for Pattern 1

        f-1

        Matches for Pattern 2

        None

        Partial matches for CEs 1 - 2

        None

        Activations

        None

        (1 0 0)

        CLIPS>

      

      

      There are several parts to the output from this command. First, for every pattern conditional element in the rule, the facts that match the pattern are printed. In this case, f-1 matches the first pattern and there are no matches for the second pattern.

      Next, the partial matches for the rule are printed. A partial match is a contiguous group of patterns, starting with the first pattern, that have been matched. So a rule with three patterns would have two groups of partial matches: one group for the first two patterns and another for the first three.

      Partial matches show all the combinations of facts that match multiple patterns. So a rule with three matches for the first pattern and two matches for the second pattern could have six partial matches for the first two patterns. As will become clear when variables in patterns are introduced, not all possible combinations of matches to individual patterns will generate partial matches for a group of those patterns. For this rule, since there are no matches for the second pattern, there are no partial matches for the first two patterns.

      After the partial matches, the activations for the rule are printed. Any activations displayed by the matches command will also be displayed by the agenda command. Once the rule associated with an activation executes, it will be removed from the activations section, but it may still appear in the partial matches section if the associated facts still exist. It’s a bit of a misnomer to refer to a complete match of a rule as a partial match, but just remember that displaying the complete matches as well as the activations allows you to see which complete matches have executed.

      Finally, the matches command returns a multifield value containing the number of pattern matches, partial matches, and activations. A multifield value is another of the CLIPS primitive data types and can be thought of as a list or one-dimensional array of primitive data values. Unlike lists in LISP, you cannot have multifield values within multifield values. In this case, the value (1 0 0) indicates that one pattern is matched, there are no partial matches, and there are no activations.

      Because this command can potentially display a large amount of output, you can optionally pass an additional argument to limit the amount of output. Specifying verbose as the second argument displays full output (which is what happens if you only specify the rule name), succinct displays just the counts, and terse displays no output at all (although the command will still return the total counts):

      
        
        CLIPS> (matches no_escape succinct)

        Pattern 1: 1

        Pattern 2: 0

        CEs 1 - 2: 0

        Activations: 0

        (1 0 0)

        CLIPS> (matches no_escape terse)

        (1 0 0)

        CLIPS>

      

      

      Giving the adventurer a climb up command will create an activation for the rule:

      
        
        CLIPS> (assert (command (action climb up)))

        <Fact-2>

        CLIPS> (matches no_escape)

        Matches for Pattern 1

        f-1

        Matches for Pattern 2

        f-2

        Partial matches for CEs 1 - 2

        f-1,f-2

        Activations

        f-1,f-2

        (2 1 1)

        CLIPS> (agenda)

        0      no_escape: f-1,f-2

        For a total of 1 activation.

        CLIPS>

      

      

      Now that f-1 matches the first pattern and f-2 matches the second pattern, the partial match f-1,f-2 for the first two patterns is created and an activation for the partial match is placed on the agenda. The rule can now be run and and the matches command output checked  again:

      
        
        CLIPS> (run)

        The walls are too slick.

        CLIPS> (matches no_escape)

        Matches for Pattern 1

        f-1

        Matches for Pattern 2

        f-2

        Partial matches for CEs 1 - 2

        f-1,f-2

        Activations

        None

        (2 1 0)

        CLIPS> (agenda)

        CLIPS>

      

      

      The matches command shows that a partial match still exists for the first two patterns of the rule, but the activation is no longer displayed because the rule has been executed.

      Removing Facts

      What happens if the player is insistent and really wants the adventurer to scale the walls? Try asserting the same fact a second time:

      
        
        CLIPS> (facts)

        f-1     (thing (id adventurer)

                       (category actor)

                       (location pit_north))

        f-2     (command (action climb up))

        For a total of 2 facts.

        CLIPS> (assert (command (action climb up)))

        <Fact-2>

        CLIPS> (agenda)

        CLIPS>

      

      

      The return value of the assert command is a fact that already exists. CLIPS doesn’t allow an exact duplicate of a fact to be created, so any attempt to do this returns the value of the existing fact. Rules exhibit a property called refraction that prevents them from executing again until a change has been made to a fact matching one of its patterns. One way to create a change is to remove an existing fact and then assert it again.

      Before doing this, the watch debugging command can be used to print changes as they occur so you don’t have to use the facts and agenda commands to look for changes:

      
        
        CLIPS> (watch facts)

        CLIPS> (watch activations)

        CLIPS>

      

      

      The (watch facts) command directs CLIPS to print a message whenever a fact is added or removed, and the (watch activations) command directs CLIPS to print a message whenever an activation is added or removed from the agenda.

      Facts can be removed using the retract command. From the command prompt, you can use the fact-index to specify which fact should be retracted:

      
        
        CLIPS> (retract 2)

        <== f-2     (command (action climb up))

        CLIPS>

      

      

      When fact f-2 was retracted, a message was printed indicating this change. The <==  symbol in a watch message indicates removal, whereas the ==> symbol indicates addition.

      Now that the prior command fact has been removed, you can try adding it again:

      
        
        CLIPS> (assert (command (action climb up)))

        ==> f-3     (command (action climb up))

        ==> Activation 0      no_escape: f-1,f-3

        <Fact-3>

        CLIPS>

      

      

      This time, the fact is successfully asserted and has been assigned a new fact-index. In addition, a new activation for the no_escape rule has been added to the agenda.

      Allowing the rule to execute prints the slick walls message again:

      
        
        CLIPS> (run)

        The walls are too slick.

        CLIPS>

      

      

      Rather than manually retracting facts from the command prompt, the rules that process commands can retract them once a command has been completed. The no_escape rule can be redefined to automatically retract the fact matching the command pattern:

      
        
        CLIPS>

        (defrule no_escape

           (thing (id adventurer)

                  (location pit_north))

           ?c <- (command (action climb up))

           =>

           (retract ?c)

           (println "The walls are too slick."))

        ==> Activation 0      no_escape: f-1,f-3

        CLIPS>

      

      

      The first change to the rule is the addition of

      
        
        ?c <-

      

      

      before the command pattern. Variables in CLIPS begin with a ? (question mark) character and are followed by a symbol beginning with an alphabetic character. Variables are used to store values that can later be retrieved. The <- symbol is an assignment operator that assigns a pattern address (the fact matching a pattern) to a variable. So the effect of this new syntax is to assign the fact-address of the fact matching the command pattern to the variable c.

      The second change to the rule is the addition of the command

      
        
        (retract ?c)

      

      

      to the actions of the rule. When this rule executes, the variable c will be bound to the fact‑address <Fact-3>, allowing the retract command to remove this fact. The retract command can remove facts specified using either a fact-index or a fact-address. In addition, the retract command accepts multiple arguments, so if several variables are bound to pattern addresses in the conditions of a rule, they can all be retracted with a single command.

      When redefining a rule, the old rule is first deleted before the new version of the rule is added. This allows the new rule to be activated by existing facts, which is why the no_escape rule is placed on the agenda again.

      Since the rule is ready to fire, run it to see what happens:

      
        
        CLIPS> (run)

        <== f-3     (command (action climb up))

        The walls are too slick.

        CLIPS>

      

      

      The command fact is now retracted as part of the actions of the rule. Note that there is no activation removal message for the rule when it executes. A removal message is only printed if an activation is removed before it has the opportunity to execute.

      Modifying Facts

      Since the adventurer can’t escape from the north side of the pit, exploration is their best option. To give them somewhere to go, add a new rule that describes the south side of the pit:

      
        
        CLIPS>

        (defrule pit_south

           (thing (id adventurer)

                  (location pit_south))

           =>

           (println "You're at the pit's south end.")

           (println "A large pile of rubble has")

           (println "collapsed from the wall above."))

        CLIPS>

      

      

      To move the adventurer from one side of the pit to the other and trigger this new rule, the value of the adventurer’s location slot needs to change from pit_north to pit_south. This could be done by explicitly retracting the old fact and then asserting a new one, but that would require specifying the slot values that hadn’t changed (id and category) as well as the location value. Fortunately, the modify command (a special form like the assert command) does this task, allowing just the desired changes to be specified:

      
        
        CLIPS> (modify 1 (location pit_south))

        <== f-1     (thing ... (location pit_north))

        ==> f-1     (thing ... (location pit_south))

        ==> Activation 0      pit_south: f-1

        <Fact-1>

        CLIPS>

      

      

      When a fact is modified, any watch message displayed will show both the retraction and assertion of the fact. Only changes to slots are displayed, and unchanged slots are signified using ellipses in the output. The fact-index and fact-address of a fact are not changed when the fact is modified; however, if the modified fact duplicates another existing fact, then the fact being modified is retracted and the fact-address of the other existing fact is returned by the command. If none of the slot values specified in the modify command are different from the existing values, then no action is taken.

      Now that the pit_south rule is activated, the run command will display the description of this new location:

      
        
        CLIPS> (watch rules)

        CLIPS> (run)

        FIRE    1 pit_south: f-1

        You're at the pit's south end.

        A large pile of rubble has

        collapsed from the wall above.

        CLIPS>

      

      

      The (watch rules) command used here causes a message to be printed each time a rule is executed. The message includes the running count of rules executed, the name of the rule being fired, and the facts matching the conditions of the rule.

      CLIPS also provides a duplicate command that behaves similarly to the modify command, but it does not retract the original fact. This allows you to create a new fact from an existing fact without having to specify the slot values that are unchanged.

      The ‘Or’ Field Constraint

      What happens if the adventurer attempts to climb the walls from the south side of the pit? Will the no_escape rule be activated? Create a climb up command to see what happens:

      
        
        CLIPS> (assert (command (action climb up)))

        ==> f-4     (command (action climb up))

        <Fact-4>

        CLIPS> (matches no_escape)

        Matches for Pattern 1

        None

        Matches for Pattern 2

        f-4

        Partial matches for CEs 1 - 2

        None

        Activations

        None

        (1 0 0)

        CLIPS>

      

      

      No activation is generated, and the matches command shows that the second pattern of the rule is matched, but not the first. That’s because the first pattern matches the location for the north side of the pit, not the south:

      
        
        (thing (id adventurer)

               (location pit_north))

      

      

      This rule can be redefined using a new type of field constraint so that it is applicable to both sides of the pit:

      
        
        CLIPS>

        (defrule no_escape

           (thing (id adventurer)

                  (location pit_north | pit_south))

           ?c <- (command (action climb | go up))

           =>

           (retract ?c)

           (println "The walls are too slick."))

        ==> Activation 0      no_escape: f-1,f-4

        CLIPS>

      

      

      The first change to the rule is adding the pit_south literal constraint after the pit_north literal constraint and then placing an or field constraint, specified using the | (vertical bar) character, between the two:

      
        
        (location pit_north | pit_south)

      

      

      The | field constraint is one of three connective field constraints provided by CLIPS. Connective constraints can’t be placed by themselves within a slot pattern; instead they allow one or more field constraints to be connected together as a combined constraint. When two or more constraints are connected using the | field constraint, the combined constraint is matched if any of the connected constraints are matched. In this case, the location slot of the thing fact matching the first pattern must be either pit_north or pit_south.

      The next change to the rule is adding the go literal constraint after the climb literal constraint and placing the | constraint between the two.

      
        
        ?c <- (command (action climb | go up))

      

      

      Now, in order for the second pattern to be matched, the first value of the action slot must be either climb or go, and the second value of that slot must be up.

      Multiple | constraints can be used within a slot pattern. For example, the slot pattern

      
        
        (action climb | go | scramble up | upward)

      

      

      would be matched if the action slot contained any of these values:

      
        
        climb up

        go up

        scramble up

        climb upward

        go upward

        scramble upward

      

      

      With these changes, the adventurer discovers there’s also no escape from the south side of the pit by climbing the walls:

      
        
        CLIPS> (run)

        FIRE    1 no_escape: f-1,f-4

        <== f-4     (command (action climb up))

        The walls are too slick.

        CLIPS>

      

      

      The ‘And’ and ‘Not’ Field Constraints

      Now that there’s a new location for the adventurer to explore, a rule can be added to move the adventurer from the north to the south side of the pit in response to a command:

      
        
        CLIPS>

        (defrule south_from_pit_north

           ?p <- (thing (id adventurer)

                        (location pit_north))

           ?c <- (command (action go south))

           =>

           (retract ?c)

           (modify ?p (location pit_south)))

        CLIPS>

      

      

      In this rule, the variables p and c are bound to the facts matching the first and second patterns. These variables are used in the actions of the rule to remove the command and change the adventurer’s location.

      Similarly, another rule can be added to move the adventurer from the south to the north side of the pit:

      
        
        CLIPS>

        (defrule north_from_pit_south

           ?p <- (thing (id adventurer)

                        (location pit_south))

           ?c <- (command (action go north))

           =>

           (retract ?c)

           (modify ?p (location pit_north)))

        CLIPS>

      

      

      Asserting a go north command causes one of the new rules to be activated:

      
        
        CLIPS> (assert (command (action go north)))

        ==> f-5     (command (action go north))

        ==> Activation 0   north_from_pit_south: f-1,f-5

        <Fact-5>

        CLIPS> (run)

        FIRE    1 north_from_pit_south: f-1,f-5

        <== f-5     (command (action go north))

        <== f-1     (thing ... (location pit_south))

        ==> f-1     (thing ... (location pit_north))

        ==> Activation 0      pit_north: f-1

        FIRE     2 pit_north: f-1

        You're at the pit's north end.

        A giant mushroom is here. The

        ground is littered with the

        bodies of dead adventurers.

        CLIPS>

      

      

      Something new has happened when the run command is processed: one rule immediately executing after another. When the north_from_pit_south rule changes the location of the adventurer, the pit_north rule is activated and then executed next.

      Asserting a go north command again doesn’t activate any rules:

      
        
        CLIPS> (assert (command (action go north)))

        ==> f-6     (command (action go north))

        <Fact-6>

        CLIPS>

      

      

      No rules are activated by the assertion because there are no rules that handle going north (or any direction other than south or up) from the north side of the pit, so this new command is not processed. A new rule can be added to check for this situation:

      
        
        CLIPS>

        (defrule bad_go_from_pit_north

           (thing (id adventurer)

                  (location pit_north))

           ?c <- (command (action go ~south&~up))

           =>

           (retract ?c)

           (println "You can't go there."))

        ==> Activation 0  bad_go_from_pit_north: f-1,f-6

        CLIPS>

      

      

      This rule introduces two new connective field constraints: the not field constraint, specified using the ~ (tilde) character, and the and field constraint, specified using the & (ampersand) character.

      The ~ field constraint can be placed before another field constraint to negate the matching logic. So the constraint ~south means any value other than south and the constraint ~up means any value other than up.

      The & field constraint is placed between two field constraints, similar to the | field constraint, but it requires all the separate constraints to be matched in order for the combined constraint to be matched. So the ~south&~up constraint matches for any value other than south or up.

      Now the player can be informed when the adventurer can’t move in the specified direction from the north side of the pit:

      
        
        CLIPS> (run)

        FIRE    1 bad_go_from_pit_north: f-1,f-6

        <== f-6     (command (action go north))

        You can't go there.

        CLIPS>

      

      

      For completeness, a similar rule can be added for the south side of the pit:

      
        
        CLIPS>

        (defrule bad_go_from_pit_south

           (thing (id adventurer)

                  (location pit_south))

           ?c <- (command (action go ~north&~up))

           =>

           (retract ?c)

           (println "You can't go there."))

        CLIPS>

      

      

      Invalid Constraints

      Not all combinations using connective constraints are useful. For example, the constraint

      
        
        pit_north&pit_south

      

      

      will never match because a value can’t be both pit_north and pit_south. Similarly, the constraint

      
        
        ~pit_north|~pit_south

      

      

      will always match, so it would be better just to use a single-field wildcard in its place.

      For some cases, CLIPS will automatically detect constraints that are unmatchable and generate an error:

      
        
        CLIPS>

        (defrule impossible

           (thing (id adventurer)

                  (location pit_north&pit_south))

           =>)

      

      

      
        
        [RULECSTR1] Pattern #1 slot 'location' has constraint conflicts which make the pattern unmatchable.

      

      

      
        
        ERROR:

        (defrule MAIN::impossible

           (thing (id adventurer)

                  (location pit_north&pit_south))

           =>)

        CLIPS>

      

      

      What’s Next

      You’ve now created and observed the behavior of some simple rules. The source code for these rules is in the file game.clp in the Chapter2 directory. In the next chapter, you’ll continue building the game and learn how to load your code from files rather than defining constructs directly at the command prompt.

      Problems

      
        
        #2-1

      

      

      Modify the exposition rule so that it places the adventurer at the north side of the pit.

      
        
        #2-2

      

      

      Add a rule that prints the following message when the player issues a climb mushroom command from the north side of the pit:

      
        
        The mushroom is not sturdy enough to climb.

      

      

      
        
        #2-3

      

      

      Write four rules. The first initially places a wand in the rubble at the south side of the pit (a thing fact with an id slot value of wand and a location slot value of rubble). The next two handle the search rubble command. One applies when the wand is undiscovered. It changes the location slot value of the wand to  adventurer and prints this message:

      
        
        You find a wand.

      

      

      The other applies once the wand has been discovered and prints this message:

      
        
        You find nothing of interest.

      

      

      The final rule prints the following message when the adventurer has the wand  (its location slot value is adventurer) and the player command is say abracadabra:

      
        
        Poof! You are teleported out of the pit.

      Files and Directories

      When creating CLIPS programs, you’ll want to place your construct definitions in files rather than entering them at the command prompt. You can then edit them using your favorite text editor. CLIPS can process both plain ASCII text files and UTF-8 unicode files.

      File paths passed as function arguments can be specified using either symbols or strings. If a file name or directory path includes spaces, it must be represented as a string. The \ (backslash) character acts as an escape code within strings (allowing you to include a " character within the string), so when specifying a directory path it is best to use a / (forward slash) character to separate directory and file names. Some examples of file paths are:

      
        
        hello.clp

        "hello world.clp"

        /Users/Garak/Desktop/hello.clp

        "C:/Users/Garak/Desktop/hello world.clp"

      

      

      When CLIPS is launched as a terminal executable, the current directory will be set to the directory containing the executable. If just file names are specified when loading a program, those files must be in the same directory as the executable. Otherwise, you will need to specify an absolute file path or a file path relative to the executable’s directory.

      When using one of the IDEs, the current directory is displayed just above the area containing the command prompt. The current directory can be changed by selecting the Set Directory… menu item in the Environment menu. The current directory is saved when you exit the IDE and restored when it is restarted.

      The Load Command

      To organize this program, the constructs created will be divided between four files:

      
        
        definitions.clp

        kernel.clp

        data.clp

        game.clp

      

      

      The first three files will include code that can be used for any text adventure you write. The game.clp file will contain code related to a specific text adventure (such as Escape from the Goblin Pit). From the previous chapter, the code for the thing and command deftemplates will be placed in definitions.clp; the eight defrule constructs in game.clp; and the files kernel.clp and data.clp will remain empty for now. For your convenience, these files are already created in the Chapter3A directory.

      Before loading the constructs from these files, first set CLIPS back to its initial state:

      
        
        CLIPS> (unwatch all)

        CLIPS> (clear)

        CLIPS>

      

      

      The unwatch command allows the effects of the watch command to be reversed. The all keyword can be used to turn off everything that is watchable. The clear command removes all existing constructs and retracts all facts.

      The construct definitions in the files you created can be loaded into CLIPS using the load command:

      
        
        CLIPS> (load definitions.clp)

        %%

        TRUE

        CLIPS> (load data.clp)

      

      

      
        
        TRUE

        CLIPS> (load kernel.clp)

      

      

      
        
        TRUE

        CLIPS> (load game.clp)

        ********

        TRUE

        CLIPS>

      

      

      If you’re using the CLIPS terminal executable, you’ll need to specify the relative or absolute paths of these files in the arguments to the load command. If you’re using one of the IDEs, select the Set Directory… menu item to change the current directory to the directory containing the files before you attempt to load them. You can also use the Load Constructs… menu item to load a file (and also set the current directory to the directory containing that file).

      The load command returns TRUE if it was able to load the file. It also prints a character for each construct that it loads. The file definitions.clp contains two deftemplates, so two % characters are printed; game.clp contains eight defrules, so eight * characters are printed; and since data.clp and kernel.clp contain no constructs, no characters are printed.

      With the constructs loaded, the program can be run:

      
        
        CLIPS> (reset)

        CLIPS> (run)

        Captured by goblins, you've been

        tossed in a pit at their lair.

        CLIPS>

      

      

      The reset command retracts all facts and activates rules that have no conditions. For this program, the exposition rule will be placed back on the agenda so that it executes when the run command is issued. You can use this command whenever you want to restart your program.

      The Batch Command

      Typing all those commands to load the program is still rather tedious, but that process can be automated. To do this, first create a file named load.bat with this content:

      
        
        (unwatch all)

        (clear)

        (load definitions.clp)

        (load kernel.clp)

        (load data.clp)

        (load game.clp)

        (reset)

      

      

      Be sure to enter a carriage return after the (reset) command so that the last line in the file contains no characters.

      The batch command can be used to automatically execute commands and define constructs using the contents of a file:

      
        
        CLIPS> (batch load.bat)

        TRUE

        CLIPS> (unwatch all)

        CLIPS> (clear)

        CLIPS> (load definitions.clp)

        %%

        TRUE

        CLIPS> (load kernel.clp)

      

      

      
        
        TRUE

        CLIPS> (load data.clp)

      

      

      
        
        TRUE

        CLIPS> (load game.clp)

        ********

        TRUE

        CLIPS> (reset)

        CLIPS> (run)

        Captured by goblins, you've been

        tossed in a pit at their lair.

        CLIPS>

      

      

      Now the only commands you have to type to run the program are the batch and run commands. If the last command in the batch file is not followed by a carriage return, then the last command will be entered at the command prompt, but will not be executed until you press the return key.

      Because the load command only allows you to define constructs, you’ll typically use this for loading constructs. When an error is encountered in a construct, it will advance to the next construct found in the file being loaded; the batch command simply moves on to the very next function call or construct in the file. For example, if you misplace a parenthesis in the actions of a rule, the batch command could execute the remaining actions of that rule or generate spurious error messages.

      During development, it’s often convenient to see the output from the commands loading your program, but this output can be eliminated using the load* and batch* commands. To see how this works, create a file named loadq.bat with this content:

      
        
        (unwatch all)

        (clear)

        (load* definitions.clp)

        (load* kernel.clp)

        (load* data.clp)

        (load* game.clp)

        (reset)

      

      

      Now there will be a minimal amount of output when loading the program:

      
        
        CLIPS> (batch* loadq.bat)

        TRUE

        CLIPS> (run)

        Captured by goblins, you've been

        tossed in a pit at their lair.

        CLIPS>

      

      

      Finally, create a file named runq.bat with this content:

      
        
        (batch* loadq.bat)

        (run)

      

      

      You can nest batch and batch* commands within other batch files, so you can create complex scripts of commands from simpler scripts. Now the program can be loaded and run with a single command when the current directory is set to Chapter3A:

      
        
        CLIPS> (batch* runq.bat)

        Captured by goblins, you've been

        tossed in a pit at their lair.

        TRUE

        CLIPS>

      

      

      Note that in this and the prior examples, the load, load*, batch, and batch* commands return the symbol TRUE, indicating that the specified files were successfully processed.

      Comments

      It’s a good idea to comment your code—if not for yourself, for the person who ends up maintaining it. In CLIPS, a comment begins with a ; (semicolon) character, and everything after that is ignored until the end of the line is reached. The only exception is that a semicolon inside a string does not start a comment. Here’s one of the previous rules with comments added:

      
        
        ;;; North From South Side Of Pit Rule

        ;;; Moves the player from the north

        ;;; to the south side of the pit

      

      

      
        
        (defrule north_from_pit_south

      

        

      
           ;; The adventurer is at the

           ;; south side of the pit

      

        

      
           ?p <- (thing (id adventurer)

                        (location pit_south))

      

      

      
        
           ;; The command is go north

      

        

      
           ?c <- (command (action go north))

      

      

      
        
           => ; then

      

      

      
        
           ;; Remove the command

      

        

      
           (retract ?c)

      

      

      
        
           ;; Move the adventurer

           ;; to the north side of the pit

      

        

      
           (modify ?p (location pit_north)))

      

      

      I like to use multiple semicolons to provide some visual distinction between comments outside constructs and comments inside constructs, but it is not necessary to use more than one semicolon. I won’t be showing comments in the rules within this tutorial since I’ll be explaining them in detail, but the downloadable source code for this program will include them.

      The Deffacts Construct

      The  exposition rule is placed on the agenda when the program is reset, but the pit_north rule is not:

      
        
        CLIPS> (reset)

        CLIPS> (facts)

        CLIPS> (agenda)

        0      exposition: *

        For a total of 1 activation.

        CLIPS> (matches pit_north)

        Matches for Pattern 1

         None

        Activations

         None

        (0 0 0)

        CLIPS>

      

      

      The reset command retracts all facts, and that includes the fact representing the adventurer and their location. This can be remedied by adding a deffacts construct:

      
        
        CLIPS>

        (deffacts player

           (thing (id adventurer)

                  (category actor)

                  (location pit_north)))

        CLIPS>

      

      

      A deffacts construct has a name (and optional comment) followed by zero or more fact definitions. You can think of it as a special kind of assert command that gets invoked when a reset command is evaluated.

      Enter a reset command again and see what happens:

      
        
        CLIPS> (watch facts)

        CLIPS> (reset)

        ==> f-1     (thing (id adventurer)

                           (category actor)

                           (location pit_north))

        CLIPS>

      

      

      Now the adventurer will be placed at the north side of the pit every time a reset command is issued.

      Salience

      The pit_north rule is now activated, but it’s been placed on the agenda before the exposition rule, so when the program runs, the output comes out in the wrong order:

      
        
        CLIPS> (agenda)

        0      pit_north: f-1

        0      exposition: *

        For a total of 2 activations.

        CLIPS> (run)

        You're at the pit's north end.

        A giant mushroom is here. The

        ground is littered with the

        bodies of dead adventurers.

        Captured by goblins, you've been

        tossed in a pit at their lair.

        CLIPS>

      

      

      The process used to determine which rule executes next when multiple rules are activated is called the conflict resolution strategy. When two rules have the same salience, CLIPS uses a depth conflict resolution strategy: the rule activated by the most recently asserted fact is placed above other rules on the agenda.

      What’s needed is a mechanism to indicate which rule should execute first when the order of execution matters. In CLIPS, this can be accomplished using the salience rule property. The exposition rule can be modified so that it will execute first:

      
        
        CLIPS>

        (defrule exposition

           (declare (salience 10))

           =>

           (println "Captured by goblins, you've been")

           (println "tossed in a pit at their lair.")

           (println))

        CLIPS>

      

      

      At the beginning of a rule, before you’ve specified any patterns, you can declare properties for the rule. In the exposition rule, the salience of the rule is declared to be 10. Salience indicates the priority with which a rule is applied. By default, the salience for a rule is 0. If more than one rule is on the agenda, the one with the highest salience is allowed to execute first. Salience values range from -10000 to 10000, but generally a well written program will only use a few.

      The exposition rule was also modified to print an additional new line to separate it from the pit_north rule output.

      Issue a reset command and check the agenda again:

      
        
        CLIPS> (watch activations)

        CLIPS> (reset)

        <== Activation 10     exposition: *

        <== f-1     (thing (id adventurer)

                           (category actor)

                           (location pit_north))

        ==> Activation 10     exposition: *

        ==> f-1     (thing (id adventurer)

                           (category actor)

                           (location pit_north))

        ==> Activation 0      pit_north: f-1

        CLIPS> (agenda)

        10     exposition: *

        0      pit_north: f-1

        For a total of 2 activations.

        CLIPS>

      

      

      This time you can see two parts to the reset command. First, existing activations and facts are removed. Next, rules with no conditions are activated and deffacts assertions are performed. Since the deffacts assertions occur last, the activation for the pit_north rule occurs last. With a depth conflict resolution strategy, the pit_north rule should be placed at the top of the agenda; however, because the exposition rule has a higher salience, it remains at the top of the agenda.

      When using one of the CLIPS IDEs, you can select the Agenda Browser menu item from the Debug menu to open a dedicated window for displaying the contents of the agenda. Figure 3-1 shows the agenda browser for the Windows IDE.

      
        
          [image: Agenda browser displaying activations]
        

        Figure 3-1
      
      The left pane in the browser shows the modules in the focus stack (modules and the focus stack will be covered in Chapter 7). Selecting one of the modules in the left pane displays the activations for that module in the right pane. The MAIN module is automatically predefined, so all of the activations have been placed there.

      The agenda browser also has Reset, Run, and Step buttons, which will enter (reset), (run), and (run 1) commands at the command prompt. The Halt Rules button will stop the execution of rules after the currently executing rule has completed.

      The updated program (with changes added to game.clp) can be run from the Chapter3B directory:

      
        
        CLIPS> (batch* runq.bat)

        Captured by goblins, you've been

        tossed in a pit at their lair.

      

        

      
        You're at the pit's north end.

        A giant mushroom is here. The

        ground is littered with the

        bodies of dead adventurers.

        TRUE

        CLIPS>

      

      

      Unanticipated Commands

      Since you can’t anticipate all the commands a player may give the adventurer, a new rule will be added that removes a command when no other rules are able to process it:

      
        
        CLIPS>

        (defrule bad_command

           (declare (salience -10))

           ?c <- (command)

           =>

           (println "I don't understand your command.")

           (retract ?c))

        CLIPS>

      

      

      This rule will match any command fact since there are no constraints on the slot values. When executed it prints an appropriate message before removing the command. It is given a salience of -10 so that it will execute after other rules for processing specific commands have had a chance to execute. To see how this rule interacts with rules that process a valid command, create a go south command:

      
        
        CLIPS> (watch facts)

        CLIPS> (watch activations)

        CLIPS> (assert (command (action go south)))

        ==> f-2     (command (action go south))

        ==> Activation -10    bad_command: f-2

        ==> Activation 0   south_from_pit_north: f-1,f-2

        <Fact-2>

        CLIPS> (agenda)

        0      south_from_pit_north: f-1,f-2

        -10    bad_command: f-2

        For a total of 2 activations.

        CLIPS>

      

      

      Both the bad_command and south_from_pit_north rules are activated by the assertion of the command fact. When two or more rules are activated by the assertion of the same fact, their placement on the agenda is arbitrary (but not random since the placement is unspecified but nonetheless deterministic). In this case, the south_from_pit_north rule has a higher salience than the bad_command rule, so it is placed at the top of the agenda (which is the desired behavior). Because the arbitrary order in which the rules were activated just happens to be the desired order, the same result would had occurred if both rules had identical salience values, but this is not something a well written program should rely on. If two rules have the same salience, the correct operation of your program should not rely on a particular order of execution.

      Allow the rules to execute to see what happens:

      
        
        CLIPS> (watch rules)

        CLIPS> (run)

        FIRE    1 south_from_pit_north: f-1,f-2

        <== f-2     (command (action go south))

        <== Activation -10    bad_command: f-2

        <== f-1     (thing ... (location pit_north))

        ==> f-1     (thing ... (location pit_south))

        ==> Activation 0      pit_south: f-1

        FIRE    2 pit_south: f-1

        You're at the pit's south end.

        A large pile of rubble has

        collapsed from the wall above.

        CLIPS>

      

      

      When the south_from_pit_north rule executes, it retracts the command fact, which removes the bad_command activation from the agenda and so only rules related to a valid command are executed.

      Next, try a command that does not have any specific rules for processing it:

      
        
        CLIPS> (assert (command (action attack goblin)))

        ==> f-3     (command (action attack goblin))

        ==> Activation -10    bad_command: f-3

        <Fact-3>

        CLIPS> (agenda)

        -10    bad_command: f-3

        For a total of 1 activation.

        CLIPS> (run)

        FIRE    1 bad_command: f-3

        I don't understand your command.

        <== f-3     (command (action attack goblin))

        CLIPS>

      

      

      In this case, there are no specific rules for the attack goblin command, so the only rule that’s applicable is the bad_command rule.

      Getting Input from the User

      It would be much more convenient for the player if they could enter a command as

      
        
        go south

      

      

      rather than

      
        
        (assert (command (action go south)))

      

      

      To do this, add a new rule to get a command from the user and convert it to the appropriate format:

      
        
        CLIPS>

        (defrule get_command

           (declare (salience -10))

           (not (command))

           =>

           (println)

           (print "> ")

           (bind ?rsp (explode$ (lowcase (readline))))

           (assert (command (action ?rsp))))

        ==> Activation -10    get_command: *

        CLIPS>

      

      

      There are several things to explain in this rule. First, it has a salience of -10 so that processing of the current command can completely finish before the player is queried for the next command. For example, if the adventurer is moving from one place to another, there might be rules that update the adventurer’s location, but there may also be rules that display what the adventurer sees in this new location, so all of these rules need to execute before the player is queried for the next command.

      This rule also introduces a new conditional element:

      
        
        (not (command))

      

      

      The not conditional element is matched if the conditional element it contains is not matched, so this rule will only be matched if no command facts exist. This will occur once any rule removes the command fact (either a rule that processed the command or the bad_command rule).

      The first two actions of the rule print a prompt for the player to enter a command:

      
        
        (println)

        (print "> ")

      

      

      The println function call will just move to a new line, and the print function will print a > character on that line but not move to the next line, so that the player can enter their input on the same line.

      The next action retrieves a line of input from the player and converts it from a string to a multifield value:

      
        
        (bind ?rsp (explode$ (lowcase (readline))))

      

      

      The bind function is a special form that allows a new variable to be created or a new value to be assigned to an existing variable. In this case, the readline function is called to retrieve a line of input from standard input (which is the keyboard). This function reads up to the point that the return key is pressed and then returns all of the input as a string. The string input is then passed to the lowcase function, which converts its input to lower case characters. This is done so that commands such as

      
        
        go south

        GO SOUTH

        Go South

      

      

      will all be converted to

      
        
        go south

      

      

      so that rules processing this command can be written to match just lower case characters.

      The return value of the lowcase function is then passed to the explode$ function, which converts a string to a multifield value. So the single string value

      
        
        "go south"

      

      

      would be converted to the multifield value

      
        
        (go south)

      

      

      which contains two symbols: go and south.

      The final action of the rule converts the player’s input to a fact:

      
        
        (assert (command (action ?rsp)))

      

      

      Issue agenda and matches commands to see how information is displayed for a rule containing a not conditional element:

      
        
        CLIPS> (agenda)

        -10    get_command: *

        For a total of 1 activation.

        CLIPS> (matches get_command)

        Matches for Pattern 1

         None

        Activations

        *

        (0 0 1)

        CLIPS>

      

      

      Both commands use the * symbol in the activation to show that there is a match, but not to a specific fact. That’s because you can’t explicitly refer to something that doesn’t exist.

      You can step through the behavior of this new rule and see how it interacts with other rules. The run command accepts an optional integer argument that specifies the maximum number of rules to execute. Use this option to execute just one rule:

      
        
        CLIPS> (run 1)

        FIRE    1 get_command: *

      

        

      
        > attack goblin

        ==> f-4     (command (action attack goblin))

        ==> Activation -10    bad_command: f-4

        CLIPS>

      

      

      You can see that the attack goblin text entered has been successfully asserted as a command fact. Next examine the get_command and bad_command rules to see how they’re effected by the assertion of this new fact:

      
        
        CLIPS> (agenda)

        -10    bad_command: f-4

        For a total of 1 activation.

        CLIPS> (matches bad_command)

        Matches for Pattern 1

        f-4

        Activations

        f-4

        (1 0 1)

        CLIPS> (matches get_command)

        Matches for Pattern 1

        f-4

        Activations

         None

        (1 0 0)

        CLIPS>

      

      

      The command fact f-4 matches both rules, but has the opposite effect for each. The bad_command rule is activated because this fact exists, but the get_command rule with a not conditional element around the same pattern is not activated.

      Allow the bad_command rule to remove the command fact to see what happens:

      
        
        CLIPS> (run 1)

        FIRE    1 bad_command: f-4

        I don't understand your command.

        <== f-4     (command (action attack goblin))

        ==> Activation -10    get_command: *

        CLIPS> (agenda)

        -10    get_command: *

        For a total of 1 activation.

        CLIPS>

      

      

      Once the command fact is removed, the get_command rule is activated again. You can now allow the program to run and the player will be able to enter one command after another. That, however, presents a different issue that will need to be resolved. If you issue a (run) command rather than just stepping through execution, you’ll end up in an endless loop with no explicit way to get out of it and stop the game.

      With most operating systems, if you’re running CLIPS as a terminal application, you can press the control and c keys to stop the execution of rules. If you’re using one of the CLIPS IDEs, you can select the Halt Rules menu item from the Environment menu to halt rule execution. You can also use the Halt Rules button in the agenda browser to halt rule execution. These methods are primarily useful if you inadvertently get stuck in a loop while developing your code.

      A new rule will be added that explicitly allows the player to stop the game using a quit command:

      
        
        CLIPS>

        (defrule quit

           ?c <- (command (action quit))

           =>

           (retract ?c)

           (halt))

        CLIPS>

      

      

      The halt command halts execution of rules after the actions of the currently executing rule have finished. Once this command is executed in the actions of this rule, control will return to the REPL. You can now get out of the program’s command loop using the quit command:

      
        
        CLIPS> (run)

        FIRE    1 get_command: *

      

      

      
        
        > quit

        ==> f-5    (command (action quit))

        ==> Activation 0      quit: f-5

        ==> Activation -10    bad_command: f-5

        FIRE    2 quit: f-10

        <== f-10    (command (action quit))

        <== Activation -10    bad_command: f-5

        ==> Activation -10    get_command: *

        [PRCCODE4] Execution halted during the actions of defrule quit.

        CLIPS>

      

      

      Note that once execution has halted, activations still remain on the agenda, so you can resume execution with a run command:

      
        
        CLIPS> (agenda)

        -10    get_command: *

        For a total of 1 activation.

        CLIPS>

      

      

      The updated program (with changes added to kernel.clp) can be run from the Chapter3C directory):

      
        
        CLIPS> (batch* runq.bat)

        Captured by goblins, you've been

        tossed in a pit at their lair.

      

      

      
        
        You're at the pit's north end.

        A giant mushroom is here. The

        ground is littered with the

        bodies of dead adventurers.

      

      

      
        
        > go south

        You're at the pit's south end.

        A large pile of rubble has

        collapsed from the wall above.

      

      

      
        
        > climb up

        The walls are too slick.

      

      

      
        
        > quit

        TRUE

        CLIPS>

      

      

      Looking Around

      You can’t get anything done in a text adventure without looking at the things the adventurer  encounters, so new rules will be added to handle a look at command:

      
        
        CLIPS>

        (defrule look_mushroom

           ?c <- (command (action look at mushroom))

           (thing (id adventurer)

                  (location pit_north))

           =>

           (retract ?c)

           (println "It's squished. I wouldn't")

           (println "try landing on it again."))

        CLIPS>

        (defrule look_bodies

           ?c <- (command (action look at bodies))

           (thing (id adventurer)

                  (location pit_north))

           =>

           (retract ?c)

           (println "Apparently this is what happens")

           (println "when you miss the mushroom."))

        CLIPS>

      

      

      Enter look at mushroom and look at bodies commands to see what happens:

      
        
        CLIPS> (run)

      

      

      
        
        > go north

        You're at the pit's north end.

        A giant mushroom is here. The

        ground is littered with the

        bodies of dead adventurers.

      

      

      
        
        > look at mushroom

        It's squished. I wouldn't

        try landing on it again.

      

      

      
        
        > look at bodies

        Apparently this is what happens

        when you miss the mushroom.

      

      

      
        
        > quit

        CLIPS>

      

      

      Now, see what happens when there isn’t a specific rule to handle a look at command:

      
        
        CLIPS> (run)

      

      

      
        
        > look at walls

        I don't understand your command.

      

      

      
        
        > quit

        CLIPS>

      

      

      That’s not the best response to give the player. The program understands the look at command for some cases, there just isn’t a specific rule that gives a response for look at walls. You can add a new rule to handle this case:

      
        
        CLIPS>

        (defrule look_default

           (declare (salience -5))

           ?c <- (command (action look at ?))

           =>

           (retract ?c)

           (println "You see nothing special."))

        CLIPS>

      

      

      This rule has a salience of -5 so that it will be at a lower salience than the specific look at command rules, but higher than the get_command and bad_command rules. A single field wildcard has been used to allow any value to match the third word in the command.

      Now when the player looks at the wall, there’s a better response:

      
        
        CLIPS> (run)

      

        

      
        > look at walls

        You see nothing special.

      

        

      
        > quit

        CLIPS>

      

      

      What’s Next

      Congratulations! You’ve now got a small program that allows you to issue commands, move around, and examine objects in the adventure. The updated program (with the look_mushroom and look_bodies rule changes added to game.clp and the look_default rule added to kernel.clp) can be run from the Chapter3D directory.

      In the next chapter, you’ll learn how to generalize rules.

      Problems

      
        
        #3-1

      

      

      Write a rule that prints a witty message when the player presses the return key without entering a command.

      
        
        #3-2

      

      

      Create a deftemplate for facts that will keep track of how many times an action has been attempted. Next, create several rules to handle an eat mushroom command when the adventurer is at the north side of the pit. The first time the command is attempted, print this message:

      
        
        You probably shouldn't do that.

        It might be poisonous.

      

      

      The second time the command is attempted, print this message:

      
        
        Seriously? You have no idea what could happen!

      

      

      The third time the command is attempted, halt the game and print this message:

      
        
        OK, you pull off a small piece and eat it. Your head spins and then your body feels light. With a mighty spring, you leap out of the pit.

      

      

      
        
        #3-3

      

      

      Create a default rule for an eat command that will apply if no other rule is applicable and display this message:

      
        
        Think about escape, not lunch.

      Generalizing the Location Rules

      In this chapter you’ll learn how to generalize rules that share similar patterns. The first two rules that will be generalized are the pit_north and pit_south rules from game.clp:

      
        
        (defrule pit_north

           (thing (id adventurer)

                  (location pit_north))

           =>

           (println "You're at the pit's north end.")

           (println "A giant mushroom is here. The")

           (println "ground is littered with the")

           (println "bodies of dead adventurers."))

      

      

      
        
        (defrule pit_south

           (thing (id adventurer)

                  (location pit_south))

           =>

           (println "You're at the pit's south end.")

           (println "A large pile of rubble has")

           (println "collapsed from the wall above."))

      

      

      There are two similarities between these rules. First, the conditions are the same except for the value of the location slot. Second, the actions in both rules print descriptions of the location. So these rules can be generalized by having the conditions match the location of the adventurer and then have the actions print the description of that specific location.

      The first set of changes will be made to the thing deftemplate in definitions.clp:

      
        
        (deftemplate thing

           (slot id)

           (slot category

              (allowed-values place item actor scenery))

           (slot location

              (default nowhere))

           (multislot description))

      

      

      Three changes have been made: the allowed-values keyword is used in the category slot; the default keyword is used in the location slot; and the description slot has been added.

      The allowed-values keyword used by the category slot restricts its value to the symbols place, item, actor, and scenery. The symbol actor will be used when representing something that can be given actions (like the adventurer); place will be used for locations an actor can move between (such as the north and south sides of the pit); scenery will be used for something that the player can see and examine at a location, but otherwise serves no purpose; and item will be used for something that can be manipulated or used by the player.

      The first value listed in the allowed-values attribute also serves as the default value for the slot (if none is specified when the fact is created) unless a different value is specified using the default attribute. If you can enumerate the values for a slot, it’s useful to use this attribute since you’ll get an error message when loading your code if an incorrect value is used.

      The default keyword used by the location slot automatically assigns nowhere to the slot if no value is provided when a thing fact is asserted. This value will indicate that a thing is not contained within something else (such as the north and south sides of the pit) or that it hasn’t been brought it into play yet.

      The multislot keyword has been used to define the description slot so that it can contain multiple strings. This will make it easier for the generalized rule to display output on multiple lines as the rules being replaced did.

      With this new deftemplate, the pit_north and pit_south rules in game.clp can be replaced with a deffacts construct that represents the two sides of the pit as facts:

      
        
        (deffacts places

           (thing (id pit_north)

                  (category place)

                  (description

                    "You're at the pit's north end."

                    "A giant mushroom is here. The"

                    "ground is littered with the"

                    "bodies of dead adventurers."))

      

      

      
        
           (thing (id pit_south)

                  (category place)

                  (description

                    "You're at the pit's south end."

                    "A large pile of rubble has"

                    "collapsed from the wall above.")))

      

      

      A new generic rule will then be added to kernel.clp for printing the description of a place:

      
        
        (defrule print_place

           ?c <- (command (action look))

           (thing (id adventurer)

                  (location ?location))

           (thing (id ?location)

                  (category place)

                  (description $?text))

           =>

           (retract ?c)

           (foreach ?line ?text

             (println ?line)))

      

      

      Rather than just matching the facts for the adventurer and their location, the first pattern of this rule will also require a command fact with the action look:

      
        
        ?c <- (command (action look))

      

      

      This pattern will only match when the action slot contains just one value: look. It would not match if the action slot contained the values look at walls. When matching a multifield slot, both the specific values and number of values in the fact must match the specific values and number of values in the pattern. The fact-address of this command is also bound to the variable c so that it can be removed by the actions of the rule to indicate that it has been processed.

      Since this rule needs a look command, one will need to be asserted whenever the adventurer moves so that the description of the new location will be printed. This rule will also allow the player to directly issue a look command to see a description of the adventurer’s current location.

      In the second pattern, the adventurer’s current location is assigned to a variable:

      
        
        (thing (id adventurer)

               (location ?location))

      

      

      Within patterns, variables that begin with a ? character are called single field variables. These are similar to single field wildcards in that they match a single value, but that value is then bound to the variable and can later be referenced either in the conditions or the actions of that rule. So this pattern will match a thing fact where the id slot value is adventurer and bind the value of the location slot to the variable location.

      The third pattern retrieves the description of the adventurer’s location:

      
        
        (thing (id ?location)

               (category place)

               (description $?text))

      

      

      This pattern will match any thing fact where the category slot value is place and bind the value of the id slot to the variable location. As part of the matching process, using the same variable multiple times in the conditions of a rule restricts the partial matches and activations to those where the variable is bound to the same value in each place it occurs. So if the variable location is bound to pit_north in the second pattern, the third pattern will only be matched if the same variable used in the id slot is bound to the same value.

      Within patterns, variables that begin with the $? characters are called multifield variables. When a slot pattern contains a single multifield variable, that variable will be bound to a multifield value containing all of the slot’s values. In this pattern, the multifield variable text contained in the description slot pattern will be bound to the strings containing the description of a location. For the north side of the pit this would be four strings, and for the south side of the pit this would be three strings.

      The actions of the rule remove the command (since it has been processed) and then print the description of the adventurer’s location:

      
        
        (retract ?c)

        (foreach ?line ?text

           (println ?line))

      

      

      The foreach function is a special form that iterates over the individual values within a multifield value. In this case each value in the variable text will be assigned to the variable line, which will then be printed by the println function. Note that within a function call, it’s not necessary to include the $ character in front of the variable. Within a pattern, when values of a slot are being bound to a variable, you must consistently use either a single field or multifield variable, but in other cases you can just use a single field variable.

      The final piece of code needed is adding a new deffacts to data.clp that contains an initial look command to print the adventurer’s location when the game starts:

      
        
        (deffacts initial_state

           (command (action look)))

      

      

      The updated program can now be loaded from the Chapter4A directory:

      
        
        CLIPS> (batch* loadq.bat)

        TRUE

        CLIPS> (facts)

        f-1 (command (action look))

        f-2 (thing (id adventurer)

                   (category actor)

                   (location pit_north)

                   (description))

        f-3 (thing (id pit_north)

                   (category place)

                   (location nowhere)

                   (description

                      "You're at the pit's north end."

                      "A giant mushroom is here. The"

                      "ground is littered with the"

                      "bodies of dead adventurers."))

        f-4 (thing (id pit_south)

                   (category place)

                   (location nowhere)

                   (description

                      "You're at the pit's south end."

                      "A large pile of rubble has"

                      "collapsed from the wall above."))

        For a total of 4 facts.

        CLIPS> (agenda)

        10     exposition: *

        0      print_place: f-1,f-2,f-3

        -10    bad_command: f-1

        For a total of 3 activations.

        CLIPS> (matches print_place)

        Matches for Pattern 1

        f-1

        Matches for Pattern 2

        f-2

        Matches for Pattern 3

        f-3

        f-4

        Partial matches for CEs 1 - 2

        f-1,f-2

        Partial matches for CEs 1 - 3

        f-1,f-2,f-3

        Activations

        f-1,f-2,f-3

        (4 2 1)

        CLIPS>

      

      

      For the print_place rule, there is one match for the first pattern: f-1, the command fact for the look action; there is one fact matching the second pattern: f-2, the thing fact representing the location of the adventurer; and there are two matches for the third pattern: f-3, the thing fact representing the north side of the pit, and f-4, the thing fact representing the south side of the pit. There is only one partial match for conditional elements one through three: f-1,f-2,f-3. The partial match f1,f-2,f-4 does not match the patterns for the rule because the location slot for f-2 is pit_north and the id slot for f-4 is pit_south. These values must match because the same variable location is used to match both slots.

      You can run the program to verify that the print_place rule prints the description of the adventurer’s starting location:

      
        
        CLIPS> (run)

        Captured by goblins, you've been

        tossed in a pit at their lair.

      

      

      
        
        You're at the pit's north end.

        A giant mushroom is here. The

        ground is littered with the

        bodies of dead adventurers.

      

      

      
        
        > quit

        CLIPS>

      

      

      There’s one issue that still needs to be addressed. When the adventurer moves from one location to another, the description of the new location is not printed:

      
        
        CLIPS> (run)

      

        

      
        > go south

      

        

      
        > quit

        CLIPS>

      

      

      This issue will be addressed later in this chapter when the rules for moving are generalized. A look command will be asserted whenever the adventurer moves to a new location.

      Generalizing the ‘Look At’ Rules

      The look at command rules (from game.clp and kernel.clp) will be generalized next:

      
        
        (defrule look_mushroom

           ?c <- (command (action look at mushroom))

           (thing (id adventurer)

                  (location pit_north))

           =>

           (retract ?c)

           (println "It's squished. I wouldn't")

           (println "try landing on it again."))

      

      

      
        
        (defrule look_bodies

           ?c <- (command (action look at bodies))

           (thing (id adventurer)

                  (location pit_north))

           =>

           (retract ?c)

           (println "Apparently this is what happens")

           (println "when you miss the mushroom."))

      

      

      
        
        (defrule look_default

           (declare (salience -5))

           ?c <- (command (action look at ?))

           =>

           (retract ?c)

           (println "You see nothing special."))

      

      

      To generalize these rules, the look_mushroom and look_bodies rules in game.clp will be replaced with a deffacts containing a thing fact for each of the items (mushroom, bodies, and rubble) mentioned in the descriptions of the north and south sides of the pit:

      
        
        (deffacts scenery

           (thing (id mushroom)

                  (location pit_north)

                  (category scenery)

                  (description

                     "It's squished. I wouldn't"

                     "try landing on it again."))

      

      

      
        
           (thing (id bodies)

                  (location pit_north)

                  (category scenery)

                  (description

                     "Apparently this is what happens"

                     "when you miss the mushroom."))

      

      

      
        
           (thing (id rubble)

                  (location pit_south)

                  (category scenery)))

      

      

      Each of these facts has been assigned a category value of scenery since the adventurer can look at them, but otherwise can’t take or interact with them.

      Now that the descriptions are stored in facts rather than the actions of the rules, some general rules for displaying them can be added. There are three cases that need to be handled for any item specified in a look at command. First, if an item has an associated thing fact, but the description slot is empty (such as for a look at rubble command), the default message indicating that there is nothing special about the item should be printed. Second, if an item has an associated thing fact and the description slot is not empty (such as for a look at mushroom command), then the values contained in that slot should be printed. Finally, if an item does’t have an associated thing fact (such as for a look at unicorn command), then a default message indicating that the item can’t be seen should be printed.

      
        
        Items Without Descriptions

      

      

      For items without descriptions, the look_default rule in kernel.clp will be replaced with a new rule:

      
        
        (defrule look_no_description

           ?c <- (command (action look at ?id))

           (thing (id adventurer)

                  (location ?location))

           (thing (id ?id)

                  (location ?location | adventurer)

                  (description))

           =>

           (retract ?c)

           (println "You see nothing special."))

      

      

      Unlike the look_default rule, which had a salience of -5, this rule has the default salience of 0. The three cases that must be handled are mutually exclusive and cover all general situations, so  there will be no conflicts between these rules that require salience to resolve.

      The first pattern of this rule binds the target item of the look at command to the variable id and the fact-address of the command fact to the variable c:

      
        
        ?c <- (command (action look at ?id))

      

      

      The second pattern assigns the location of the adventurer to the variable location:

      
        
        (thing (id adventurer)

               (location ?location))

      

      

      The third pattern is matched if the adventurer can see an item:

      
        
        (thing (id ?id)

               (location ?location | adventurer)

               (description))

      

      

      In this pattern, the value of the item’s location slot must either be the same as the variable location (which was previously bound to the adventurer’s current location) or it must be the literal adventurer (which means the adventurer is holding it). The description slot is also specified in the pattern, but contains no field constraints, so the pattern will only be matched if the description slot of the fact contains no values.

      If the prior patterns are matched, the actions of the rule retract the command fact since it has been processed and print the generic message for an object with no description:

      
        
        (retract ?c)

        (println "You see nothing special.")

      

      

      
        
        Items With Descriptions

      

      

      Items with descriptions will be handled with a new rule added to kernel.clp:

      
        
        (defrule look_description

           ?c <- (command (action look at ?id))

           (thing (id adventurer)

                  (location ?location))

           (thing (id ?id)

                  (location ?location | adventurer)

                  (description $?text))

           (test (> (length$ ?text) 0))

           =>

           (retract ?c)

           (foreach ?line ?text

              (println ?line)))

      

      

      The first two patterns of this rule are identical to the prior rule.

      The third pattern is also identical except for the addition of the multifield variable text in the description slot:

      
        
        (thing (id ?id)

               (location ?location | adventurer)

               (description $?text))

      

      

      This additional field constraint will match a description slot containing any number of values (including none at all).

      The fourth and final pattern introduces a new type of conditional element that will exclude the case where the description slot is empty (since this will be handled by the prior rule for items with no descriptions):

      
        
        (test (> (length$ ?text) 0))

      

      

      The test conditional element allows an expression to be evaluated. It is matched if the expression evaluates to any value other than FALSE. The length$ function returns the number of values contained in a multifield value, so in this pattern, passing it the multifield variable text will return the number of values stored in the description slot. Since the prior rule already handles the case where the description slot is empty, this rule shouldn’t be activated for the same case, so the > (greater than) function is used to compare the length$ return value to 0. If the return value is greater than 0, then the > function will return TRUE and this pattern will be satisfied; otherwise the pattern will not be satisfied.

      If the prior patterns are matched, the actions of the rule retract the command fact and use the foreach function to print the description of the object:

      
        
        (retract ?c)

        (foreach ?line ?text

           (println ?line))

      

      

      
        
        Predicate Functions

      

      

      Before proceeding to the final case in generalizing the look at action, let’s take a brief aside to discuss the types of functions that you’ll use with the test conditional element. Functions that return a non-FALSE or FALSE value to indicate whether a group of arguments satisfy a relationship are referred to as predicate functions. The topmost function passed to a test conditional should always be a predicate function since a function that never returns FALSE will always satisfy a test conditional element.

      Predicate functions always return the same value when passed the same arguments. For example, the open function is used to open a file and returns either the value TRUE or FALSE depending upon whether the file is successfully opened, but it is not a predicate function because it can return a different value when called multiple times with the same arguments (such as when attempting to open the same file more than once). Because rules can share patterns and it can be difficult to predict the order is which patterns will be evaluated when asserting facts, calling a non-predicate function within a test conditional element can produce unexpected results.

      The > function in the prior rule is one of several predicate functions shown in Table 4-1 that can be used for comparing numeric values (either integers or floats). These functions return TRUE if the relationship tested is true; otherwise they return FALSE.

      When comparing mixed arguments (integers and floats), all arguments are converted to floats for comparison. For example, the following function calls all return TRUE:

      
        
        CLIPS> (= 3 3.0)

        TRUE

        CLIPS> (= 3 3)

        TRUE

        CLIPS> (= 3.0 3.0)

        TRUE

        CLIPS>

      

      

      The functions eq (equal) and neq (not equal) can be used to compare values of any type. Both the values and types of the arguments are considered. For example, these function calls would both return FALSE:

      
        
        CLIPS> (eq 3 3.0)

        FALSE

        CLIPS> (eq red "red")

        FALSE

        CLIPS>

      

      

      
        
          [image: =, <>, <, >, <=, and >= predicate functions.]
        

        Table 4-1
      
      Table 4-2 shows a list of functions that can be used to determine if a value is of a specific type. These functions take a single argument and return TRUE if the argument is of the specified type; otherwise FALSE is returned. For example:

      
        
        CLIPS> (integerp 3)

        TRUE

        CLIPS> (floatp 3)

        FALSE

        CLIPS> (numberp 3)

        TRUE

        CLIPS>

      

      

      
        
          [image: Symbolp, stringp, lexemep, integers, floatp, numberp, multifieldp, and fact-addressp predicate functions.]
        

        Table 4-2
      
      CLIPS also provides three functions for performing boolean operations: the and function returns TRUE if none of its arguments are FALSE; the or function returns TRUE if at least one argument is not FALSE; and the not function returns TRUE if its single argument is FALSE. Otherwise, these functions return FALSE. For example, if the thing fact had a weight slot, the following patterns using the and function would be matched if that slot value was a number between 100 and 150:

      
        
        (thing (weight ?weight))

        (test (and (numberp ?weight)

                   (>= ?weight 100)

                   (<= ?weight 150)))

      

      

      The predicate field constraint (indicated by the : character) can also be used to evaluate an expression. It is satisfied if the expression evaluates to any value other than FALSE. It’s similar to the test conditional element, but is placed inside of a pattern. For example, these patterns from the look_description rule:

      
        
        (thing (id ?id)

               (location ?location | adventurer)

               (description $?text))

        (test (> (length$ ?text) 0))

      

      

      could be replaced with this pattern:

      
        
        (thing (id ?id)

               (location ?location | adventurer)

               (description

                  $?text&:(> (length$ ?text) 0)))

      

      

      You can also tie several : constraints together using the & connective constraint. For example, the prior patterns testing the value of a hypothetical weight slot could be replaced with this pattern:

      
        
        (thing (weight ?weight&:(numberp ?weight)

                              &:(>= ?weight 100)

                              &:(<= ?weight 150)))

      

      

      
        
        Items that Can’t be Seen

      

      

      The final case for the look at command is when the adventurer attempts to look at an item that doesn’t have a corresponding thing fact (like a unicorn) or is somewhere the adventurer can’t see. A new rule will be added to kernel.clp to handle this case:

      
        
        (defrule look_can't_see

           ?c <- (command (action look at ?id))

           (thing (id adventurer) (location ?location))

           (not (thing (id ?id)

                       (location

                          ?location | adventurer)))

           =>

           (retract ?c)

           (println "You can't see any such thing."))

      

      

      Again, the first two patterns of this rule are identical to the two prior look at rules.

      The third pattern uses a not conditional element so that this rule will only be satisfied if the target of the look at command doesn’t exist (the mythical unicorn), isn’t held by the adventurer, or isn’t at the same location:

      
        
        (not (thing (id ?id)

                    (location ?location | adventurer)))

      

      

      If the prior patterns have been matched, the actions of the rule retract the command fact and print a generic message for an item that the adventurer can’t see:

      
        
        (retract ?c)

        (println "You can't see any such thing.")

      

      

      Decomposing Patterns

      The number of options for composing patterns can be daunting at first. At this point, you’ve been introduced to no less than three different types of negation that can be used in patterns: the ~ field constraint, the not conditional element, and the not function. Complex nested boolean expressions in any language can be difficult to understand, and in CLIPS you have an added complexity in that matching logic can apply to a specific slot or to the entire pattern. If you have problems understanding how a pattern will be matched, it can be helpful to decompose the pattern into smaller parts for analysis. For example, in the third pattern of the previous rule, the inner component of the not conditional element is

      
        
        (thing (id ?id)

               (location ?location | adventurer))

      

      

      The id slot value must match the prior value bound to the variable id, otherwise the pattern will not be matched. The location slot value must be the same value as either the variable location or the literal adventurer. This can be expressed as

      
        
        The id slot value is the variable id and

          (the location slot value is the variable location or

           the location slot value is the literal adventurer)

      

      

      Adding the negation of the not conditional element gives

      
        
        It is not true for any thing fact that

          the id slot value is the variable id and

            (the location slot value is the variable location or

             the location slot value is the literal adventurer)

      

      

      Converting the pattern to this more verbose form can make the logic easier to understand.

      Setting Breakpoints

      CLIPS allows you to set breakpoints on rules that automatically halt execution before the corresponding rule is fired. You can load the updated program from the Chapter4B directory to observe the behavior of the new look at command rules using breakpoints:

      
        
        CLIPS> (batch* loadq.bat)

        TRUE

        CLIPS> (set-break look_no_description)

        CLIPS> (set-break look_description)

        CLIPS> (set-break look_can't_see)

        CLIPS> (show-breaks)

        look_description

        look_no_description

        look_can't_see

        CLIPS>

      

      

      The set-break command sets a breakpoint on a rule. Execution will halt before executing a rule with a breakpoint, allowing you to examine the state of the rule before it executes. The list of existing breakpoints is displayed using the show-breaks command.

      Now when you issue a run command, execution will halt before the rules with breakpoints are fired:

      
        
        CLIPS> (run)

        Captured by goblins, you've been

        tossed in a pit at their lair.

      

      

      
        
        You're at the pit's north end.

        A giant mushroom is here. The

        ground is littered with the

        bodies of dead adventurers.

      

      

      
        
        > look at mushroom

        Breaking on rule look_description.

        CLIPS>

      

      

      In this case, the mushroom is the target of the look at command, it’s at the same location as the adventurer, and it has a description, so execution breaks  on the look_description rule. You can see how the rule was satisfied using the matches and ppfact commands:

      
        
        CLIPS> (matches look_description)

        Matches for Pattern 1

        f-8

        Matches for Pattern 2

        f-2

        Matches for Pattern 3

        f-2

        f-3

        f-4

        f-5

        f-6

        f-7

        Partial matches for CEs 1 - 2

        f-8,f-2

        Partial matches for CEs 1 - 3

        f-8,f-2,f-5

        Activations

        f-8,f-2,f-5

        (8 2 1)

        CLIPS> (ppfact 5)

        (thing

           (id mushroom)

           (category scenery)

           (location pit_north)

           (description

              "It's squished. I wouldn't"

              "try landing on it again."))

        CLIPS>

      

      

      From the output it can be seen that the look_description rule is activated by the presence of a thing fact for the mushroom, f-5, matching the third pattern.

      Allow the description of the mushroom to be printed and then see what happens when the adventurer looks at something that doesn’t have a thing fact representing it:

      
        
        CLIPS> (run)

        It's squished. I wouldn't

        try landing on it again.

      

      

      
        
        > look at unicorn

        Breaking on rule look_can't_see.

        CLIPS> (matches look_can't_see)

        Matches for Pattern 1

        f-9

        Matches for Pattern 2

        f-2

        Matches for Pattern 3

        f-2

        f-3

        f-4

        f-5

        f-6

        f-7

        Partial matches for CEs 1 - 2

        f-9,f-2

        Partial matches for CEs 1 - 3

        f-9,f-2,*

        Activations

        f-9,f-2,*

        (8 2 1)

        CLIPS>

      

      

      In this case, there are several facts matching the third pattern, but there isn’t a thing fact with an id slot value of unicorn, so the look_can't_see rule is activated by the absence of that fact. As is done for a rule with no conditions, the absence of a fact is shown in an activation using the * character.

      Breakpoints can be removed using the remove-break command. You can remove a breakpoint for a specific rule, or if you provide no arguments to remove-break, all breakpoints will be removed:

      
        
        CLIPS> (show-breaks)

        look_description

        look_no_description

        look_can't_see

        CLIPS> (remove-break look_description)

        CLIPS> (show-breaks)

        look_no_description

        look_can't_see

        CLIPS> (remove-break)

        CLIPS> (show-breaks)

        CLIPS>

      

      

      In order to examine a rule that isn’t firing, you need to set a breakpoint on the rule that executes before the one you’re interested in examining. In this case, if the look_description rule wasn’t firing and you think it should, you could set a breakpoint on the get_command rule, issue a (run 1) command, and then issue a matches command for the look_description rule.

      Generalizing The Path Rules

      The next rules to be generalized are the go command rules from game.clp:

      
        
        (defrule south_from_pit_north

           ?p <- (thing (id adventurer)

                        (location pit_north))

           ?c <- (command (action go south))

           =>

           (retract ?c)

           (modify ?p (location pit_south)))

      

      

      
        
        (defrule north_from_pit_south

           ?p <- (thing (id adventurer)

                        (location pit_south))

           ?c <- (command (action go north))

           =>

           (retract ?c)

           (modify ?p (location pit_north)))

      

      

      Generalization will start by adding a path deftemplate to definitions.clp that will be used to represent connections between locations:

      
        
        (deftemplate path

           (multislot direction)

           (multislot from)

           (slot to (default nowhere))

           (slot blocked (default FALSE))

           (slot blocked_message

              (default "The way is blocked.")))

      

      

      The direction slot will contain the direction of the go command (such as north or south). The from slot will contain the location at which the path starts (such as pit_north or pit_south). The multislot keyword has been used for these slots so that they can contain multiple values. This will allow multiple paths to be combined when convenient. For example, if the adventurer starts in a clearing in a forest, going either east or west could lead them to the castle. Similarly, going north from either the castle or swamp could lead them to the cave.

      The to slot will contain the location at which the path ends. This slot will contain just a single value since you can’t have the adventurer moving to two different places at the same time. The default for the slot is nowhere, which will be used for paths that are permanently blocked.

      The blocked slot will contain a boolean value indicating whether the adventurer can proceed along a path, and the blocked_message slot will contain the message to print if a move is attempted along the blocked path.

      The south_from_pit_north and north_from_pit_south rules in game.clp can be replaced with a deffacts using this new deftemplate:

      
        
        (deffacts paths

           (path (direction south)

                 (from pit_north)

                 (to pit_south))

      

      

      
        
           (path (direction north)

                 (from pit_south)

                 (to pit_north))

      

      

      
        
           (path (direction up)

                 (from pit_north pit_south)

                 (blocked TRUE)

                 (blocked_message

                    "The walls are too slick.")))

      

      

      This deffacts also contains a path that will be used to generalize the rule used when the player issues a go up command.

      
        
        Moving Along a Valid Path that is Not Blocked

      

      

      A new generic rule will now be added to kernel.clp for processing a go command when the adventurer is directed to move along a valid path:

      
        
        (defrule go_valid_path

           ?c <- (command (action go ?direction))

           ?p <- (thing (id adventurer)

                        (location ?location))

           (path (direction $? ?direction $?)

                 (from $? ?location $?)

                 (to ?new_location)

                 (blocked FALSE))

           =>

           (retract ?c)

           (modify ?p (location ?new_location))

           (assert (command (action look))))

      

      

      The first pattern binds the direction in which the adventurer is attempting to move to the variable direction:

      
        
        ?c <- (command (action go ?direction))

      

      

      The second pattern binds the current location of the adventurer to the variable location:

      
        
        ?p <- (thing (id adventurer)

                     (location ?location))

      

      

      The third pattern checks that the matching path is not blocked; binds the end of the path to the variable new_location; and uses a new type of field constraint, the multifield wildcard constraint, specified using the $? (dollar question mark) characters, in the direction and from slot patterns for the path fact:

      
        
        (path (direction $? ?direction $?)

              (from $? ?location $?)

              (to ?new_location)

              (blocked FALSE))

      

      

      A multifield wildcard is similar to a multifield variable in that it can match zero or more values in a slot, but the values matched can’t be referenced later since they are not bound to a variable. When there are multiple multifield wildcards and/or variables contained in a slot pattern, CLIPS will generate all possible combinations. When matching, all values from the slot must be matched by a constraint in the pattern. For the direction and from slot patterns, this means that these slots must contain at least one value, otherwise there would be nothing for the variables direction and location to match.

      When the direction slot contains the value up, Table 4-3 shows the only possible match for the direction slot pattern. The variable from must be bound to the value up, and the multifield wildcards each match no values.

      
        
          [image: $? = () ?from = up $? = ()]
        

        Table 4-3
      
      Table 4-4 shows all the possible matches for the constraints in the from pattern in the case where the from slot contains the values

      
        
        pit_north pit_south

      

      

      The variable location must either be bound to the literal pit_north or pit_south, and either the preceding or following multifield wildcard will match the remaining value in the slot.

      
        
          [image: $? = () ?location = pit_north $? = (pit_south) or $? = (pit_north) ?location = pit_south $? = ()]
        

        Table 4-4
      
      Essentially, the patterns for the direction and from slots are performing membership tests. First, the value of the variable direction from the command fact must be in the direction slot of the path fact. Second, the value of the variable location from the thing fact representing the adventurer must be found in the from slot of the path fact.

      To see how the actions of this rule will behave, suppose that the adventurer is at the north side of the pit and receives a go south command. In this case, the first two patterns would bind the variable direction to south and the variable location to pit_north. Those bindings would only permit one fact from the paths deffacts  to match the third pattern:

      
        
        (path (direction south)

              (from pit_north)

              (to pit_south))

      

      

      Since the blocked slot is unspecified, it’s assigned the default value of FALSE, which allows it to match the slot constraint in the third pattern. The pit_south value from the to slot will be assigned to the variable new_location for use in the actions of the rule.

      The actions of the rule retract the command fact; update the player’s location to the value stored in the variable new_location; and assert a command fact to look at the adventurer’s new location (which will trigger the print_place rule and fix the issue that arose when the location rules were generalized):

      
        
        (retract ?c)

        (modify ?p (location ?new_location))

        (assert (command (action look)))

      

      

      
        
        Moving Along an Invalid Path

      

      

      Next the two rules, which check for an invalid path, will now be generalized:

      
        
        (defrule bad_go_from_pit_north

           (thing (id adventurer)

                  (location pit_north))

           ?c <- (command (action go ~south&~up))

           =>

           (retract ?c)

           (println "You can't go there."))

      

      

      
        
        (defrule bad_go_from_pit_south

           (thing (id adventurer)

                  (location pit_south))

           ?c <- (command (action go ~north&~up))

           =>

           (retract ?c)

           (println "You can't go there."))

      

      

      These rules will be deleted from game.clp, and a new generalized rule will be added to kernel.clp:

      
        
        (defrule go_invalid_path

           ?c <- (command (action go ?direction))

           ?p <- (thing (id adventurer)

                        (location ?location))

           (not (path (direction $? ?direction $?)

                      (from $? ?location $?)))

           =>

           (retract ?c)

           (println "You can't go there."))

      

      

      The conditions of this rule are similar to the go_valid_path rule, but there are several changes to the third pattern in order to determine that there are no valid paths:

      
        
        (not (path (direction $? ?direction $?)

                   (from $? ?location $?)

                   (to ?new_location)

                   (blocked FALSE)))

      

      

      The not conditional element flips the logic of the pattern by checking that there is no path fact matching the constraints for a valid path. It does not matter whether the path is blocked or not, so the blocked slot constraint is removed. The to slot constraint is also unnecessary since the location of the adventurer will not be changed. A variable that is first bound within a not conditional element can’t be referenced outside the scope of that conditional element, so changing the adventurer’s location to the variable new_location in the actions of the rule does not make sense anyway because no valid path fact existed from which the value of this variable could have been retrieved.

      The actions of this rule remove the player command and print a message indicating the adventurer can’t move in the desired direction:

      
        
        (retract ?c)

        (println "You can't go there."))

      

      

      
        
        Moving Along a Valid Path that is Blocked

      

      

      The rule that prevents the adventurer from climbing up the walls is the final rule that will be generalized in this chapter:

      
        
        (defrule no_escape

           (thing (id adventurer)

                  (location pit_north | pit_south))

           ?c <- (command (action go | climb up))

           =>

           (retract ?c)

           (println "The walls are too slick."))

      

      

      This rule will be removed from game.clp and a new generalized rule will be added to kernel.clp:

      
        
        (defrule go_valid_path_blocked

           ?c <- (command (action go ?direction))

           ?p <- (thing (id adventurer)

                        (location ?location))

           (path (direction $? ?direction $?)

                 (from $? ?location $?)

                 (to ?new_location)

                 (blocked TRUE)

                 (blocked_message ?text))

           =>

           (retract ?c)

           (println ?text))

      

      

      The conditions of this rule are similar to the go_valid_path rule, but there are two changes to the path pattern to check for a valid path that is blocked:

      
        
        (path (direction $? ?direction $?)

              (from $? ?location $?)

              (to ?new_location)

              (blocked FALSE TRUE)

              (blocked_message ?text))

      

      

      First, the value of the blocked slot must be TRUE, indicating that the path is blocked. Second, the string value in the blocked_message slot is bound to the variable text so that it can be displayed in the actions of the rule. This pattern will match the blocked path from the paths deffacts that was previously created:

      
        
        (path (direction up)

              (from pit_north pit_south)

              (blocked TRUE)

              (blocked_message

                 "The walls are too slick."))

      

      

      Once this rule has determined the adventurer’s path is blocked, the actions of the rule retract the player command and print the message indicating why the path is blocked:

      
        
        (retract ?c)

        (println ?text)

      

      

      The no_escape rule also handled the case when the player entered a climb up command. These rules have been generalized to work with just the go command, so this case will be handled by adding a new rule to kernel.clp to translate the climb up command to a go up command instead:

      
        
        (defrule climb_action

           ?c <- (command (action climb up))

           =>

           (modify ?c (action go up)))

      

      

      You can run the updated program with the new path rules from the Chapter4C directory:

      
        
        CLIPS> (batch* runq.bat)

        Captured by goblins, you've been

        tossed in a pit at their lair.

      

      

      
        
        You're at the pit's north end.

        A giant mushroom is here. The

        ground is littered with the

        bodies of dead adventurers.

      

      

      
        
        > go south

        You're at the pit's south end.

        A large pile of rubble has

        collapsed from the wall above.

      

      

      
        
        > go south

        You can't go there.

      

      

      
        
        > climb up

        The walls are too slick.

      

      

      
        
        > go up

        The walls are too slick.

      

      

      
        
        > quit

        TRUE

        CLIPS>

      

      

      What’s Next

      The next chapter will investigate a better way to generalize the translation of command variations (such as look at mushroom and examine the giant mushroom).

      Problems

      
        
        #4-1

      

      

      Write a  rule that displays a message when a direction other than north, south, east, west, up, or down is used with the go command. For example, a go sideways command would display this message:

      
        
        I don't understand the direction 'sideways'.

      

      

      
        
        #4-2

      

      

      Create a new location called cavern with this description:

      
        
        You're in a cavern lit by glowing algae with a well in the center. A path made of rubble leads up.

      

      

      The well in the cavern is scenery and has this description:

      
        
        This is no ordinary well.

      

      

      When the player first enters a search rubble command at the south side of the pit, display this message:

      
        
        From the top of the rubble, you can see that the ground has collapsed creating a path down to a cavern below.

      

      

      Additionally, create a path down from the south side of the pit to the cavern and a path up from the cavern to the south side of the pit. Add this string to the description of the south side of the pit:

      
        
        A path through the rubble leads down.

      

      

      When the player subsequently enters a search rubble command, display this message:

      
        
        You find nothing of interest.

      

      

      Finally, when the player enters a make wish command in the cavern, halt the game and display this message:

      
        
        Your wish is granted. You've been

        magically transported to safety.

      

      

      
        
        #4-3

      

      

      Add a symmetric slot to the path deftemplate with a default value of FALSE. Add code that will automatically generate a reverse path for any path that has this slot value set to TRUE. For example, if there is a symmetric path fact from the north side of the pit to the south side of the pit in the direction south, a path fact from the south side of the pit to the north side of the pit in the direction north should be created. Print an error message if a symmetric path has more than one location in its from slot.

      The ‘Or’ Conditional Element

      Rather than having the player type go south or go east, it would be better if they could just type south or east. One way this could be done is to create additional rules to handle each variation of a command. For example, a variation of the go_valid_path rule could be added that handles a player command containing just a direction:

      
        
        (defrule go_valid_path_direction_only

           ?c <- (command (action go ?direction))

           ?p <- (thing (id adventurer)

                        (location ?location))

           (path (direction $? ?direction $?)

                 (from $? ?location $?)

                 (to ?new_location)

                 (blocked FALSE))

           =>

           (retract ?c)

           (modify ?p (location ?new_location))

           (assert (command (action look))))

      

      

      The only changes from the original rule is to rename it and modify the command pattern to remove the literal go. The remaining lines don’t require any modifications.

      Creating variations of the rules using this technique would be tedious to maintain. Fortunately, CLIPS provides an or conditional element for handling just this type of situation. Using the or conditional element, the go_valid_path rule can be modified to handle both command variations:

      
        
        (defrule go_valid_path

           ?p <- (thing (id adventurer)

                        (location ?location))

           (or ?c <- (command (action go ?direction))

               ?c <- (command (action ?direction)))

           (path (direction $? ?direction $?)

                 (from $? ?location $?)

                 (to ?new_location)

                 (blocked FALSE))

           =>

           (retract ?c)

           (modify ?p (location ?new_location))

           (assert (command (action look))))

      

      

      The or conditional element works by generating multiple rules where each rule replaces the or conditional element with one of the conditional elements contained within it. For example, this rule contains two or conditional elements each containing two placeholders representing patterns:

      
        
        (defrule example

           (or <pattern-a1> <pattern-a2>)

           (or <pattern-b1> <pattern-b2>)

           =>)

      

      

      This single rule is implemented by generating four rules:

      
        
        (defrule example_1

           <pattern-a1>

           <pattern-b1>

           =>)

      

      

      
        
        (defrule example_2

           <pattern-a1>

           <pattern-b2>

           =>)

      

      

      
        
        (defrule example_3

           <pattern-a2>

           <pattern-b1>

           =>)

      

      

      
        
        (defrule example_4

           <pattern-a2>

           <pattern-b2>

           =>)

      

      

      Even though multiple rules are generated internally by CLIPS, any commands that have rule name arguments (such as the matches command) should be passed the original name of the rule (i.e., there are no suffixes appended to the names of the generated rules).

      Because the actions of the original rule are the same for every generated rule, any variables referenced in the actions of the original rule must be bound in every generated rule. For example, this rule would generate an error because the variable c will be referenced by both generated rules, but is only bound in one of the generated rules:

      
        
        (defrule example

           (or ?c <- (command (action go ?direction))

                     (command (action ?direction)))

           =>

           (retract ?c))

      

      

      Generally, you should only use the or conditional element when it would otherwise be cumbersome to write the rule without it. For example, this rule will be activated by the commands climb up, go up, and climb walls, but also go walls:

      
        
        (defrule example

           (command (action climb | go up | walls))

           =>)

      

      

      It’s possible to exclude the command go walls without using the or conditional element, but this makes it more difficult to see what the conditions are trying to match:

      
        
        (defrule example

           (command (action ?w1&climb|go ?w2&up|walls))

           (test (or (neq ?w1 go) (neq ?w2 walls)))

           =>)

      

      

      Using the or conditional element instead makes it much easier to see what the rule will match:

      
        
        (defrule example

           (or (command (action climb up))

               (command (action go up))

               (command (action climb walls)))

           =>)

      

      

      You can also use the or conditional element when the enclosed patterns match facts with different deftemplates. For example, this rule allows the adventurer to move from the east to the west side of the chasm if the bridge isn’t blocked or they’re wearing the boots of leaping:

      
        
        (defrule cross_chasm

           (or ?c <- (command (action go west))

               ?c <- (command west))

           (or (path (from east_of_chasm)

                     (to west_of_chasm)

                     (blocked FALSE))

               (thing (id boots_of_leaping)

                      (location adventurer)))

           ?a <- (thing (id adventurer)

                        (location east_of_chasm))

           =>

           (retract ?c)

           (modify ?a (location west_of_chasm)))

      

      

      Recognizing Verb Variations

      Rather than using the or conditional element to recognize every command variation, recognition can be simplified by converting actions to a canonical form that is the one variation that rules match. For example, the commands up, go up, and climb up can all be converted to the canonical command go up. Similarly, noun variations such as mushroom, the mushroom, giant mushroom, and the giant mushroom can all be converted to the canonical noun mushroom (which is conveniently the id slot value of the thing fact representing the mushroom).

      Conversion to canonical forms will begin with a modification to the command deftemplate in definitions.clp:

      
        
        (deftemplate command

           (multislot text)

           (multislot action))

      

      

      The text slot is now going to correspond to the command that the player enters, and the action slot is going to correspond to the action that implements the command. So there will be a many-to-one relationship between text and actions. For example, the text values go south and south will both translate to the action values go south.

      With this change, the initial_state deffacts in data.clp will be modified to use the text slot rather than the action slot to store the initial command when the game starts:

      
        
        (deffacts initial_state

           (command (action text look)))

      

      

      A command_pattern deftemplate for specifying the translations between a text command and its associated action will be added to definitions.clp:

      
        
        (deftemplate command_pattern

           (multislot text)

           (multislot action))

      

      

      All commands will now need a command_pattern fact to convert the player command to the corresponding canonical action, even for those cases where there is just one command that translates to a canonical action. Several deffacts will be added to data.clp  to handle the translation of existing commands that will be common to all text adventures. The first will handle the simple translation of the quit command to the quit action:

      
        
        (deffacts command_patterns_core

           (command_pattern (text quit)

                            (action quit)))

      

      

      A separate deffacts will be added to handle translations of direction commands:

      
        
        (deffacts command_patterns_go

           (command_pattern (text south)

                            (action go south))

           (command_pattern (text go south)

                            (action go south))

           (command_pattern (text north)

                            (action go north))

           (command_pattern (text go north)

                            (action go north))

           (command_pattern (text east)

                            (action go east))

           (command_pattern (text go east)

                            (action go east))

           (command_pattern (text west)

                            (action go west))

           (command_pattern (text go west)

                            (action go west))

           (command_pattern (text up)

                            (action go up))

           (command_pattern (text go up)

                            (action go up))

           (command_pattern (text climb up)

                            (action go up))

           (command_pattern (text down)

                            (action go down))

           (command_pattern (text go down)

                            (action go down))

           (command_pattern (text climb down)

                            (action go down)))

      

      

      Now the commands up, go up, and climb up will all translate to the canonical action go up. Other directions are handled similarly. Since the translation of the command climb up to the action go up will now be handled by generic translation rules, the climb_action rule in kernel.clp is no longer needed and can be deleted:

      
        
        (defrule climb_action

           ?c <- (command (action climb up))

           =>

           (modify ?c (action go up)))

      

      

      The last deffacts added will handle translation of the look and look at commands:

      
        
        (deffacts command_patterns_look

           (command_pattern (text look around)

                            (action look))

           (command_pattern (text look)

                            (action look))

           (command_pattern (text look at <thing>)

                            (action look at))

           (command_pattern (text examine <thing>)

                            (action look at)))

      

      

      Translation to the canonical action look is handled similarly to prior actions. Translation to the look at action, however, needs to be handled differently because there is a reference to the target of that action. It would be impractical to write a separate translation for every noun that can be used with this action: look at mushroom, look at bodies, look at rubble, and so forth. Instead, the literal <thing> will be used in the text slot to indicate a placeholder for the target of the action. When the command is translated, the target that is in the placeholder position will be appended to the canonical action.

      Now that the translations are represented by facts, the rules can be modified to support this new approach. First, the get_command rule in kernel.clp will be updated:

      
        
        (defrule get_command

           (declare (salience -10))

           (not (command))

           =>

           (println)

           (print "> ")

           (bind ?rsp (explode$ (lowcase (readline))))

           (assert (command (action text ?rsp))))

      

      

      The single modification to this rule is to store the player’s input in the text slot rather than the action slot. Since the value of the action slot in the assert command is unspecified, by default no values will be stored in that multifield slot when the fact is created. New code in the following sections will populate this slot.

      
        
        Translations Without a Placeholder

      

      

      The simplest translations to handle are those that don’t use the literal <thing> as a placeholder. A new rule added to kernel.clp will handle this situation:

      
        
        (defrule translate

           ?c <- (command (text $?text) (action))

           (test (not (member$ <thing> ?text)))

           (command_pattern (text $?text)

                            (action $?action))

           =>

           (modify ?c (action ?action)))

      

      

      In the first conditional element, the player’s input is assigned to the multifield variable text:

      
        
        ?c <- (command (text $?text) (action))

      

      

      This pattern also checks that the action slot has no values. This indicates that the player’s input has not yet been translated to its canonical form.

      The second conditional element will only be satisfied if the literal <thing> is not contained in the player’s input:

      
        
        (test (not (member$ <thing> ?text)))

      

      

      Since other rules will be handling the placeholder translation, this conditional element excludes those translations. The member$ function returns the position of the first argument within the second argument (which must be a multifield value). If the first argument is within the second argument, an integer  value indicating that position is returned (starting at 1 for the first position); otherwise FALSE is returned. Passing an integer value to the not function will return FALSE, so this conditional element will not be matched if <thing> is found within the player’s input.

      The final conditional element matches the fact that translates the player’s input to its canonical form:

      
        
        (command_pattern (text $?text)

                         (action $?action))

      

      

      The multifield variable text must match the values assigned to it from the first pattern matching the player’s input. The multifield variable action will then be bound to the canonical form for the command.

      In the actions of the rule, the canonical form is assigned to the action slot of the player command:

      
        
        (modify ?c (action ?action))

      

      

      Once the action slot receives a value, this rule will no longer be applicable since the first pattern only matches the player command when the action slot contains no values.

      
        
        Translations With a Placeholder

      

      

      The next rule added to kernel.clp will handle the situation where a translation contains a placeholder and the referenced item is either held by the adventurer or at the same location:

      
        
        (defrule translate_one_thing_match

           ?c <- (command (text $?before ?id) (action))

           (command_pattern (text $?before <thing>)

                            (action $?action))

           (thing (id adventurer)

                  (location ?location))

           (thing (id ?id)

                  (location ?location | adventurer))

           (not (command_pattern (text $?before ?id)))

           =>

           (modify ?c (action ?action ?id)))

      

      

      Consider how this rule would handle a command such as look at mushroom. The first pattern

      
        
        ?c <- (command (text $?before ?id) (action))

      

      

      would bind the multifield variable before to the multifield value

      
        
        (look at)

      

      

      and the variable id would be bound to the literal mushroom. The second pattern

      
        
        (command_pattern (text $?before <thing>)

                         (action $?action))

      

      

      would then only match a command_pattern fact where the text slot contained

      
        
        (look at <thing>)

      

      

      Basically, this pattern maps the item from the first pattern bound to the variable id to the same position in the command as the placeholder <thing> in the second pattern.

      The third pattern assigns the location of the adventurer to the variable location:

      
        
        (thing (id adventurer)

               (location ?location))

      

      

      The fourth pattern verifies that the object represented by the variable id is either in the same room as the adventurer or held by the adventurer:

      
        
        (thing (id ?id)

               (location ?location | adventurer))

      

      

      For the look at mushroom command being considered, the variable id will have the value mushroom, so this pattern will verify that the mushroom is in the same location as or held by the adventurer.

      Finally, the fifth pattern verifies that there is not a translation that handles this specific item without using placeholders:

      
        
        (not (command_pattern (text $?before ?id)))

      

      

      This will allow a more specific translation to override a general translation with a placeholder. For example, if the player issues a look at gorgon command, the command_pattern fact

      
        
        (command_pattern (text look at gorgon)

                         (action turn to stone))

      

      

      would assign the action turn to stone, overriding the more general command_pattern fact

      
        
        (command_pattern (text look at <thing>)

                         (action look at))

      

      

      that would assign the action look at gorgon.

      The only action of the rule is to modify the action of the command to include the action and append the value of the variable id (which is the item assigned to the <thing> placeholder):

      
        
        (modify ?c (action ?action ?id))

      

      

      For the look at mushroom command being considered, if all of the patterns are satisfied, the command is verified and translated to the action look at mushroom.

      
        
        Translations With a Placeholder but no Matching Item

      

      

      The final translation rule added to kernel.clp will print a message when a command is recognized but there is no item mapped to that placeholder (either because the item doesn’t exist or when it does, but the adventurer can’t see it):

      
        
        (defrule translate_one_thing_no_match

           (declare (salience -5))

           ?c <- (command (text $?before ?id) (action))

           (command_pattern (text $?before <thing>))

           =>

           (retract ?c)

           (println "You can't see any such thing."))

      

      

      This rule has been given a salience of -5 so that the other translation rules have the opportunity to execute first. The patterns of this rule check that there’s a command translation for the player’s input, but not whether the item matching the placeholder is something that the adventurer can see. If the prior translation rules are applicable, then one of those rules will execute first, and this rule will be removed from the agenda.

      If there isn’t any command translation at all for the player’s input, the bad_command rule will execute and the

      
        
        I don't understand your command.

      

      

      message will be displayed.

      The updated program can be run from the Chapter5A directory:

      
        
        CLIPS> (batch* runq.bat)

        Captured by goblins, you've been

        tossed in a pit at their lair.

      

      

      
        
        You're at the pit's north end.

        A giant mushroom is here. The

        ground is littered with the

        bodies of dead adventurers.

      

      

      
        
        > look at mushroom

        It's squished. I wouldn't

        try landing on it again.

      

      

      
        
        > examine mushroom

        It's squished. I wouldn't

        try landing on it again.

      

      

      
        
        > look at rubble

        You can't see any such thing.

      

      

      
        
        > look at the mushroom

        I don't understand your command.

      

      

      
        
        > quit

        CLIPS>

      

      

      You can now use either look at or examine commands to get further information, but rather than recognizing just a single noun such as mushroom, it would be better if you could also use the mushroom, giant mushroom, or the giant mushroom. That functionality will be implemented in the next section.

      Recognizing Noun Variations

      To recognize noun variations, the thing deftemplate in definitions.clp will be modified so that prefixes for the noun placeholder can be specified in a translation:

      
        
        (deftemplate thing

           (slot id)

           (slot category

              (allowed-values place item actor scenery))

           (slot location

              (default nowhere))

           (multislot description)

           (multislot prefixes (default "")))

      

      

      The prefixes slot will hold one or more string prefixes for a noun. By default, the value of the slot will be the string "", which will be used to indicate that no prefix is an acceptable option.

      The scenery deffacts in game.clp can now be updated to include the appropriate prefixes for the items that the adventurer can interact with:

      
        
        (deffacts scenery

           (thing (id mushroom)

                  (location pit_north)

                  (category scenery)

                  (prefixes

                     "" "the" "giant" "the giant")

                  (description

                     "It's squished. I wouldn't"

                     "try landing on it again."))

      

      

      
        
           (thing (id bodies)

                  (location pit_north)

                  (category scenery)

                  (prefixes "" "the")

                  (description

                     "Apparently this is what happens"

                     "when you miss the mushroom."))

      

      

      
        
           (thing (id rubble)

                  (location pit_south)

                  (category scenery)

                  (prefixes

                     "" "the" "pile of" "the pile of"

                     "large pile of"

                     "the large pile of")))

      

      

      
        
        Updating the Translation Placeholder Rule

      

      

      The translate_one_thing_match rule from kernel.clp can now be modified to allow prefixes using the updated thing deftemplate:

      
        
        (defrule translate_one_thing_match

           ?c <- (command (text $?before $?prefix ?id)

                          (action))

           (command_pattern (text $?before <thing>)

                            (action $?action))

           (thing (id adventurer)

                  (location ?location))

           (thing (id ?id)

                  (prefixes $? =(implode$ ?prefix) $?)

                  (location ?location | adventurer))

           (not (command_pattern

                   (text $?before $?prefix ?id)))

           =>

           (modify ?c (action ?action ?id)))

      

      

      This rule will be examined in some detail, so run the program from the Chapter5B directory up to the point where the text of the command look at the giant mushroom is translated to an action:

      
        
        CLIPS> (batch* loadq.bat)

        TRUE

        CLIPS> (set-break translate_one_thing_match)

        CLIPS> (run)

        Captured by goblins, you've been

        tossed in a pit at their lair.

      

      

      
        
        You're at the pit's north end.

        A giant mushroom is here. The

        ground is littered with the

        bodies of dead adventurers.

      

      

      
        
        > look at the giant mushroom

        Breaking on rule translate_one_thing_match.

        CLIPS>

      

      

      The first pattern of this rule has been modified to add the multifield variable prefix to match one or more values that may be a prefix for the target noun of the command:

      
        
        ?c <- (command (text $?before $?prefix ?id)

                       (action))

      

      

      This pattern will be matched by f-30, the command fact with a text slot value of look at the giant mushroom.

      
        
        CLIPS> (ppfact 30)

        (command

           (text look at the giant mushroom)

           (action))

        CLIPS>

      

      

      Table 5-1 shows all of the possible ways the first pattern can be matched by this fact.

      
        
          [image: The five ways fact f-30 can match the command fact.]
        

        Table 5-1
      
      Using multiple multifield variables within a pattern generates all possible combinations of values bound to the variables. Not all of these matches contain the values of interest for the player command. For the multifield variable before, the matches where this variable is bound to the multifield values (look) and (look at) correspond to valid commands. Similarly for the multifield variable prefix, the matches where this variable is bound to the multifield values (the giant), (giant), and () correspond to valid prefixes for the mushroom. The match where the variable before is bound to (look at) and the variable prefix is bound to (the giant) is the one match where both values correspond to a valid command.

      The second pattern of the rule is unchanged:

      
        
        (command_pattern (text $?before <thing>)

                         (action $?action))

      

      

      There are only two facts containing placeholders that can match this pattern: f-19 and f-20.

      
        
        CLIPS> (ppfact 19)

        (command_pattern

           (text look at <thing>)

           (action look at))

        CLIPS> (ppfact 20)

        (command_pattern

           (text examine <thing>)

           (action look at))

        CLIPS>

      

      

      Table 5-2 shows how this pattern can be matched by these facts.

      
        
          [image: How facts f-19 and f-20 can match the second pattern.]
        

        Table 5-2
      
      The only match in Table 5-2 where the value for the variable before matches the value in Table 5-1 is when the variable is bound to the multifield value (look at). In order for the rule to match, any subsequent binding of the variable prefix must be the multifield value (the giant).

      The third pattern of the rule is also unchanged and simply retrieves the location of the adventurer:

      
        
        (thing (id adventurer)

               (location ?location))

      

      

      The fourth pattern of the rule is modified to include a prefixes slot pattern using a new type of constraint:

      
        
        (thing (id ?id)

               (prefixes $? =(implode$ ?prefix) $?)

               (location ?location | adventurer))

      

      

      The return value constraint, specified using the symbol =, is satisfied if the field being matched has the same value as the return value of the function call following the = constraint. This constraint is used in this pattern to determine whether the value of the multifield variable prefix, matched and constrained by the first two patterns, is a valid prefix for the item associated with the variable id.

      Because the variable id was previously bound to mushroom, the mushroom thing fact, f-27, is the only fact of interest that matches the fourth pattern.

      
        
        CLIPS> (ppfact 27)

        (thing

           (id mushroom)

           (category scenery)

           (location pit_north)

           (description "It's squished. I wouldn't"

                        "try landing on it again.")

           (prefixes "" "the" "giant" "the giant"))

        CLIPS>

      

      

      The prefixes slot value for this fact is the multifield value

      
        
        ("" "the" "giant" "the giant")

      

      

      Table 5-3 shows how each field constraint in the prefixes slot pattern can be matched by this value. The return value constraint matches a single value, so the multifield wildcards will match the leading and trailing values to the left and right.

      
        
          [image: How the prefixes slot pattern can be matched by fact f-27.]
        

        Table 5-3
      
      For the current command, the multifield variable prefix is bound to the multifield value (the giant). The implode$ function converts a multifield value to a string value. So if the multifield value (the giant) is passed to this function, the return value would be "the giant". So the fourth pattern of the rule would be matched by fact f-27 for the fourth row of Table 5-3 where the = constraint matches the slot value "the giant".

      If, however, the command had been look at the white mushroom, the multifield variable prefix would be bound to (the white) instead, so the fourth pattern would not be matched since the function call for the = constraint would return "the white".

      The matches command confirms this analysis of how the patterns are matched:

      
        
        CLIPS> (matches translate_one_thing_match)

        Matches for Pattern 1

        f-30

        f-30

        f-30

        f-30

        f-30

        Matches for Pattern 2

        f-19

        f-20

        Matches for Pattern 3

        f-21

        Matches for Pattern 4

          .

          .

          .

        f-27

        f-27

        f-27

        f-27

          .

          .

          .

        Matches for Pattern 5

          .

          .

          .

        Partial matches for CEs 1 - 2

        f-30,f-19

        Partial matches for CEs 1 - 3

        f-30,f-19,f-21

        Partial matches for CEs 1 - 4

        f-30,f-19,f-21,f-27

        Partial matches for CEs 1 - 5

        f-30,f-19,f-21,f-27,*

        Activations

        f-30,f-19,f-21,f-27,*

        (54 4 1)

        CLIPS>

      

      

      Note that the output of the matches command shows that fact f-30 matches the first pattern in five different ways (as shown in Table 5-1) and that fact f-27 matches the fourth pattern in four different ways (as shown in Table 5-3).

      
        
        Updating the Translation Placeholder With no Matching Item

      

      

      The translate_one_thing_no_match rule in kernel.clp also needs to be modified to allow a prefix between the target item and the action portion of the command:

      
        
        (defrule translate_one_thing_no_match

           (declare (salience -5))

           ?c <- (command (text $?before $?prefix ?id)

                          (action))

           (command_pattern (text $?before <thing>))

           =>

           (retract ?c)

           (println "You can't see any such thing.")

      

      

      This modification simply adds the multifield variable prefix to capture the portion of the command that comes between the action and the target of the action. Since the translation rules that detect a valid translation have a higher salience, this rule will only have the chance to execute when the multifield variable prefix does not correspond to a valid prefix for the target item bound to the variable id.

      Recognizing Unknown Words

      Finally, a new rule will be added to kernel.clp that informs the player when a specific word is not in the game’s vocabulary:

      
        
        (defrule unrecognized_word

           ?c <- (command (text $? ?word $?))

           (not (command_pattern

                   (text $? ?word&~<thing> $?)))

           (not (thing (id ?word)

                       (category scenery | item)))

           (not (and (thing (prefixes $? ?str $?))

                     (test (member$

                              ?word (explode$ ?str)))))

           =>

           (retract ?c)

           (println "I don't know the word '"

                    ?word "'."))

      

      

      This functionality will be implemented by checking each word in the player’s input to see if it can’t be found in the text of a command pattern, the name of an item, or the prefixes of an item.

      The first pattern generates a match for every word within the text slot of the command fact and binds that value to the variable word:

      
        
        ?c <- (command (text $? ?word $?))

      

      

      The second pattern will be satisfied if the value in the variable word is not contained in the text slot of any command_pattern fact:

      
        
        (not (command_pattern

                (text $? ?word&~<thing> $?)))

      

      

      Although it’s unlikely the player will use the symbol <thing> in one of their commands, this literal will be explicitly excluded so that it’s not recognized by this pattern as valid.

      The third patten will be satisfied if the value in the variable word is not contained in the id slot of any thing fact with a category of scenery or item:

      
        
        (not (thing (id ?word)

                    (category scenery | item)))

      

      

      Restricting the category prevents literals such as pit_north and pit_south from being recognized as valid words in commands.

      The fourth pattern checks to see if the word is included in any of the values found in the prefixes slot of the thing facts:

      
        
        (not (and (thing (prefixes $? ?str $?))

                  (test (member$

                           ?word (explode$ ?str)))))

      

      

      This pattern uses the and conditional element to nest multiple conditional elements within a not conditional element. The and conditional element is satisfied if every conditional element within it is satisfied. The conditions of a rule are implicitly contained within an and conditional element, so you only need to use one when you want to nest patterns within an or or not conditional element. You can freely nest and, or, and not conditional elements within each other.

      For the and conditional element in this pattern, the constraints try to locate the symbol bound to the variable word within a string containing one or more symbols. The nested thing pattern binds the variable str to each value contained in the prefix slot. The nested test pattern uses the explode$ function to convert the string value in the variable str to a multifield value containing symbols. The member$ function is then used to determine if the variable word from the player’s input is contained in that multifield value. If no fact can be found that contains the word within one of the prefix strings, then the inner and conditional element will not be satisfied, which will cause the outer not conditional element to be satisfied indicating that there is no prefix containing the word.

      You may have thought that it wasn’t absolutely necessary to use the and conditional element for the fourth pattern, and you’d be right. In this particular case, the test conditional element could have been moved inside the prefixes slot constraint and attached to the variable str using the : field constraint:

      
        
        ?str&:(member$ ?word (explode$ ?str))

      

      

      Sometimes you might find that your code is more readable when you split some of the expressions from your pattern and place them in a separate test conditional element.

      When all of the patterns of this rule have been satisfied, then the word contained in the player’s input is not part of the game’s vocabulary. The actions of the rule can then be applied to remove the player command and print a message indicating that the word isn’t recognized:

      
        
        (retract ?c)

        (println "I don't know the word '" ?word "'.")

      

      

      The updated program can be run from the Chapter5C directory:

      
        
        CLIPS> (batch* runq.bat)

        Captured by goblins, you've been

        tossed in a pit at their lair.

      

      

      
        
        You're at the pit's north end.

        A giant mushroom is here. The

        ground is littered with the

        bodies of dead adventurers.

      

      

      
        
        > look at mushroom

        It's squished. I wouldn't

        try landing on it again.

      

      

      
        
        > examine the giant mushroom

        It's squished. I wouldn't

        try landing on it again.

      

      

      
        
        > look mushroom

        I don't understand your command.

      

      

      
        
        > look at the white mushroom

        I don't know the word 'white'.

      

      

      
        
        > look at the pile of rubble

        You can't see any such thing.

      

      

      
        
        > look around

        You're at the pit's north end.

        A giant mushroom is here. The

        ground is littered with the

        bodies of dead adventurers.

      

      

      
        
        > quit

        CLIPS>

      

      

      What’s Next

      The next chapter will introduce fact queries, which provide the ability to iterate over a set of facts outside of the conditions of a rule. Commands will also be added to allow the adventurer to acquire and use items discovered in the pit.

      Problems

      
        
        #5-1

      

      

      Write a program that will take a phrase such as

      
        
        the pit's south end is south of

        the pit's north end

      

      

      and automatically create path facts for the specified direction as well as the return direction. For example, the phrase fact

      
        
        (phrase (words the pit's south end

                       is south of

                       the pit's north end)))

      

      

      should generate the following path facts:

      
        
        (path (direction south)

              (from the pit's north end)

              (to the pit's south end))

      

      

      
        
        (path (direction north)

              (from the pit's south end)

              (to the pit's north end))

      

      

      
        
        #5-2

      

      

      Write a program that will take a phrase such as the giant white mushroom is scenery and automatically generate a thing fact containing a list of all possible prefixes. For example, the phrase fact

      
        
        (phrase (words the giant white

                       mushroom is scenery))

      

      

      should generate a thing fact such as:

      
        
        (thing (id mushroom)

               (category scenery)

               (prefixes

                  "" "the giant white" "white"

                  "giant white" "giant" "the"

                  "the white" "the giant"))

      Discovering Things

      The adventurer is never getting out of this pit unless they can find something to help them, so the next addition to the adventure will be a goblin with some magic beans that can be discovered in the rubble.

      To begin, the thing deftemplate in definitions.clp will be modified to include some additional slots:

      
        
        (deftemplate thing

           (slot id)

           (slot category  

              (allowed-values place item actor scenery))

           (slot location

              (default nowhere))

           (multislot description)

           (multislot prefixes (default ""))

           (slot definite (default ""))

           (slot indefinite (default ""))

           (multislot attributes))

      

      

      The definite slot will contain the phrase used to refer to a specific thing (such as “the giant mushroom”), whereas the indefinite slot will contain the phrase for generally referring to an item (such as “a giant mushroom”). The attributes slot will be used to store attributes of a thing. For example, if a thing can be picked up by the adventurer, the symbol can_be_taken will be added to the attributes slot.

      Next, the scenery deffacts in game.clp will be modified to include definite and indefinite slot values for the mushroom, bodies, and rubble items:

      
        
        (deffacts scenery

           (thing (id mushroom)

                  (location pit_north)

                  (category scenery)

                  (definite "the giant mushroom")

                  (indefinite "a giant mushroom")

                  (prefixes

                     "" "the" "giant" "the giant")

                  (description

                     "It's squished. I wouldn't"

                     "try landing on it again."))

      

      

      
        
           (thing (id bodies)

                  (location pit_north)

                  (category scenery)

                  (definite "the bodies")

                  (indefinite "bodies")

                  (prefixes "" "the")

                  (description

                     "Apparently this is what happens"

                     "when you miss the mushroom."))

      

      

      
        
           (thing (id rubble)

                  (location pit_south)

                  (category scenery)

                  (definite "the rubble")

                  (indefinite "rubble")

                  (prefixes

                     "" "the" "pile of" "the pile of"

                     "large pile of"

                     "the large pile of")))

      

      

      The goblin and magic beans will be created in a new deffacts in game.clp:

      
        
        (deffacts things

           (thing (id goblin)

                  (location rubble)

                  (category item)

                  (prefixes "" "the" "dead" "the dead")

                  (definite "the dead goblin")

                  (indefinite "a dead goblin"))

      

      

      
        
           (thing (id beans)

                  (location goblin)

                  (category item)

                  (prefixes "" "the")

                  (definite "the beans")

                  (indefinite "some beans")

                  (description

                     "These are no ordinary beans.")

                  (attributes can_be_taken)))

      

      

      The category values of the goblin and the rubble have been set to item rather than scenery since they aren’t included in the description of the south side of the pit.

      The program will treat any thing with a category slot value of scenery as something that can’t be taken. A thing with a category slot value of item may or may not be taken. If a thing can be taken, the literal can_be_taken will be added to its attributes slot (as is done for the beans).

      The goblin will be buried in the rubble, so the adventurer won’t be able to discover the creature just by looking at the rubble—they’ll have to search through it. A new search command will be added to  the end of  the command_patterns_look deffacts in data.clp:

      
        
        (deffacts command_patterns_look

           .

           .

           .

           (command_pattern (text search <thing>)

                            (action search)))

      

      

      Implementation of the search command will begin by adding a general rule to kernel.clp that will be the default result if no other rule processes the command:

      
        
        (defrule search_nothing_found

           (declare (salience -5))

           ?c <- (command (action search ?id))

           =>

           (retract ?c)

           (println "You find nothing of interest."))

      

      

      Since this rule will have a lower salience than other search rules, it will only be executed if no other rule handles the command. The actions remove the command and print a message indicating that nothing of interest was found.

      Next, a rule will be added to game.clp to reveal the goblin when the adventurer searches the rubble:

      
        
        (defrule search_rubble

           ?c <- (command (action search rubble))

           (thing (id adventurer)

                  (location ?location))

           ?g <- (thing (id goblin)

                        (location rubble))

           =>

           (retract ?c)

           (modify ?g (location ?location))

           (println "You find a dead goblin.")

           (println "He probably fell from above."))

      

      

      The salience of this rule is the default value of 0, so if its conditions are satisfied, it will execute before the generic search rule. The conditions bind the current location of the adventurer to the variable location and check that the goblin is currently located in the rubble.

      The actions remove the current command, change the location of the goblin from the rubble to the current location of the adventurer, and print a message. The command translation rules handle the situation where the rubble and the adventurer aren’t at the same location, so this rule doesn’t have to check for this condition. By retrieving the location from the adventurer in the conditions rather than using the pit_south symbol, the location of the rubble can be changed and the search_rubble rule will continue to work properly.

      Once the goblin has been discovered, its location slot value is no longer rubble, so this rule will no longer be applicable if the player issues another search rubble command. This allows the generic search rule to execute and indicate that there is nothing more of interest in the rubble.

      A similar rule will be added to game.clp for revealing the beans:

      
        
        (defrule search_goblin

           ?c <- (command (action search goblin))

           (thing (id adventurer)

                  (location ?location))

           ?b <- (thing (id beans)

                        (location goblin))

           =>

           (retract ?c)

           (modify ?b (location ?location))

           (println "Some beans are in his pockets."))

      

      

      Again, the command translation rules handle the situation where the goblin and the adventurer are not at the same location, so until the goblin is revealed, this rule can’t be activated.

      The updated program can be run from the Chapter6A directory:

      
        
        CLIPS> (batch* runq.bat)

        Captured by goblins, you've been

        tossed in a pit at their lair.

      

        

      
        You're at the pit's north end.

        A giant mushroom is here. The

        ground is littered with the

        bodies of dead adventurers.

      

        

      
        > search the mushroom

        You find nothing of interest.

      

        

      
        > go south

        You're at the pit's south end.

        A large pile of rubble has

        collapsed from the wall above.

      

        

      
        > search rubble

        You find a dead goblin.

        He probably fell from above.

      

        

      
        > search goblin

        Some beans are in his pockets.

      

        

      
        > quit

        CLIPS>

      

      

      Ordered Facts

      The adventurer’s search has revealed the goblin and the goblin’s beans, but a look command does not include these newly discovered items in the description of the adventurer’s location:

      
        
        CLIPS> (run)

      

      

      
        
        > look

        You're at the pit's south end.

        A large pile of rubble has

        collapsed from the wall above.

      

      

      
        
        > quit

        CLIPS>

      

      

      To include these items in a location’s description, the print_place rule in kernel.clp could be modified:

      
        
        (defrule print_place

           ?c <- (command (action look))

           (thing (id adventurer) (location ?location))

           (thing (id ?location)

                  (category place)

                  (description $?text))

           =>

           (retract ?c)

           (foreach ?line ?text

              (println ?line))

           (assert (list-things ?location)))

      

      

      The modification to the actions of this rule now asserts a list-things fact after the basic description of the adventurer’s location is printed.

      All of the facts that have been created up to this point are associated with a previously defined deftemplate (and are referred to as deftemplate facts). The list-things fact is an example of an ordered fact (a type of fact that does not require the prior definition of a deftemplate). Ordered facts have a single multifield slot that is not specified when you assert a fact or match that fact in a pattern. When CLIPS encounters an assert command or a pattern that references a deftemplate that does not exist, it automatically creates an implied deftemplate that has a single multifield slot. This is convenient when you want to assert a fact that has a single slot or no slots at all—you just need to be careful that your assertions and patterns use the same symbol (for example, if you assert a list-things fact but your pattern matches a list-thing fact, you’ll have a rule that doesn’t activate).

      An additional rule could now be added that lists every item with the same location as the adventurer:

      
        
        (defrule list-things

           (list-things ?location)

           (thing (location ?location)

                  (category item)

                  (indefinite ?text))

           =>

           (println "You can see " ?text "."))

      

      

      This rule will be activated when a list-things fact is asserted and will fire once for each thing fact that has a category slot value of item and is at the same location as the adventurer. This will list things such as the goblin and beans, but not things like the pile of rubble.

      Once listing is complete, a low salience rule could be used to remove the list-things fact so that a subsequent look command will be able to assert this fact again:

      
        
        (defrule done-list-things

           (declare (salience -5))

           ?lt <- (list-things $?)

           =>

           (retract ?lt))

      

      

      Unlike a deftemplate pattern, matching any set of values in an ordered pattern requires the explicit use of a multifield wildcard constraint. However, in this case, a single-field wildcard could also have been used since the list-things fact is always asserted with a single value in the implied slot.

      Although these rules work together to print the list of things the adventurer can see, it would be better if this functionality was completely implemented in the actions of the print_place rule. The next two sections will show how this can be accomplished.

      Fact Query Functions

      The fact query functions are used to collect sets of facts matching a condition or to iterate over these sets performing an action. Normally you’d use rules for performing this type of task, but there are some situations where it is more appropriate to use these functions.

      The simplest query to make is retrieving all facts of a particular type:

      
        
        CLIPS> (find-all-facts ((?t thing)) TRUE)

        (<Fact-22> <Fact-23> <Fact-24> <Fact-28>

         <Fact-29> <Fact-30> <Fact-31> <Fact-32>)

        CLIPS>

      

      

      The find-all-facts function iterates over one or more sets of facts, determines if the facts satisfy a condition, and then returns a multifield value containing the matching facts. The fact query functions are special forms, so their syntax differs from standard function calls. The name of the query function is followed by a specification of the sets of facts over which iteration will occur. In this example, iteration will occur over a single set of thing facts:

      
        
        ((?t thing))

      

      

      A fact sets specification begins and ends with matching parentheses, and each fact set of the specification also begins and ends with matching parentheses. Each fact set specifies a variable followed by a template name. As iteration occurs over the facts of the specified deftemplate, the current value being examined is stored in the fact set variable. In this example, as iteration occurs over the thing facts, the current thing fact is stored in the variable t.

      The fact sets specification is followed by a condition that must evaluate to a value other than FALSE in order for the facts in the current iteration to be included in the return value of the function. In this example, the condition is the symbol TRUE, so every thing fact will be included in the return value of the function.

      The ppfact command can be used to display one of the facts from the query return value:

      
        
        CLIPS> (ppfact 22)

        (thing

           (id adventurer)

           (category actor)

           (location pit_south)

           (description)

           (prefixes "")

           (definite "")

           (indefinite "")

           (attributes))

        CLIPS>

      

      

      Several slots from this fact have default slot values. The display of default values can be restricted by providing additional arguments to the ppfact command:

      
        
        CLIPS> (ppfact 22 stdout TRUE)

        (thing

           (id adventurer)

           (category actor)

           (location pit_south))

        CLIPS>

      

      

      The second argument to the ppfact command, stdout, is a logical name. Logical names are used to refer to input sources and output destinations. For example, when a file is opened in CLIPS using the open command, it is associated with a logical name. Some logical names, such as stdout for standard output and stdin for standard input, are automatically provided by CLIPS. If the ppfact command is only given a single argument, it displays its output to stdout.

      The third argument to the ppfact command indicates whether slots with default values are excluded from display. If unspecified, this argument defaults to FALSE. In this example, TRUE is specified, so the default values for the description, prefixes, definite, indefinite, and attributes slots are not displayed.

      The facts returned by the previous example can be restricted by changing the query condition so that only some facts satisfy it:

      
        
        CLIPS>

        (find-all-facts ((?t thing))

           (eq ?t:category item))

        (<Fact-31> <Fact-32>)

        CLIPS> (ppfact 31)

        (thing

           (id goblin)

           (category item)

           (location pit_south)

           (description)

           (prefixes "" "the" "dead" "the dead")

           (definite "the dead goblin")

           (indefinite "a dead goblin")

           (attributes))

        CLIPS>

      

      

      The query condition has been changed to

      
        
        (eq ?t:category item)

      

      

      In this example, t:category is a special kind of variable that can be used in the condition and actions of a query function. A variable slot shortcut consists of a fact set variable and a slot name separated by a : character. It allows a slot value to be retrieved from a fact bound to a fact set variable. In this example, the variable slot shortcut t:category retrieves the category slot value from the thing fact bound to the fact set variable t. The query condition in this case compares the slot shortcut value to the symbol item using the eq function. Thus the only facts returned by this query function are thing facts with a category slot value of item.

      Rather than using the ppfact command to inspect the facts returned by the query, a group of actions can be performed during the query iteration using the do-for-all-facts function:

      
        
        CLIPS>

        (do-for-all-facts ((?t thing))

           (eq ?t:category item)

           (println ?t:id))

        goblin

        beans

        CLIPS>

      

      

      The single action in this query prints the id slot value for each thing fact with a category slot value of item. Since only two of the eight thing facts satisfy the query condition, only the id slot values for those facts are printed.

      The find-fact function can be used to retrieve the first fact found matching a query. In this example, it’s used to retrieve the fact representing the adventurer:

      
        
        CLIPS>

        (find-fact ((?a thing))

           (eq ?a:id adventurer))

        (<Fact-22>)

        CLIPS>

      

      

      The do-for-fact query function can be used to perform an action on the first fact found. In this example, it’s used to retrieve the current location of the adventurer:

      
        
        CLIPS>

        (do-for-fact ((?a thing))

           (eq ?a:id adventurer)

           ?a:location)

        pit_south

        CLIPS>

      

      

      Since the variable slot shortcut a:location is the last (and only) action executed, it becomes the return value of the function.

      It’s also possible to iterate over multiple sets of facts. For example, this query lists the scenery at each location:

      
        
        CLIPS>

        (do-for-all-facts ((?p thing) (?s thing))

           (and (eq ?p:category place)

                (eq ?s:category scenery)

                (eq ?s:location ?p:id))

           (println ?p:id ": " ?s:id))

        pit_north: mushroom

        pit_north: bodies

        pit_south: rubble

        CLIPS>

      

      

      This query iterates over all combinations of two thing facts and binds the first to the fact set variable p and the second to the fact set variable s. If the first fact has a category slot value of place, the second fact has a category slot value of scenery, and the location slot value of the second fact is the id slot value of the first fact, then the id slot values of the two facts are printed.

      Nested query calls can also be used for this example to achieve the same effect:

      
        
        CLIPS>

        (do-for-all-facts ((?p thing))

           (eq ?p:category place)

           (do-for-all-facts ((?s thing))

              (and (eq ?s:category scenery)

                   (eq ?s:location ?p:id))

              (println ?p:id ": " ?s:id)))

        pit_north: mushroom

        pit_north: bodies

        pit_south: rubble

        CLIPS>

      

      

      Listing The Items At A Place

      In this section, fact queries will be used to update the actions of the print_place rule to display the list of items the adventurer can see that aren’t scenery included in the description of the adventurer’s location.

      First, a few minor changes will be made to the places deffacts in game.clp:

      
        
        (deffacts places

           (thing (id pit_north)

                  (category place)

                  (definite "the pit's north end")

                  (description

                     "You're at the pit's north end."

                     "A giant mushroom is here. The"

                     "ground is littered with the"

                     "bodies of dead adventurers."))

      

      

      
        
           (thing (id pit_south)

                  (category place)

                  (definite "the pit's south end")

                  (description

                     "You're at the pit's south end."

                     "A large pile of rubble has"

                     "collapsed from the wall above.")

                  (attributes hard_ground)))

      

      

      The general name of a location has been moved from the description slot to the definite slot. This allows some flexibility in the amount of information that’s displayed when the adventurer moves from one location to another. For example, the general name and description of a location could be displayed the first time the adventurer visits a location, but only the general name could be displayed on subsequent visits.

      In addition, hard_ground has been added to the attributes slot of the south side of the pit. This will be used later when rules for planting are added.

      The print_place rule can now be modified to include listing the room inventory:

      
        
        (defrule print_place

           ?c <- (command (action look))

           (thing (id adventurer) (location ?location))

           (thing (id ?location)

                  (category place)

                  (definite ?definite)

                  (description $?text))

           =>

           (retract ?c)

           (println "You're at " ?definite ".")

           (foreach ?line ?text

              (println ?line))

           (do-for-all-facts ((?t thing))

              (and (eq ?t:location ?location)

                   (eq ?t:category item))

              (println "You can see " ?t:indefinite

                       ".")))

      

      

      Since the general name of a location has been moved to the definite slot, this value is retrieved in the conditions of the rule. This text is then displayed by adding a println command in the actions of the rule. A do-for-all-facts command is also added to iterate over each thing fact and output the indefinite text for each one that has a category slot value of item and is at the same location as the adventurer.

      Now items discovered by the adventurer will be listed when the adventurer enters a room or is given a look command. The updated program can be run from the Chapter6B directory:

      
        
        CLIPS> (batch* runq.bat)

        Captured by goblins, you've been

        tossed in a pit at their lair.

      

      

      
        
        You're at the pit's north end.

        A giant mushroom is here. The

        ground is littered with the

        bodies of dead adventurers.

      

      

      
        
        > south

        You're at the pit's south end.

        A large pile of rubble has

        collapsed from the wall above.

      

      

      
        
        > search rubble

        You find a dead goblin.

        He probably fell from above.

      

      

      
        
        > look

        You're at the pit's south end.

        A large pile of rubble has

        collapsed from the wall above.

        You can see a dead goblin.

      

      

      
        
        > search goblin

        Some beans are in his pockets.

      

      

      
        
        > look

        You're at the pit's south end.

        A large pile of rubble has

        collapsed from the wall above.

        You can see a dead goblin.

        You can see some beans.

      

      

      
        
        > quit

        CLIPS>

      

      

      The Take, Drop, and Inventory Commands

      Now that the adventurer can see the items at their location, they’ll need new commands to pick them up, drop them, and list the ones they’re currently holding. Several new command patterns will be added to the command_patterns_core deffacts in data.clp file to recognize these commands :

      
        
        (deffacts command_patterns_core

           (command_pattern (text quit)

                            (action quit))

           (command_pattern (text get <thing>)

                            (action take))

           (command_pattern (text take <thing>)

                            (action take))

           (command_pattern (text grab <thing>)

                            (action take))

           (command_pattern (text drop <thing>)

                            (action drop))

           (command_pattern (text inventory)

                            (action inventory)))

      

      

      The get, take, and grab commands will allow the adventurer to pick up items from their current location; the drop command will allow the adventurer to drop items at their current location; and the inventory command will list all items held by the adventurer. These commands all represent core functionality for any adventure, so the rules implementing these commands will all be added to kernel.clp.

      
        
        The Take Command

      

      

      The preexisting rules for processing commands will handle the situation where the adventurer attempts to take an item that they aren’t holding or isn’t at the same location, so the specific rules for the take command don’t have to check for this as well. The first new rule handles the case where the adventurer attempts to pick up something that can be taken (such as the beans):

      
        
        (defrule take_valid_thing

           ?c <- (command (action take ?id))

           ?t <- (thing (id ?id)

                        (attributes $? can_be_taken $?)

                        (location ?location)

                        (definite ?text))

           (thing (id adventurer)

                  (location ?location))

           =>

           (retract ?c)

           (modify ?t (location adventurer))

           (println "You take " ?text "."))

      

      

      The two key portions of this rule’s conditions check that the attributes slot contains the value can_be_taken and that the location slot of the adventurer and the item being taken are the same. In the actions of the rule, the location slot of the item taken is set to adventurer, indicating that it is now held.

      The next rule handles the case where the adventurer attempts to pick up something that cannot be taken (such as the goblin or the rubble):

      
        
        (defrule take_invalid_thing

           ?c <- (command (action take ?id))

           (thing (id ?id)

                  (attributes $?attributes)

                  (location ?location)

                  (definite ?text))

           (test (not (member$

                         can_be_taken ?attributes)))

           (thing (id adventurer)

                  (location ?location))

           =>

           (retract ?c)

           (println "You can't take " ?text "."))

      

      

      The key difference between this rule and the prior one is that the content of the attributes slot is bound to the multifield variable attributes and then used in a test conditional element to check that can_be_taken is not a member of this slot.

      The final rule handles the case when the adventurer is already holding an item:

      
        
        (defrule take_carried_thing

           ?c <- (command (action take ?id))

           (thing (id ?id)

                  (location adventurer)

                  (definite ?text))

           =>

           (retract ?c)

           (println "You already have " ?text "."))

      

      

      The key condition in this rule checks that the location slot of the item being taken is already set to adventurer.

      
        
        The Drop Command

      

      

      Two cases need to be handled for the drop command. The first case is when the adventurer is holding the item being dropped:

      
        
        (defrule drop_valid_thing

           ?c <- (command (action drop ?id))

           ?t <- (thing (id ?id)

                        (location adventurer)

                        (definite ?text))

           (thing (id adventurer)

                  (location ?location))

           =>

           (retract ?c)

           (modify ?t (location ?location))

           (println "You drop " ?text "."))

      

      

      The key pattern for this rule checks that the location slot of the item begin dropped contains the value adventurer. In the actions of the rule, the location slot of the item dropped is set to the location of the adventurer (which was assigned to the variable location in the conditions of the rule).

      The next case is when the adventurer is not holding the item:

      
        
        (defrule drop_invalid_thing

           ?c <- (command (action drop ?id))

           (thing (id ?id)

                  (location ~adventurer)

                  (definite ?text))

           =>

           (retract ?c)

           (println "You don't have " ?text "."))

      

      

      The key pattern for this rule checks that the location slot of the item begin dropped does not contain the value adventurer.

      
        
        The Inventory Command

      

      

      Two cases need to be handled for the inventory command. The first case handles the situation when the adventurer is not holding anything:

      
        
        (defrule inventory_empty

           ?c <- (command (action inventory))

           (not (thing (category item)

                       (location adventurer)))

           =>

           (retract ?c)

           (println "You are not carrying anything."))

      

      

      The key pattern in this rule uses a not conditional element to check that no thing fact exists with a category slot value of item and a location slot value of adventurer.

      The other case handles the situation when the adventurer is holding at least one item:

      
        
        (defrule inventory_not_empty

           ?c <- (command (action inventory))

           (exists (thing (category item)

                          (location adventurer)))

           =>

           (retract ?c)

           (println "You are carrying:")

           (do-for-all-facts ((?t thing))

              (and (eq ?t:category item)

                   (eq ?t:location adventurer))

              (println  "   " ?t:indefinite)))

      

      

      The second pattern in this rule introduces the exists conditional element:

      
        
        (exists (thing (category item)

                       (location adventurer)))

      

      

      The outer exists CE is satisfied if there is at least one occurrence of the inner CE that is satisfied. So for this rule, the exists CE will be satisfied if there is at least one thing fact with a category slot value of item and a location slot value of adventurer.

      The exists CE is functionally equivalent to the use of two nested not conditional elements:

      
        
        (not (not (thing (category item)

                         (location adventurer))))

      

      

      If there are no thing facts held by the adventurer, then the inner not CE will be satisfied, which causes the outer not CE to be unsatisfied. Conversely, if there are thing facts held by the adventurer, then the inner not CE is unsatisfied, which causes the outer not CE to be satisfied.

      Writing this rule using the exists CE guarantees that there is just one activation for the rule regardless of how many items the adventurer is holding. This pattern for the rule could also have written without using the exists CE:

      
        
        (thing (category item)

               (location adventurer))

      

      

      In this case, the rule would be activated multiple times, once for every item held by the adventurer. When the first activation fires, the inventory command will be retracted causing the remaining activations for the rule to be removed, achieving the desired result. However, since the exists CE most accurately reflects the intent of the rule, it’s preferable to use it since it prevents unwanted activations from being added to the agenda.

      The key actions of the rule display the items held by the adventurer:

      
        
        (println "You are carrying:")

        (do-for-all-facts ((?t thing))

           (and (eq ?t:category item)

                (eq ?t:location adventurer))

           (println  "   " ?t:indefinite))

      

      

      The do-for-all-facts function is used to iterate over and print each item that is held by the adventurer.

      With the take, drop, and inventory commands implemented, the updated program can be run from the Chapter6C directory:

      
        
        CLIPS> (batch* runq.bat)

        Captured by goblins, you've been

        tossed in a pit at their lair.

      

      

      
        
        You're at the pit's north end.

        A giant mushroom is here. The

        ground is littered with the

        bodies of dead adventurers.

      

      

      
        
        > inventory

        You are not carrying anything.

      

      

      
        
        > go south

        You're at the pit's south end.

        A large pile of rubble has

        collapsed from the wall above.

      

      

      
        
        > search rubble

        You find a dead goblin.

        He probably fell from above.

      

      

      
        
        > search goblin

        Some beans are in his pockets.

      

      

      
        
        > get beans

        You take the beans.

      

      

      
        
        > inventory

        You are carrying:

           some beans

      

      

      
        
        > look at the beans

        These are no ordinary beans.

      

      

      
        
        > quit

        CLIPS>

      

      

      The adventurer is now in possession of some magic beans. The next section will implement functionality that will allow the adventurer to use them.

      Planting the Beans and Climbing the Stalk

      In order to escape, the adventurer will need to plant the magic beans and climb up the stalk that grows from them. To represent the stalk, a new item will be added to the things deffacts in game.clp:

      
        
        (deffacts things

           .

           .

           .

           (thing (id stalk)

                  (category item)

                  (prefixes

                     "" "the" "giant" "the giant")

                  (definite "the giant stalk")

                  (indefinite "a giant stalk")

                  (description

                     "It's sturdy enough to climb.")))

      

      

      The plant command will be different from prior commands. The adventurer will have to be holding the beans in order to plant them. The command deftemplate in definitions.clp will be modified to allow attributes to be associated with commands:

      
        
        (deftemplate command_pattern

           (multislot text)

           (multislot action)

           (multislot attributes))

      

      

      Next, a new deffacts will be created in game.clp to contain the plant and climb commands:

      
        
        (deffacts command_patterns_adventure

           (command_pattern (text plant <thing>)

                            (action plant)

                            (attributes must_be_held))

           (command_pattern (text climb <thing>)

                            (action climb)))

      

      

      
        
        Planting in Hard Ground

      

      

      Previously in this chapter, the south side of the pit was assigned the hard_ground attribute in its attributes slot. The adventurer won’t be able to plant the beans at a location where the ground is too hard. A new rule will be added to game.clp to handle this situation:

      
        
        (defrule plant_beans_hard_ground

           ?c <- (command (action plant beans))

           (thing (id adventurer)

                  (location ?location))

           (thing (id ?location)

                  (attributes $?attributes))

           (test (member$ hard_ground ?attributes))

           =>

           (retract ?c)

           (println "The ground is too hard."))

      

      

      The first CE checks for a plant beans command:

      
        
        ?c <- (command (action plant beans))

      

      

      The second and third CEs bind the multifield variable attributes to the attribute values of the adventurer’s current location:

      
        
        (thing (id adventurer)

               (location ?location))

        (thing (id ?location)

               (attributes $?attributes))

      

      

      Lastly, the fourth CE checks that the attributes of the adventurer’s current location contains the hard_ground attribute:

      
        
        (test (member$ hard_ground ?attributes))

      

      

      If all of the conditions of the rule are satisfied, the actions remove the plant beans command and print a message indicating that the ground is too hard.

      
        
        Planting in Soft Ground

      

      

      Another rule will be added to game.clp to handle the situation where the ground is soft enough for planting:

      
        
        (defrule plant_beans_soft_ground

           ?c <- (command (action plant beans))

           ?b <- (thing (id beans)

                        (location ~nowhere))

           (thing (id adventurer)

                  (location ?location))

           (thing (id ?location)

                  (attributes $?attributes))

           (test (not (member$

                         hard_ground ?attributes)))

           ?s <- (thing (id stalk)

                        (location nowhere))

           =>

           (retract ?c)

           (modify ?b (location nowhere))

           (modify ?s (location ?location))

           (println "Apparently the beans are magical.")

           (println "They grow into a giant stalk."))

      

      

      Many of the CEs in this rule are similar to the prior rule for planting the beans in hard ground. A new CE is added to match the item representing the beans that have been discovered:

      
        
        ?b <- (thing (id beans)

                     (location ~nowhere))

      

      

      The fifth CE is changed to check that the ground is not hard:

      
        
        (test (not (member$ hard_ground ?attributes)))

      

      

      And another CE is added to match the item representing the stalk:

      
        
        ?s <- (thing (id stalk)

                     (location nowhere))

      

      

      When the location slot value of the stalk is nowhere, the beans have not yet been planted.

      The first action of the rule removes the plant beans command:

      
        
        (retract ?c)

      

      

      The second action of the rule changes the location of the beans item to nowhere:

      
        
        (modify ?b (location nowhere))

      

      

      The beans are now no longer visible to the adventurer. If the beans item had been retracted instead, then the word beans would no longer be recognized as part of the game’s vocabulary.

      The third action places the stalk at the player’s current location:

      
        
        (modify ?s (location ?location))

      

      

      The remaining actions simply print a message indicating the beans have grown into a giant stalk.

      
        
        Items that Must be Held

      

      

      A generic rule will be added to kernel.clp to verify that an item that must be held for a specific command is actually held by the adventurer:

      
        
        (defrule thing_must_be_held

           (declare (salience 10))

           (command_pattern

              (text $?action <thing>)

              (attributes $? must_be_held $?))

           ?c <- (command (action $?action ?thing))

           (thing (id ?thing)

                  (location ~adventurer)

                  (definite ?definite))

           =>

           (retract ?c)

           (println "You're not holding " ?definite "."))

      

      

      The rule has been given a salience of 10. This will allow it to execute before any rules that process the adventurer’s command (such as the plant_beans_hard_ground rule).

      The first pattern matches any command pattern that matches a <thing> placeholder that has the attribute must_be_held:

      
        
        (command_pattern

           (text $?action <thing>)

           (attributes $? must_be_held $?))

      

      

      The second pattern is matched when the player’s current command is the same as the matched command pattern:

      
        
        ?c <- (command (action $?action ?thing))

      

      

      The third pattern matches the item associated with the player command, but only if it is not held by the adventurer:

      
        
        (thing (id ?thing)

               (location ~adventurer)

               (definite ?definite))

      

      

      The actions of the rule remove the command and print a message indicating that the item is not held by the adventurer.

      
        
        Up And Out

      

      

      Now that the adventurer can plant the beans and watch them grow, one additional rule will be added to game.clp to allow the adventurer to climb the stalk:

      
        
        (defrule climb_stalk

           ?c <- (command (text climb stalk))

           (thing (id adventurer)

                  (location ?location))

           (thing (id stalk)

                  (location ?location))

           =>

           (retract ?c)

           (println "You have escaped!")

           (println)

           (halt))

      

      

      The first pattern of the rule determines that the player has issued a climb stalk command:

      
        
        ?c <- (command (text climb stalk))

      

      

      The second and third patterns establish that the adventurer is at the same location as the stalk:

      
        
        (thing (id adventurer)

               (location ?location))

        (thing (id stalk)

               (location ?location))

      

      

      The actions of the rule remove the command, print a message indicating that the adventurer has escaped, and halt execution.

      
        
        Trying the Plant and Climb Stalk Commands

      

      

      The updated program can be run from the Chapter6D directory:

      
        
        CLIPS> (batch* runq.bat)

        Captured by goblins, you've been

        tossed in a pit at their lair.

      

      

      
        
        You're at the pit's north end.

        A giant mushroom is here. The

        ground is littered with the

        bodies of dead adventurers.

      

      

      
        
        > south

        You're at the pit's south end.

        A large pile of rubble has

        collapsed from the wall above.

      

      

      
        
        > search rubble

        You find a dead goblin.

        He probably fell from above.

      

      

      
        
        > search goblin

        Some beans are in his pockets.

      

      

      
        
        > plant beans

        You're not holding the beans.

      

      

      
        
        > get beans

        You take the beans.

      

      

      
        
        > plant beans

        The ground is too hard.

      

      

      
        
        > north

        You're at the pit's north end.

        A giant mushroom is here. The

        ground is littered with the

        bodies of dead adventurers.

      

      

      
        
        > plant beans

        Apparently the beans are magical.

        They grow into a giant stalk.

      

      

      
        
        > climb stalk

        You have escaped!

      

      

      
        
        TRUE

        CLIPS>

      

      

      Player Help

      Rather than have the player guess which commands are available, it would be preferable to have a help command that lists the available commands alphabetically.

      
        
        Sorting

      

      

      To implement the help command, the do-for-all-facts query function will be used to construct the list of available commands and then the sort function will be used to order them alphabetically. The arguments to the sort function are a function name followed by a list of values to be sorted. For example, the > and < functions can be used to sort a list of numbers in ascending or descending order:

      
        
        CLIPS> (sort > 4 8 2 4 5 3)

        (2 3 4 4 5 8)

        CLIPS> (sort < 4 8 2 4 5 3)

        (8 5 4 4 3 2)

        CLIPS>

      

      

      The sort function uses the function specified as the first argument to determine if two values in the list need to be swapped. If the function returns FALSE, then the values are not swapped; otherwise they are swapped.

      CLIPS provides several predicate functions for comparing numbers that can be directly used with the sort function (such as the > and < functions in the prior example), but no such predicate function exists for comparing strings. The str-compare function, however, can be indirectly used by the sort function to alphabetize lists of strings and symbols. This function accepts two arguments and returns 1 if the first argument is lexicographically greater than the second argument; 0 if the first and second arguments are the same; and -1 if the first argument is lexicographically less than the second argument.

      To sort a list of strings into alphabetic order, the str-compare return values of 0 and -1 will need to be converted to FALSE when comparing two string values that are in the correct order. The deffunction construct will be used to create a new function that makes this conversion:

      
        
        CLIPS>

        (deffunction sort-string (?s1 ?s2)

           (> (str-compare ?s1 ?s2) 0))

        CLIPS>

      

      

      The deffunction keyword is followed by the name of the new function being defined. The parameter list follows the function name and is delimited by left and right parentheses. The parameter list consists of zero or more variable parameters. When a deffunction-defined function is called, the arguments passed to it are assigned to the parameters, which act as variables within the body of the function. Changes to these parameter variables within the scope of the function do not affect the values that were originally passed as arguments to the function. The parameter list of the function is followed by zero or more function calls. The return value of the function is the last function call evaluated.

      The sort-string function has two parameters: s1 and s2. When the function is called, the value of the variables s1 and s2 are passed to the str-compare function, which will return either -1, 0, or 1. This value is then compared to 0 using the > function. Thus if the first argument to the > function is -1 or 0, the return value of the > function will be FALSE; otherwise if the first argument is 1, the return value of the > function will be TRUE. Since the > function is the last function call in the deffunction, the return value of that function will also be the return value of the sort-string function.

      Once defined, the sort-string function can now be used with the sort function to sort strings:

      
        
        CLIPS> (sort sort-string "red" "green" "blue")

        ("blue" "green" "red")

        CLIPS>

      

      

      The sort-string deffunction will be added to definitions.clp for use in the next section.

      
        
        Implementing the Help Command

      

      

      Now that strings can be sorted alphabetically, the help command can be implemented. First, a new command pattern will be added to data.clp:

      
        
        (deffacts command_patterns_core

           .

           .

           .

           (command_pattern (text help)

                            (action help)))

      

      

      A new rule will be added to kernel.clp to process the help command:

      
        
        (defrule help

           ?c <- (command (action help))

           =>

           (retract ?c)

           (bind ?list (create$))

           (do-for-all-facts ((?f command_pattern)) TRUE

              (bind ?text (implode$ ?f:text))

              (bind ?list (create$ ?list ?text)))

           (bind ?list (sort sort-string ?list))

           (println "Commands I understand are:")

           (foreach ?item ?list

              (println "   " ?item)))

      

      

      The only pattern in the rule matches the help command:

      
        
        ?c <- (command (action help))

      

      

      In the actions of this rule, the create$ function is used to initially bind the variable list to a multifield value with no values:

      
        
        (bind ?list (create$))

      

      

      The create$ function creates a multifield value by appending all of its arguments together. If no arguments are specified, an empty multifield value (one with no values) will be created.

      The next action iterates over every command pattern and appends the associated command to the current value of the variable list:

      
        
        (do-for-all-facts ((?f command_pattern)) TRUE

           (bind ?text (implode$ ?f:text))

           (bind ?list (create$ ?list ?text)))

      

      

      Since the value of the text slot in the command pattern is a multifield value, the implode$ function is used to convert that value to a string, which is then appended to the variable list using the create$ function. This maps each command (possibly containing multiple symbols) to a single string. If the multifield value from the text slot had been appended instead, it would not be possible to tell where one command ends and another begins in the values in the variable list.

      Once the list of command strings is constructed, the sort function is used to order them alphabetically:

      
        
        (bind ?list (sort sort-string ?list))

      

      

      Finally, the last actions of the rule print the list of available commands:

      
        
        (println "Commands I understand are:")

        (foreach ?item ?list

           (println "   " ?item))

      

      

      
        
        Trying the Help Command

      

      

      The updated program can be run from the Chapter6E directory:

      
        
        CLIPS> (batch* runq.bat)

        Captured by goblins, you've been

        tossed in a pit at their lair.

      

      

      
        
        You're at the pit's north end.

        A giant mushroom is here. The

        ground is littered with the

        bodies of dead adventurers.

      

      

      
        
        > help

        Commands I understand are:

           climb <thing>

           climb down

           climb up

           down

           drop <thing>

           east

           examine <thing>

           get <thing>

           go down

           go east

           go north

           go south

           go up

           go west

           grab <thing>

           help

           inventory

           look

           look around

           look at <thing>

           north

           plant <thing>

           quit

           search <thing>

           south

           take <thing>

           up

           west

      

      

      
        
        > quit

        TRUE

        CLIPS>

      

      

      What’s Next

      The next chapter will introduce functionality that allows rules to be grouped together to provide an explicit mechanism for controlling the order in which they execute.

      Problems

      
        
        #6-1

      

      

      Modify the code so that the full description of a place is only printed once. For example, the first time the adventurer moves to the south side of the pit, this message should be displayed:

      
        
        You're at the pit's south end.

        A large pile of rubble has

        collapsed from the wall above.

      

      

      Subsequent moves to the south side of the pit should display this message:

      
        
        You're at the pit's south end.

      

      

      The full description of a place should still be printed when the adventurer gives a look or look around command.

      
        
        #6-2

      

      

      Modify the look_no_description and search_nothing_found rules to include the name of the item being examined or searched.

      
        
        #6-3

      

      

      Replace the search_rubble and search_goblin rules with a single generic rule for revealing an item contained within another item. You can use a generic message that includes the name of the revealed item in your rule:

      
        
        You find a dead goblin.

        You find some beans.

      

      

      
        
        #6-4

      

      

      Add generic rules to implement commands that just require a message to be displayed. Add this deftemplate for associating commands with messages to definitions.clp:

      
        
        (deftemplate action_default

           (multislot action)

           (multislot text))

      

      

      Test your code by adding this deffacts to game.clp:

      
        
        (deffacts action_defaults

           (command_pattern (text eat <thing>)

                            (action eat))

      

      

      
        
           (action_default

              (action eat <thing>)

              (text "Think about escape, not lunch."))

      

      

      
        
           (action_default

              (action eat beans)

              (text "They can be put to better use.")))

      

      

    

  



  
    
      
        
          
            Chapter 7

          

          
            Modules

          

        

      

    

    
      It’s common to have groups of related rules where each group needs to execute in order as part of a sequence. In prior chapters, salience has been used to ensure that some rules execute before others. For example, the general search_nothing_found rule was given a salience of -5 so that it would only execute if more specific rules (such as the search_rubble rule) did not.

      Rather than assigning different saliences to specific and general rules, it would be better to place the specific rules used only by Escape from the Goblin Pit in one group and the general core rules that can be used by any adventure in another group. If the specific group is processed before the core group, then the most specific rule will be applied first without the need to use salience.

      Related rules can be grouped together using the defmodule construct. Each module has its own agenda. Activations for rules belonging to a module are placed on that module’s agenda. During execution, each module updates its agenda in response to changes as facts are added and removed, but only activations from the agenda of the module with the current focus are eligible for execution. The focus command can be used to place modules on the focus stack. The module at the top of the focus stack always receives the current focus. When the module having the current focus no longer has any activations on its agenda, it is removed from the focus stack and the next module on the stack is given the current focus. When the focus stack is empty, execution halts.

      CLIPS automatically defines a module called MAIN. If no other modules are defined, CLIPS will place all constructs in this module. For example:

      
        
        CLIPS> (clear)

        CLIPS> (watch focus)

        CLIPS> (watch rules)

        CLIPS>

        (defrule hello_world

           =>

           (println "Hello, world!"))

        CLIPS> (list-focus-stack)

        MAIN

        CLIPS> (run)

        FIRE    1 hello_world: *

        Hello, world!

        <== Focus MAIN

        CLIPS> (list-focus-stack)

        CLIPS> (reset)

        ==> Focus MAIN

        CLIPS> (list-focus-stack)

        MAIN

        CLIPS> (reset)

        <== Focus MAIN

        ==> Focus MAIN

        CLIPS> (unwatch all)

        CLIPS>

      

      

      The focus keyword is another watchable option that can be used with the watch command. When this debugging option is enabled, a message will be printed whenever a change is made to the focus stack. The list-focus-stack command shows the list of modules on the focus stack. After the hello_world rule executes, the MAIN module is removed from the focus stack since its agenda is empty. An additional effect of the reset command is to remove all modules from the focus stack and then focus on the MAIN module.

      Module Overview

      In this chapter, the rules created in the prior chapters will be divided among six modules: MAIN, INITIAL, COMMAND, TRANSLATE, GAME, and CORE.

      The MAIN module will contain all deftemplate and deffunction constructs. In addition to grouping rules, modules can also be used to group other constructs that are shared among modules.

      The INITIAL module will contain rules (such as the exposition rule) that need to execute when the program starts.

      The COMMAND module will contain rules (such as the get_command rule) for requesting player commands.

      The TRANSLATE module will contain rules (such as the translate rule) for validating a player command and converting it into its canonical form.

      The GAME module will contain constructs specific to Escape from the Goblin Pit (such as the places deffacts and the search_rubble rule).

      Finally, the CORE module will contain rules (such as the search_nothing_found rule) for handling general commands common to all text adventures.

      Figure 7-1 shows the relationships between these modules.

      
        
          [image: MAIN module invokes INITIAL and COMMAND. COMMAND module invokes TRANSLATE. TRANSLATE module invokes GAME and CORE.]
        

        Figure 7-1
      
      The arrows in the diagram denote the program flow. The MAIN module will focus on the INITIAL and COMMAND modules. Once the rules in the INITIAL module have executed, the rules in the COMMAND module will request the next player command and then focus on the TRANSLATE module. If the TRANSLATE module is able to translate the command, it will focus on the GAME and CORE modules. The rules in the GAME module will have the opportunity to execute first, and then the rules in the CORE module will be able to execute. Once the GAME and CORE modules have finished execution, focus will return to the TRANSLATE module and then the COMMAND module. Rules in the COMMAND module will then request the next player command and the cycle of focusing on the TRANSLATE and then the GAME and CORE modules will repeat.

      The MAIN  Module

      The simplest way to define a module is to follow the defmodule keyword with the name of the module. The MAIN module is automatically defined with the following code when CLIPS is  launched or a clear command is issued:

      
        
        (defmodule MAIN)

      

      

      The module definition can also specify which constructs (such as deftemplates and deffunctions) in the module are exported (made available) to other modules. By default, a module does not export any constructs. For this program, every construct that needs to be utilized by other modules will be placed in the MAIN module, which will be redefined to export all constructs:

      
        
        (defmodule MAIN (export ?ALL))

      

      

      The addition of the (export ?ALL) syntax in the module definition indicates that every construct defined in the module is accessible to other modules.

      After a module is defined, any constructs that follow the definition (with the exception of another module definition) will be placed in that module. Placing the MAIN module redefinition at the top of definitions.clp will cause the remaining construct definitions in that file to be placed in the MAIN module as well. Any subsequent files loaded will also place their constructs in the MAIN module unless one of those files creates a new module definition.

      
        
        The Defglobal Construct

      

      

      One of the goals for using modules is to reduce the use of salience values in rules. Too many salience values can make it difficult to understand how a group of rules will interact with one another. Sometimes, however, judicious use of salience makes a program more maintainable than attempting to remove all salience values. To strike a balance between too few and too many salience values, it is helpful to explicitly define the set of salience values that will be used. For this program, this will be accomplished by adding the definition of a new type of construct to  definitions.clp after the MAIN module definition:

      
        
        (defglobal ?*high* = 10

                   ?*low* = -10)

      

      

      The defglobal construct allows the definition of one or more global variables with an initial value. Global variables can be distinguished from local variables used within rules and functions by the addition of the asterisks at the beginning and end of the variable name. Unlike a local variable, which can only be accessed within the rule or function in which it is defined, a global variable can be accessed by any construct. The bind function is used to change the value of a global variable. When a reset command is executed, the values of global variables are reset to the initial values given in their definitions.

      Once defined, the global variables high and low will have initial values of 10 and -10, respectively. These variables will be used later to assign salience values to rules. Using global variables for salience values makes it easier to change the salience for groups of rules as well as encouraging some consideration of the salience values used by rules.

      
        
        Module Specifiers

      

      

      The existing rules of this program will be spread across several modules, but none will be placed in the MAIN module since this module is being used for shared constructs. Since the MAIN module receives the current focus when a reset command is issued, a new rule will be added to the top of kernel.clp to focus on the other modules containing rules:

      
        
        (defrule MAIN::start

           =>

           (focus INITIAL COMMAND))

      

      

      In this case, the rule name MAIN::start uses a module specifier to specify both the rule name and the module in which it is defined. The symbol to the left of the :: character sequence is the name of the module, and the symbol to the right is the rule name. In order to use a module specifier, the module must already be defined. A module specifier not only indicates in which module the new defined construct is placed, but any subsequent constructs that lack a module specifier will be placed in this module as well.

      Since kernel.clp is loaded after definitions.clp, the start rule would be included in the MAIN module anyway, but using the module specifier explicitly indicates that this rule and any subsequent constructs will be placed in the MAIN module.

      The start rule has no conditions, so it will be automatically activated once defined or after a reset command. Since the MAIN module will initially be at the top of the focus stack, this rule will be the first to execute. The action of this rule uses the focus command to place the modules INITIAL and COMMAND on the focus stack. The modules are placed on the focus stack in the reverse order that they appear in the focus command, so the COMMAND module will first be placed on top of the MAIN module, and then the INITIAL module will be placed on top of the COMMAND module. Since the INITIAL module will now be on the top of the focus stack, it will receive the current focus.

      The CORE Module

      The CORE module will be added after the start rule in kernel.clp:

      
        
        (defmodule CORE (import MAIN ?ALL))

      

      

      This module will contain the print_place rule as well as the rules for the go, look, quit, help, take, drop, and inventory actions. The import syntax used in this definition indicates that the CORE module will import all of the constructs exported by the MAIN module.

      The search_nothing_found rule in kernel.clp will be modified to remove its salience:

      
        
        (defrule search_nothing_found

           (declare (salience -5))

           ?c <- (command (action search ?id))

           =>

           (retract ?c)

           (println "You find nothing of interest."))

      

      

      Since the GAME module will be processed before the CORE module, there is no longer any need for this rule to have a lower salience as the search_rubble and search_goblin rules will have the opportunity to execute before this general rule.

      The COMMAND Module

      The definition for the COMMAND module will be placed in kernel.clp after the rules that handle the inventory command:

      
        
        (defmodule COMMAND (import MAIN ?ALL))

      

      

      The bad_command rule in this module will have its salience modified:

      
        
        (defrule bad_command

          (declare (salience -10 ?*low*))

          ?c <- (command)

          =>

          (println "I don't understand your command.")

          (retract ?c))

      

      

      Its salience is now set to the global variable low (that has been set to -10) instead of directly setting it to -10. This implements the desired practice of establishing a set of predefined salience values. The salience for this rule needs to remain, so that it is executed only after an attempt is made to translate and process the player command. The salience for the rule could have been removed and the rule placed in a separate module to achieve the same result, but given that there’d only be a single rule in this module, it makes more sense just to keep it in the same module with the other related rules.

      The get_command rule in this module will have its salience removed:

      
        
        (defrule get_command

           (declare (salience -10))

           (not (command))

           =>

           (println)

           (print "> ")

           (bind ?rsp (explode$ (lowcase (readline))))

           (assert (command (text ?rsp)))

      

      

      A lower salience value is no longer necessary since there will be no other rules in the COMMAND module for processing and removing the player command (other than the bad_command rule).

      A new rule will be added to the COMMAND module that will initiate the translation of the player’s input:

      
        
        (defrule focus_translate

           (command (text ? $?) (action))

           =>

           (focus TRANSLATE))

      

      

      This rule is matched by a command fact that has at least one value in its text slot (which indicates there are values to translate) and no values in its action slot (which indicates no translation has been performed). The action of this rule focuses on the TRANSLATE module.

      When the game is first started, the player’s initial command is automatically asserted by the initial_state deffacts:

      
        
        (command (text look))

      

      

      The first time the COMMAND module has the current focus, the get_command rule will not be applicable since a command fact exists. The focus_translate rule, however, will be applicable, so when it executes, the TRANSLATE module will receive focus.

      Upon return to the COMMAND module, the look command will have been processed, removing the command fact. Now the get_command rule will be applicable, and when it executes it will assert a command fact, causing the focus_translate rule to be activated again.

      The TRANSLATE Module

      The definition for the TRANSLATE module will be placed in kernel.clp after the focus_translate rule:

      
        
        (defmodule TRANSLATE (import MAIN ?ALL))

      

      

      A new rule will be added to the TRANSLATE module that will initiate the processing of the player command when a successful translation is made and at least one value has been assigned to the action slot of the command fact:

      
        
        (defrule process_action

           (command (action ? $?))

           =>

           (focus GAME CORE))

      

      

      When this rule executes, the GAME module will receive the current focus, and subsequently the CORE module will receive the current focus when there are no more rules to execute in the GAME module.

      The thing_must_be_held rule will have its salience value modified to the global variable high:

      
        
        (defrule thing_must_be_held

           (declare (salience 10 ?*high*))

           .

           .

           .

           (println "You're not holding " ?definite "."))

      

      

      Finally, the translate_one_thing_no_match rule will have its salience value modified to the global variable low:

      
        
        (defrule translate_one_thing_no_match

           (declare (salience -5 ?*low*))

           .

           .

           .

           (println "You can't see any such thing."))

      

      

      Since this rule and the bad_command rule are now in separate modules, they can share the same salience value without conflicting.

      The INITIAL Module

      The INITIAL module will contain startup code that is specific to a text adventure. It will be added at the end of kernel.clp:

      
        
        (defmodule INITIAL (import MAIN ?ALL))

      

      

      This definition is included as part of the general code for any adventure. This allows the INITIAL module to receive focus even if the designer of a text adventure does not have any startup code. In that case, the module can receive the current focus, and since there will be no rules to execute, the module will be immediately removed from the focus stack.

      A module specifier will be used to place the exposition rule in game.clp in the INITIAL module:

      
        
        (defrule INITIAL::exposition

           (declare (salience 10))

           =>

           (println "Captured by goblins, you've been")

           (println "tossed in a pit at their lair.")

           (println))

      

      

      Since the exposition rule is in a separate module from the rules that will process the initial player look command and the INITIAL module receives focus first, there is no longer any need to assign this rule a salience value. The exposition text for the adventure will always be printed before the description of the adventurer’s starting location.

      The GAME Module

      The final module to create is the GAME module, which will contain all other code that is specific to a text adventure. It will be placed after the exposition rule in game.clp:

      
        
        (defmodule GAME (import MAIN ?ALL))

      

      

      The salience values for the search_rubble and search_goblin rules can now be removed:

      
        
        (defrule search_rubble

           (declare (salience 5))

           .

           .

           .

           )

      

      

      
        
        (defrule search_goblin

           (declare (salience 5))

           .

           .

           .

           )

      

      

      Since the GAME module receives focus before the CORE module, there is no longer a need to assign these rules a higher salience so they will execute before the general rules for processing a search command.

      Watching The Modules

      The module redesign of this program is now complete, and the number of salience values used by the rules has been reduced from four to two. The updated program can be loaded from the Chapter7A directory to observe how the changed rules and modules interact:

      
        
        CLIPS> (batch* loadq.bat)

        TRUE

        CLIPS> (agenda)

        0      start: *

        For a total of 1 activation.

        CLIPS> (list-focus-stack)

        MAIN

        CLIPS>

      

      

      After the commands in loadq.bat have loaded the program and reset CLIPS, the start rule is now on the agenda and the MAIN module is at the top of the focus stack.

      If the agenda command is called without any arguments, only the activations on the agenda for the module with the current focus are displayed. The agendas of other modules can be displayed by specifying the module name as an additional argument to the agenda command:

      
        
        CLIPS> (agenda INITIAL)

        0      exposition: *

        For a total of 1 activation.

        CLIPS>

      

      

      Specifying * as the module name will display all agendas:

      
        
        CLIPS> (agenda *)

        MAIN:

           0      start: *

        CORE:

        COMMAND:

           0      focus_translate: f-1

           -10     bad_command: f-1

        TRANSLATE:

           0      translate: f-1,f-24

        INITIAL:

           0      exposition: *

        ADVENTURE:

        For a total of 5 activations.

        CLIPS>

      

      

      This output shows that even though the MAIN module has the current focus, the rules in other modules are being matched and their agendas are being updated. When using one of the CLIPS IDEs, the agenda browser (introduced in Chapter 3) can also be used for displaying the contents of the focus stack and the agenda of the selected module from the focus stack. This is usually more convenient than entering commands at the REPL.

      Allowing the start rule to execute will push the COMMAND and then the INITIAL modules onto the focus stack:

      
        
        CLIPS> (watch focus)

        CLIPS> (watch rules)

        CLIPS> (watch activations)

        CLIPS> (run 1)

        FIRE    1 start: *

        ==> Focus COMMAND from MAIN

        ==> Focus INITIAL from COMMAND

        CLIPS> (list-focus-stack)

        INITIAL

        COMMAND

        MAIN

        CLIPS> (agenda)

        0      exposition: *

        For a total of 1 activation.

        CLIPS>

      

      

      The current focus is now the INITIAL module, and the previously activated exposition rule is ready to be executed:

      
        
        CLIPS> (run 1)

        FIRE    1 exposition: *

        Captured by goblins, you've been

        tossed in a pit at their lair.

      

      

      
        
        <== Focus INITIAL to COMMAND

        CLIPS>

      

      

      The exposition rule prints the starting message for the text adventure. Since the INITIAL module no longer has any activations on its agenda, it is removed from the focus stack.

      Now the COMMAND module has the current focus, and the focus_translate rule is ready to execute to initiate translation of the player’s initial look command.

      
        
        CLIPS> (agenda)

        0      focus_translate: f-1

        -10    bad_command: f-1

        For a total of 2 activations.

        CLIPS> (ppfact 1)

        (command

           (text look)

           (action))

        CLIPS>

      

      

      Allowing the rule to execute will push the TRANSLATE module onto the focus stack:

      
        
        CLIPS> (run 1)

        FIRE    1 focus_translate: f-1

        ==> Focus TRANSLATE from COMMAND

        CLIPS> (list-focus-stack)

        TRANSLATE

        COMMAND

        MAIN

        CLIPS>

      

      

      The translate rule is now ready to fire:

      
        
        CLIPS> (agenda)

        0      translate: f-1,f-24

        For a total of 1 activation.

        CLIPS> (watch facts)

        CLIPS> (run 1)

        FIRE    1 translate: f-1,f-24

        <== f-1     (command ... (action))

        <== Activation -10    bad_command: f-1

        ==> Activation 0      get_command: *

        ==> f-1     (command ... (action look))

        ==> Activation 0      process_action: f-1

        <== Activation 0      get_command: *

        ==> Activation -10    bad_command: f-1

        ==> Activation 0      print_place: f-1,f-28,f-29

        CLIPS>

      

      

      When the translate rule fires, it assigns look to the action slot of the command fact. This causes several activations to be added and removed as the fact is first retracted and then asserted with the modified value. The net result is that the process_action rule is activated (which will focus on the CORE and GAME modules), the bad_command rule is activated (which will eventually fire if the command is not processed by another rule), and the print_place rule is also activated (which will process the look command).

      The process_action rule is now ready to fire:

      
        
        CLIPS> (agenda)

        0      process_action: f-1

        For a total of 1 activation.

        CLIPS> (run 1)

        FIRE    1 process_action: f-1

        ==> Focus CORE from TRANSLATE

        ==> Focus GAME from CORE

        <== Focus GAME to CORE

        CLIPS> (list-focus-stack)

        CORE

        TRANSLATE

        COMMAND

        MAIN

        CLIPS> (agenda)

        0      print_place: f-1,f-28,f-29

        For a total of 1 activation.

        CLIPS>

      

      

      Both the CORE and GAME modules are pushed onto the focus stack, but since there are no activations in the GAME module (because the generic look rules in the CORE module handle this situation), the GAME module is immediately removed from the stack.

      Allowing the print_place rule to execute will print the description for the north side of the pit:

      
        
        CLIPS> (run 1)

        FIRE    1 print_place: f-1,f-28,f-29

        <== f-1     (command (text look) (action look))

        <== Activation -10    bad_command: f-1

        ==> Activation 0      get_command: *

        You're at the pit's north end.

        A giant mushroom is here. The

        ground is littered with the

        bodies of dead adventurers.

        <== Focus CORE to TRANSLATE

        <== Focus TRANSLATE to COMMAND

        CLIPS> (list-focus-stack)

        COMMAND

        MAIN

        CLIPS> (agenda)

        0      get_command: *

        For a total of 1 activation.

        CLIPS> (unwatch all)

        CLIPS>

      

      

      In addition to printing the place description, processing the look command removes the command fact, and as a result the bad_command activation is removed from the agenda and an activation for the get_command rule is added. Since there are no activations remaining on the agenda of the CORE and TRANSLATE modules, these are removed from the focus stack and the COMMAND module now has the current focus.

      At this point, continuing to run the program and entering additional commands would similarly invoke the TRANSLATE, CORE, and GAME modules.

      What’s Next

      Escape from the Goblin Pit is almost complete. The next chapter will add a few finishing touches and then the game will be ready to distribute.

      Now that the game can be won, there are just a few remaining steps before it’s ready for others to play. First, the player should be able to quit the game and return later to continue playing. Second, there needs to be a framework to allow automated testing of the game to detect bugs and ensure that future additions to the game do not break what’s already working. Finally, the application needs to be deployed in a format that’s suitable for distribution. That will require embedding CLIPS in an executable that does not require the player to use the REPL to load and run the game.

      Remembering The Game State

      
        
        Saving the Game State

      

      

      To allow game progress to be saved and restored, the quit rule in kernel.clp will be modified to save the list of facts representing the current game state:

      
        
        (defrule quit

           ?c <- (command (action quit))

           =>

           (retract ?c)

           (bsave-facts "game.fct" visible)

           (halt))

      

      

      The bsave-facts command saves the existing facts in a binary format to the file game.fct. The visible keyword indicates that facts with deftemplates defined in the current module or imported by the current module will be saved. By default, only those facts with deftemplates defined in the current module are saved.

      In addition to the bsave-facts command, there is also a save-facts command that saves facts in a plain text format rather than a binary format. Saving with a binary format obfuscates the information contained within the file; players seeking “hints” can glean some information from the binary format because text strings will be embedded within the file, but it is difficult to casually modify the saved binary game state.

      
        
        Restoring the Game State

      

      

      The start rule in kernel.clp will also be modified to look for and load the saved game state:

      
        
        (defrule MAIN::start

           =>

           (if (open game.fct game)

              then

              (close game)

              (retract *)

              (bload-facts game.fct)

              (assert (command (text look)))

              (focus COMMAND)

              else

              (focus INITIAL COMMAND)))

      

      

      The actions of this rule include a call to the if function, which is another special form. Similar to many other programming languages, the if function evaluates the expression that follows the function name, and if the expression evaluates to a value that is not FALSE, the actions following the then keyword are evaluated. Optionally, the else keyword can be specified. If so and the expression evaluates to FALSE, then the actions following the else keyword are evaluated.

      The if statement in this rule evaluates a call to the open function. The open function opens a file and associates it with a logical name that can be referenced by other functions for performing input/output operations. By default, a file is opened for reading input. In this rule, the function attempts to open game.fct and associate it with the logical name game. The open function returns TRUE if it was successfully able to open a file; otherwise it returns FALSE.

      If the function returns TRUE, then game.fct exists and its binary fact content can be loaded by actions in the then portion of the if special form. First, the close function is called and passed the logical name game. This will close the file that was just opened in preparation for using the  bload-facts command to load the saved game state. Before loading this new state, the initial game state must be removed. To do this, the retract command is called passing the argument *, which causes all facts to be retracted. The bload-facts function is then called to load the saved game state. Next, a command fact is asserted to create a look command to describe the adventurer’s current location since this fact was retracted along with the initial game state. The focus command is then used to place the COMMAND module on the focus stack. The actions performed by the INITIAL module have previously been processed, so this module no longer needs to be placed on the focus stack.

      If the game state was not saved, then the actions in the else portion of the if special form are executed, which place the INITIAL and COMMAND modules on the focus stack as was done previously for the initial game state.

      
        
        Removing the Game State

      

      

      Once the adventurer has escaped, the game state must be removed so that the next time the player starts the game, they’ll restart at the beginning. The climb_stalk rule in game.clp will be modified to delete game.fct:

      
        
        (defrule climb_stalk

           ?c <- (command (action climb stalk))

           (thing (id adventurer)

                  (location ?location))

           (thing (id stalk)

                  (location ?location))

           =>

           (retract ?c)

           (println "You have escaped!")

           (println)

           (remove game.fct)

           (halt))

      

      

      The remove function call in the actions of this rule will delete game.fct before execution is halted.

      Restarting The Game

      A restart command fact will also be added to the deffacts in data.clp to give the player the explicit ability to restart the game:

      
        
        (deffacts command_patterns_core

           .

           .

           .

           (command_pattern (verb restart)

                            (text restart)))

      

      

      
        
        Asking a Question

      

      

      Since the restart command will discard all of the player’s progress, they should be asked for confirmation before proceeding. A new deffunction that repeatedly asks the player a question until a valid response is given will be added to definitions.clp:

      
        
        (deffunction ask-question (?question $?values)

           (bind ?answer (void))

           (while (not (member$ ?answer ?values)) do

              (print ?question)

              (bind ?answer (read))

              (if (lexemep ?answer)

                 then (bind ?answer (lowcase ?answer))))

           ?answer)

      

      

      The ask-question function has two parameters. The parameter question is a string with the text of the question to be asked. The parameter values is a multifield value containing all of the responses that are valid. Normally, the parameters to a deffunction are specified using single-field variables (even if they contain multifield values). Using a multifield variable in a pattern of a rule indicates that it can be bound to any number of values. Using a multifield variable as the last parameter in a deffunction indicates that the function can take a variable number of arguments (and the last parameter is referred to as a wildcard parameter). When invoked, a deffunction must be supplied with an argument for each single-field variable in its parameter list. Any remaining arguments will be combined into a multifield value that will be assigned to the wildcard parameter. If a deffunction does not specify a wildcard parameter, then the number of arguments passed to the function must exactly match the number of parameters.

      The actions of the ask-question deffunction can be broken down into several parts. First, the variable answer is bound to the return value of the void function:

      
        
        (bind ?answer (void))

      

      

      In many programming languages, the void keyword is used to indicate a function that returns no value. In CLIPS, you can use the void function to return a value that indicates no value. By binding this value to a variable, it can be used to indicate that no value has yet been assigned to that variable. The void function can also be returned by a deffunction if for whatever reason you don’t want a function to return a value. It is a bit of an oxymoron to have a “no value” value, but it can only be used in these limited circumstances; you can’t, for example, assign it as a slot value in a fact.

      The next line of code in the deffunction initiates a loop:

      
        
        (while (not (member$ ?answer ?values)) do

      

      

      The while function is another special form. The expression following the function name (the condition) is evaluated, and if the return value is any value other than FALSE, then the actions in the body of the while function are executed. The body follows the expression, and optionally the do keyword can be placed between the two.

      After the body finishes executing, the condition for the while loop is again evaluated, and the body executes if the condition result is not FALSE. This process loop of evaluating the expression and executing the body repeats until the expression evaluates to FALSE.

      For this while loop, the member$ function determines whether or not the value in the variable answer is contained within the multifield variable values. This loop will continue as long as the value is not found. Since the variable answer is initially set to the “no value” value from the void function, the loop will be executed at least once.

      The next four lines of the deffunction code comprise the body of the loop:

      
        
        (print ?question)

        (bind ?answer (read))

        (if (lexemep ?answer)

           then (bind ?answer (lowcase ?answer))))

      

      

      The first line prints the question. The second line binds the variable answer to the return value of the read function. This function is similar to the readline function except rather than returning an entire line of input, it returns just the first primitive value that can be parsed from the input. Like the readline function, the player must still press a carriage return before a value is returned. For example, if the user types go north followed by a carriage return, the readline function would return "go north" but the read function would return go.

      The final two lines allow the case of the characters to be ignored when determining whether the player’s answer matches a valid answer. For example, if one of the valid answers is yes, then yes, Yes, YES, and any other combination of upper and lower case in the characters yes will be considered a valid answer.

      The lexemep function is used to determine if the value in the variable answer is a string or symbol. If so, then this value is converted to lowercase using the lowcase function. Thus the member$ call in the while loop will always compare a lowercase string or symbol to the valid answers. This assumes that any strings or symbols in the valid answers are in lowercase as well, although for completeness it is easy to add some additional code that converts any symbols or strings in the valid answers to lowercase as well.

      Finally, the last line of the deffunction contains just the variable answer:

      
        
        ?answer)

      

      

      The return value of a function is simply the last expression that’s evaluated, so once a valid answer is given and the while loop exits, the variable answer is evaluated and that’s the return value of the function. The return function can also be used to return from a function, so the last line of the deffunction could have been:

      
        
        (return ?answer))

      

      

      Typically the return function is used to return from the body of a function (such as within the body of an if or while special form) before the last line of code is reached.

      
        
        Asking a Yes or No Question

      

      

      The ask-question function can now be used to create another deffunction in definitions.clp for asking the player a yes or no question:

      
        
        (deffunction yes-or-no-p (?question)

           (bind ?answer

              (ask-question ?question yes no y n))

           (if (or (eq ?answer yes) (eq ?answer y))

              then yes

              else no))

      

      

      The yes-or-no-p function accepts a single parameter, a question expecting a yes or no response, and returns either yes or no. The first line of code passes five arguments to the ask‑question function:

      
        
        (bind ?answer

           (ask-question ?question yes no y n))

      

      

      The first argument is the parameter question, which is passed directly. The remaining four arguments

      
        
        yes no y n

      

      

      are combined into the multifield value

      
        
        (yes no y n)

      

      

      and assigned to the wildcard parameter variable values in the ask-question deffunction.

      The final three lines of the yes-or-no-p deffunction  return the player’s answer:

      
        
        (if (or (eq ?answer yes) (eq ?answer y))

           then yes

           else no))

      

      

      Since the if special form is the last expression evaluated in the yes-or-no-p function, the values yes or no can be placed in the then and else portions and those values will be the return value of the yes-or-no-p function.

      
        
        The Restart Command

      

      

      With the yes-or-no-p function defined, a new rule will be added to kernel.clp to implement the restart command:

      
        
        (defrule restart

           ?c <- (command (action restart))

           =>

           (retract ?c)

           (if (eq (yes-or-no-p "Restart game? ") yes)

              then

              (remove game.fct)

              (reset)))

      

      

      If the player affirms that the game should be restarted, the saved game state will be discarded by deleting game.fct, and then the reset command will restore the original game state.

      Testing the Program

      As you add, remove, and modify CLIPS code, it’s important to verify that changes haven’t unintentionally broken the behavior of your program. This can be accomplished with regression testing. In prior chapters, the batch command was used to automatically execute a series of commands to load and start execution of the text adventure. The batch command can also be used to simulate the player interacting with the game.

      For the text adventure regression tests, the batch command will be used to run through various scenarios, and the output of those interactions will be captured in a file. The content of the actual output of the program can then be compared to the expected output to see if anything has changed.

      The testing framework will be created in a Testing subdirectory of the directory containing the rules. Since the rules won’t be in the same directory as the framework, a new batch file called loadq.bat will be created in the Testing directory with content that is similar to the same file from the rules directory:

      
        
        (unwatch all)

        (clear)

        (load* "../definitions.clp")

        (load* "../kernel.clp")

        (load* "../data.clp")

        (load* "../game.clp")

        (reset)

        (remove game.fct)

      

      

      The first difference between this file and the prior version is that the rules are loaded from the parent directory rather than the current directory. Using ../ in the file path to specify the parent directory will work with Windows, macOS, and Linux. The second difference is the addition of the remove command to delete game.fct. This allows each test case to run independently of any saved progress made in a previous test case.

      
        
        Testing the North Side of the Pit

      

      

      The first test case will test some of the actions that the adventurer can take on the north side of the pit. The tst file extension will be used for batch files that run a test case. For the north side test case, create the file northpit.tst with the following content in the Testing directory:

      
        
        (batch* loadq.bat)

        (dribble-on "Actual/northpit.out")

        (batch "northpit.bat")

        (dribble-off)

      

      

      The batch* command will process the commands in loadq.bat to load and reset the program. The dribble-on command is used to capture I/O and place it in northpit.out in the Actual subdirectory (which you will need to create in the Testing directory). The dribble-on command will look for any I/O read from standard input or written to standard output and write it to the dribble file as well. Next, the batch command is used to process the commands in northpit.bat (which will be discussed next). Finally, once the commands from northpit.bat are processed, the dribble-off command is used to stop capturing output to northpit.out.

      Next, create the file northpit.bat with the following content in the Testing directory:

      
        
        (run)

        look at mushroom

        look at bodies

        go up

        quit

      

      

      This file begins execution of the test case using the run command and contains the simulated player input to test commands on the north side of the pit. There should be a carriage return after the final line of input for the quit command.

      
        
        The Difference Between the Batch and Batch* Commands

      

      

      When the batch command is invoked, it does not immediately execute the commands contained inside the batch file. Instead the batch file is opened, and the content is only processed when CLIPS makes a request from standard input. So when a batch command is executed from the command prompt, CLIPS only begins reading from the batch file when the prompt next appears and only reads until it finds a complete command. If the carriage return is left off the end of the last command, CLIPS will read up to that point and will require the user to enter the final carriage return before executing the command. When a batch command is called within a batch command, a stack of batch files is created that is used when input is requested from standard input. When the batch file at the top of the stack no longer has any content to be read, it’s removed from the top of the stack, and input is then read from the next batch file. This continues until there are no batch files that remain open. Subsequent input is then read from the keyboard.

      For this test case, the batch files northpit.tst and northpit.bat will both be open when the run command is executed. When a command is requested from the player by the get_command rule, the line of text look at mushroom will be read from northpit.bat and returned as input to the rule. Other rules will process that command, and then northpit.bat will provide input again when the get_command rule is next executed. This will continue until the quit command is reached at the bottom of northpit.bat. Then, the run command will be halted, northpit.bat will be removed from the stack of open batch files, and control will return to the REPL. When the REPL requests input, the dribble-off command will be read from northpit.tst, and when executed, northpit.out will be closed. Since all input has been read from northpit.tst, it will be removed from the stack of batch files. The REPL will now wait for input from the keyboard.

      There’s a difference between the batch and batch* commands that’s important. The batch* command isn’t just a different version of the batch command that doesn’t display the output of the commands it executes. The batch command redirects any requests from standard input to instead take input from the contents of the batch file. So if no requests are made from standard input, the contents of the batch file will never be processed. The batch* command immediately parses and executes the commands contained in the batch* file. This means that you can place a function inside the batch* file that requests input from standard input, but the input will not be read from the batch* file (and will either be read from the keyboard or a batch file). So for these test cases, the batch command needs to be used to simulate player input for the game.

      
        
        Testing the South Side of the Pit

      

      

      A test case similar to the north side of the pit can be created for the south side of the pit. Create the file southpit.tst file in the Testing directory with the following content:

      
        
        (batch* loadq.bat)

        (dribble-on "Actual/southpit.out")

        (batch "southpit.bat")

        (dribble-off)

      

      

      Next create the file southpit.bat with the following content in the Testing directory:

      
        
        (run)

        south

        look at rubble

        go up

        search rubble

        search goblin

        get beans

        inventory

        quit

      

      

      
        
        Running The Test Cases

      

      

      Finally, create the file testall.tst with the following content that will run all of the test cases:

      
        
        (batch northpit.tst)

        (batch southpit.tst)

      

      

      Once testall.tst in the Testing directory from the Chapter8A directory is processed by the batch command, the Actual directory will contain the output files northpit.out and southpit.out from the two test cases. For example, northpit.out will contain the following content:

      
        
        TRUE

        CLIPS> (batch "northpit.bat")

        TRUE

        CLIPS> (run)

        Captured by goblins, you've been

        tossed in a pit at their lair.

      

      

      
        
        You're at the pit's north end.

        A giant mushroom is here. The

        ground is littered with the

        bodies of dead adventurers.

      

      

      
        
        > look at mushroom

        It's squished. I wouldn't

        try landing on it again.

      

      

      
        
        > look at bodies

        Apparently this is what happens

        when you miss the mushroom.

      

      

      
        
        > go up

        The walls are too slick.

      

      

      
        
        > quit

        CLIPS> (dribble-off)

      

      

      Once the content of the output files has been verified to be correct, create an Expected directory in the Testing directory and move the verified output to this new directory. When the test cases are subsequently run, a file comparison utility can be used to compare the contents of the Actual and Expected directories to see if any changes have occurred. On macOS, I use BBEdit from Bare Bones Software for comparing files; on Windows, I use WinMerge from Thingamahoochie Software; and on Linux, I use DiffMerge from SourceGear.

      Distributing The Program

      Escape from the Goblin Pit is almost ready for distribution. However, the code for the game is stored in text files that players can easily read, so it would be preferable to obfuscate the code so that the players aren’t tempted to peek. Also, rather than having players launch CLIPS and issue commands from the REPL to start the game, it’s preferable to create an executable that will automatically do this when launched.

      
        
        The Bsave and Bload Commands

      

      

      The text code for the game can be save in a binary format to a single file using the bsave command:

      
        
        CLIPS> (batch* loadq.bat)

        TRUE

        CLIPS> (bsave escape.bin)

        [CSTRNBIN1] WARNING: Constraints are not saved with a binary image when dynamic constraint checking is disabled.

        TRUE

        CLIPS>

      

      

      The bsave command is similar to the bsave-facts command except that it creates a binary file of constructs rather than facts. All of the constructs for the adventure program are now contained in the single file escape.bin.

      Constraints on the valid values for fact slots, such as the allowed-values keyword, are normally checked only when code is parsed (e.g., for constructs loaded from a file or entered at the REPL). Dynamic constraint checking occurs when facts are asserted by a running program (and this is off by default and must be explicitly enabled). So the warning issued when the bsave command is processed just indicates that the constraints that are associated with a deftemplate for dynamic constraint checking will not be saved if dynamic constraint checking is disabled.

      The binary file can be loaded using the bload command (after first removing game.fct so that the game starts at the beginning):

      
        
        CLIPS> (bload escape.bin)

        TRUE

        CLIPS> (reset)

        CLIPS> (run)

        Captured by goblins, you've been

        tossed in a pit at their lair.

      

      

      
        
        You're at the pit's north end.

        A giant mushroom is here. The

        ground is littered with the

        bodies of dead adventurers.

      

      

      
        
        > quit

        CLIPS>

      

      

      The bload command automatically removes all existing constructs before loading the contents of the binary file. For large programs, loading a binary file can be much faster than loading the equivalent text files with constructs. When a text file of constructs is loaded, the text must be parsed and analyzed to create the data structures for implementing those constructs. The content of the binary file already contains those data structures, so no parsing or analysis is needed. In addition, the contents of the binary file are read in large blocks, which is typically faster than reading character by character from a text file (even if the reads are buffered).

      Saving constructs as binary content does have some restrictions, however: you can only have the content of one binary file loaded at a time, you can’t load constructs from a text file while binary content is loaded, and the files are generally not portable to other software/hardware platforms.

      
        
        Embedding the Program

      

      

      The last step needed before distributing Escape from the Goblin Pit is to embed CLIPS within a C program so that players do not need to interact with the REPL to load and run the program. The source directory in the Chapter8B directory contains the source code for CLIPS. It also contains an additional file named adventure.c, which includes a replacement for the main function in main.c normally used by CLIPS. If you’re not familiar with the C programming language, the main function is the one that is first executed when a program is run.

      The main function in main.c used when CLIPS is run as a terminal application initializes CLIPS and starts the REPL. The replacement main function in adventure.c will load and launch the adventure game instead:

      
        
        int main(

          int argc,

          char *argv[])

          {

           Environment *theEnv;

      

      

      
        
           theEnv = CreateEnvironment();

      

      

      
        
           if (Bload(theEnv,"escape.bin"))

             {

              Reset(theEnv);

              Run(theEnv,-1);

             }

      

      

      
        
           DestroyEnvironment(theEnv);

      

      

      
        
           return 0;

          }

      

      

      The CreateEnvironment function creates and initializes an environment in which a CLIPS program can be loaded and run. The Bload, Reset, and Run functions are the C equivalents of the CLIPS bload, reset, and run commands. If the Bload function is able to load escape.bin, then the Reset and Run functions will be called to start the text adventure running; otherwise, the Bload function prints an error message and the Reset and Run functions are not called. Once the Bload function fails or the player issues a quit command, the DestroyEnvironment function is called to destroy the environment that was created, and then the main function completes.

      On macOS and Linux, the game executable can be created by launching a terminal application, navigating to the source directory, renaming the file makefile.unx to makefile, and then executing a make command:

      
        
        .../Chapter8B/source$ make

           .

           .

           .

        .../Chapter8B/source$

      

      

      Once compilation is complete, you can move the escape executable to the directory containing escape.bin, remove game.fct so that the game starts at the beginning, and then start the game with a ./escape command:

      
        
        .../Chapter8B$ ./escape

        Captured by goblins, you've been

        tossed in a pit at their lair.

      

      

      
        
        You're at the pit's north end.

        A giant mushroom is here. The

        ground is littered with the

        bodies of dead adventurers.

      

        

      
        >

      

      

      On Windows, the game executable can be created by launching a Visual Studio Command Prompt, navigating to the source directory, renaming the file makefile.win to makefile, and then executing an nmake command:

      
        
        c:\Users\...\Chapter8B\source> nmake

           .

           .

           .

        c:\Users\...\Chapter8B\source>

      

      

      Once compilation is complete, you can move the escape.exe executable to the directory containing escape.bin and start the game with an escape command.

      
        
        c:\Users\...\Chapter8B> escape

        Captured by goblins, you've been

        tossed in a pit at their lair.

      

      

      
        
        You're at the pit's north end.

        A giant mushroom is here. The

        ground is littered with the

        bodies of dead adventurers.

      

        

      
        >

      

      

      What’s Next

      Game over, man. You’ve reached the end of the book, but your CLIPS adventure doesn't have to end here. You can hone your CLIPS programming skills by adding additional rooms and puzzles to the goblin pit or create a completely new adventure of your own. Additional examples of CLIPS programs can be downloaded from this web page:

      
        
        sourceforge.net/p/clipsrules/code/HEAD/tree/branches/64x/examples/

      

      

      There are also several online forums where you can ask questions about CLIPS:

      
        
        groups.google.com/g/CLIPSESG

        sourceforge.net/p/clipsrules/discussion/

        stackoverflow.com/questions/tagged/clips

